
マニュアルの表記方法について 1

マニュアルの表記方法について

書体と色

このマニュアルは、LightWave 3D [7] 日本語マニュアルの表記方法に準じて記述されていますが、
プログラミングの解説という性質上、以下の違いがあります。

機能や用語

LScriptやLightWave 3Dの機能、メニュー名、および一般的なプログラミング用語が初めて出て
くる箇所は、黒の太字で書かれています。

例：

このような場合、変数を使えばスクリプト実行中に、このデータを…

コマンドや構文

コマンドや構文は、青で書かれています。

例：

main

{

// このスクリプトは infoボックスを画面上に表示します

// 変数テキストの値は "Hello World!"です

…

}

コメント行

本文のスクリプト中に、"//"や"/*"で囲まれたコメント行があります。このマニュアルでは、
分かりやすく解説するためにコメント行を日本語で記述していますが、実際のLScriptでは英数字
しか使用できません。

第１章：スクリプト概論 1.1

第 1 章 : スクリプト概論

スクリプトの作成法を学びたいのですね？素晴らしいことです！独自のスクリプトを作成出来る
ということは、アーティストにとって有力な技能です。望みどおりのカスタムツールが作成できれ
ば、組織やLightWaveコミュニティにとってあなたは貴重な人材となるでしょう。またアーティス
トとプログラマという両方の側面を持つ非常に稀な人材となります。素晴らしいツールを設計し、
特定の人々つまりアーティストのために特別なツールを作ることが出来る人材になるのです。

NewTekのようなソフトウェア開発者にとって、ユーザーが必要とする機能を、必要な時に正確
に処理出来る LightWaveツールを設計するなんてことは、とても出来るものではありません。
LightWaveのユーザーは広範囲にわたっており、フィルムやビデオ、印刷やゲームなどのユーザー
はカスタムツールをますます必要とするようになるでしょう。しかしNewTekはLightWaveコミュ
ニティが全体として満足できるようにツールセットを設計しました。このツールセットは
LightWaveの核から派生したものであり、独自のツールが作成できます。それがLScriptです。

LightWaveのスクリプト言語を学び始める前に、一つはっきりさせておかなければなりません。
スクリプトはプログラム形式です。こんなことを聞くと芸術家体質の体中の骨はきしんでしまうこ
とでしょう。でもLScriptでは通常のプログラミングから苦痛と悩みが取り去られていると保証し
ます。CやC++言語での機能性の大半を保ちながらも、それに伴う頭痛の種は取り除きました。

簡単だとは言い切れません。ここからは今までアーティストにとっては無縁だった話題や手続き
などについてカバーしていきます。章を読み進み、順番どおりにサンプルを実行しながら一度に一
つずつ段階を重ね全てを根気良く続けていけば、きっと楽しくなってくるはずです。LScriptの素晴
らしいプログラミングの世界をぜひ楽しんでください。



この本を読むべきは人は誰？
LightWaveのスクリプト言語でプログラミング方法を学ぶためには、まずLightWaveがどのよう

に動作しているのかについて理解しておかなければなりません。基本的なモデリングやアニメー
ションについて話題に上りますし、サーフェイスやポリゴンも関係してきます。たとえ他にアプリ
ケーションを十分に理解しており、LightWaveを学び始めたばかりだとしても、LScriptを学び始め
る前にまずLightWaveの基本的な技術を習得しておいてください。

LScriptのツールはLightWaveにあるツールのサブセットとなっていますから、これから解説し
実践していく内容の大半は、LightWaveとLScriptがどのようにして相互に対話しているのかとい
うことと関わります。

このことを踏まえておかないと、わけもなくテーブルに頭を打ち付けたくなるかもしれません。

このマニュアルはLScriptでプログラミングを学ぼうとしている人を意図して設計されています。
主に初心者や趣味で使用する人を養成するのが目的です。もちろん専門家でも大歓迎ですが、専門
家が望むよりもはるかにゆっくりとしたスピードで進めていくことになります。また専門家には必
要とされる、より難しいテクニックについてここでは取り組みません。経験を積んだプログラマに
はLScriptドキュメントやリリースノートを読むことをお勧めします。

ですからLightWaveの動作については良く理解しており、なおかつ未知なるプログラミング領域
に足を踏み入れたい人に、最適なマニュアルといえるでしょう。

何を学ぶの？
プログラミング習得において最も大変な個所といえば、初めてのスクリプトを作る前にまず、プ

ログラミングのあらゆる面について少しずつ学んでおかなければならない点にあります。プログラ
ミングについてより理解を深め知識が増えてくるにつれて、既知の知識を広げていくことが出来る
のです。ここでは学生時代に教師から教わったようにきちんとした基礎に基づいて教えていくとい
う方式を取っていきます。学ぶべき基礎知識は非常に膨大ですが、その見返りは十分にあります。

スクリプトを学びやすくするために、いくつか実世界の例を用いて難しい概念を説明していきま
す。ある問題に対し異なった視点から見つめることで、日常の概念と関連付けられればと考えてい
ます。

マニュアルはいくつかのセクションに分かれています。

第1部 プログラミング概論（第１章～第５章）

コマンドや変数、コントロールの構造、プログラムの流れ、関数といったプログラミングの概念
について紹介していきます。

第2部 例（第6章～第8章）

サンプルを使用してモデラーでのスクリプトの作成法、またモデラーとLScript相互の動作法に
ついて学んでいきます。

第3部 LScript とレイアウト（第9章～第19章）

実例を挙げながらLScriptとレイアウトについての関係を解説していきます。

第4部 LScript プログラミング環境（第20章～第24章）

LScript Editor、Interface Designerなどを含むLScriptのプログラミング環境についてカバーし
ていきます。

第１章：スクリプト概論 1.3



全てを終えると何が起きる？
全て終了する頃には、夢中になってスクリプトを書くようになっているでしょう！ですが、この

マニュアルを読み終えた時点でプログラミング教育が終了したわけではありません。このマニュア
ルは実際の教育である実践と経験についての入門書としてのみ提供されるものなのです。机に向か
い様々なスクリプトを書くことで、独自のプログラミング形態を発見し、本当に必要とされる便利
なスクリプトを作成出来るようになるでしょう。また幾千通りも誤った方法を取ってしまうことも
あるでしょうが、この分野にはつきものなのです。

幸いなことに、プログラミングのための様々な方法をカバーした何千もの本やオンライン記事が
あります。LScriptの動作法において全てが適用可能であるとは限りませんが、一般的なプログラミ
ング、また 3Dに特化したプログラミングに適用できるテクニックを教えてくれます。また討議グ
ループやLScript愛好者達を探してみてください。彼らはプログラミングに関する情報を持ち、初
心者に喜んで手を貸してくれるでしょう。

第２章：変数 2.1

第2章 : 変数

スクリプト実行時において、後々使用するために情報を保存しておく必要が出てくることでしょ
う。記憶しておく必要がある情報が、押されていたボタンの状態なのか、それともオブジェクトの
位置なのか、いずれにせよ後でアクセスしたいデータを持っているとします。このような場合、変
数を使えばスクリプト実行中に、このデータを一時的に保存することが出来ます。変数を使用せず
に多目的のスクリプトを作ることはまず不可能です。そのため、変数がどのように動作するのかを
理解することは、プログラミングの不可欠な要素となります。

ここに幾つか、スクリプト内にある変数の例を挙げてみましょう。

myVariable
control01
frameNumber
frame_step

これらの変数には何も入っていません。変数に対して何の値も割り当てていないからです。
LScriptにより確保された保存領域を示しているに過ぎません。値を保存したい場合、この変数の名
称をリファレンスとして、情報を記憶領域内に挿入します。

その言葉からも連想できるとおり、変数内の値は動的なものであり、スクリプト実行中に修正す
ることが可能です。変数内にどの値を保存するのか、いつ値を変更するのかを完全にコントロール
します。しかしエラーが発生した場合、責任はあなたにあります。値が変わる場合、値を修正する
コードを書いたのはあなたなのですから。変数内の値が独力で変更することはありえません。

トラッキングとデバッグ
スクリプトを書いていくうちに、変数内の値が想定していた値とは全く異なる値になってしまう

場合もあるかもしれません。これは変数の責任ではなく、コードが間違っています。コードのある
どこかの部分で、誤って値を変更しているのです。でも心配しないで下さい。よくある間違いなの
です。こういうエラーを見つけ出し修正する方法を把握していることが、変数を扱う際に常に正し
さを保つ鍵となります。値がおかしくなっている箇所を追跡するのは必要悪でもあり、プログラミ
ング初心者にとっては予想される事態なのです。

バグを見つけ出し修正するのに長い時間がかかる場合もあります。この処理をデバッグといいま
す。最初は時間もかかり非常にいらだたしい思いもするでしょう。でも時間がたてば、より経験を
積んできて、間違いなく過ちが少なくなっていきます。



変数名
変数の名前付けにはいくつか規則があります。

1 名称は唯一無二のものにして下さい。コマンドや文、関数、他の変数と同じ名称を持たせること
は出来ません。スクリプト実行時に名称で混乱を生じさせないためです。

2 名称にはスペースやシンボル（アンダースコア（ _ ）を除く）を含んではいけません。

3 名称内部に数字を挟むことは出来ますが、冒頭に数字をいれることは出来ません。

例：

・変数名 cricket は有効

・変数名 cricket3 は有効

・変数名 3cricket は無効

大文字/小文字の区別
LScriptの要素は全て大文字/小文字を区別しますが、変数も同様です。つまり大文字と小文字の

違いで異なる二つの変数名称をつけることが出来るのです。コードを書く際、このことをしっかり
と頭に入れておいてください。大文字/小文字の区別による単純な間違いが重大なバグになりえる
のです。以下の変数名称は全て異なる変数となります。

cricket

Cricket

CriCKet

Cricket

変数の宣言
宣言は変数に対し単に名称と値を設定します。通常、変数の宣言は名称の後に等記号（=）、そ

の後に特定の値を書きこみます。以下の段落では変数の種類を示し、その使い方について説明して
いきます。それではまず非常に簡単な値、整数から始めていきましょう。

注意
変数タイプには高等な機能や使用法を持つものもありますが、ここでは記しません。最初から混乱さ
せたくないからです。残りの変数タイプについてはサンプルコードで使用する場合に、たっぷりと説
明します。

整数
整数はスクリプトで使用する最も一般的なデータタイプの一つです。このデータタイプは正、負

どちらも 4,294,967,294未満の値を表現します。中学校の数学を思いおこしてもらうと、整数とは小
数点無しの値をさします。

有効な整数値：0, 1, 1481, -253, 823523

無効な整数値：1.0, 3532.99, -25.38, 21.001, 4294967295

例 宣言：

myVariable = 0;

myVariable = 1481;

注意
セミコロンは行の終端を表しており、プログラム内ではこのようにコードを表示します。例を示す必
要もないのですが、実際にコードがどのような外観になるのかを見慣れておいて欲しいのです。

整数は非三次元数学的処理の大半で使用されるデータタイプです。整数はプログラミングにおい
て広くは使用されないというわけではなく、非少数値全体が三次元数学的処理において無効になる
部分が多いのです。整数値だけでアニメーターがアイテムを移動させたり回転させたりスケールさ
せることは非常にまれだからです。非少数値は数値変数タイプへと変換します。

数値
数値データはLScriptで広く使用されるデータタイプの一つです。整数データタイプがカバーし

きれない小数値を、この数値タイプがカバーしているからです。値の小数部分は小数点もしくは指
数表記を使用して表されます。

有効な数値: 2.45, -1.5, 22.1, 9e-10

無効な数値: (none)

例 宣言：

myVariable = 2.45;

myVariable = -1.5;

文字列
文字列は変数内に保存される言葉またはフレーズです。文字列の決まりごととして、変数用の値

は二重引用符（"）で囲んでおかなければならず、また、全て一行に収まるようにしなければなり
ません。LScriptは、引用符の外側の値は他の変数もしくはコマンド/関数であるとみなします。

有効な文字列値："apples", "oranges", "Hello!", "1", "true"

無効な文字列値：apples, oranges, hello!, 1, true

例 宣言：

myVariable = "apples";

myVariable = "oranges";

ベクトル
ベクトルはカンマで区切った三つの数値データタイプの値を、小なり記号（<）と大なり記号

（>）で囲み、一つの型として表します。このタイプは座標値やRGB値を保存する場合に便利です。

注意
ベクトルはオブジェクトメッセージもサポートしており、この章の後の方で解説しています。メッ
セージのリスト全体については、リファレンスマニュアル（12章 : 変数）を参照してください。

有効なベクトル値：<1.9, -2.4, 0.23>, <124.1, -2.35, 1>

無効なベクトル値：<"one", 5.0, 102.4>

例 宣言：

myVariable = <1.8, -2.5, 0.23>;

myVariable = <124.1, -2.35, 1>;

ブール値（ブーリアン）
最後のデータタイプはブール値です。ブール値はバイナリと整数値双方を表します。ブール値と

して保存可能な値は0、1、trueそれにfalseの4個だけです。

有効なブール値：0, 1, true, それにfalse

無効なブール値：4, -2

例 宣言：

myVariable = true;

myVariable = false;

配列
配列は、解説する変数タイプのうち最も複雑なタイプかもしれません。配列は一つの変数名称の

もとに多くのデータの断片を寄せ集めたものです。配列はどのタイプの変数でも持つことが出来ま
す。実際、LScriptにおける配列はタイプが異なる複数の変数をあわせることさえ可能です。

例えば、一つの変数に信号機の色を保存してみましょう。

最初の色は"red"

次の色は"yellow"

3番目の色は"green"

これを書くもう一つの方法が

1 "red"

2 "yellow"

3 "green"

色を表す文字列の前に、それがリストのどの位置にあるのかを表す数値を付け加えてみました。
"2番目の色は？"と問われたら、自然と"yellow"と答えるでしょう。これが配列の動作なのです。

文字列名称を用いて色を書いてきましたが、配列にはどのタイプを用いることも出来ますから、
このように簡単に表すことも出来ます。

<255, 0, 0>

<255, 255, 0>

<0, 100, 0 >

標準の変数名に括弧（[]）と、インデックスと呼ばれる連続する整数値を用いることで、配列
に値を保存したり取り出すことが出来るようになります。インデックスは正の整数値、もしくは正
の整数値を含んだ変数でなくてはなりません。ゼロ、負の値、小数などは使用できません。信号機
の例では変数名称は以下のようになります。

trafficColor[1]

trafficColor[2]

trafficColor[3]

有効な配列インデックス:

trafficColor[1]

studentNumber[114]

trafficColor[lightNumber]

無効な配列インデックス:

trafficColor[0]

accountNumber[-4]

trafficColor[1.5]

pointNumber[<9,1.0, 12>]

例 宣言：

trafficColor[1] = "red";

trafficColor[2] = "yellow";

trafficColor[3] = "green";

配列のある部分をより細かく見てみると、インデックスとして変数を使用していますね。これを
利用して少し複雑に設定してみましょう。

trafficColor[1] = "red";

trafficColor[2] = "yellow";

trafficColor[3] = "green";

lightNumber = 1;

trafficColor[lightNumber]には文字列値"red"が入っています。

少し説明を加えましょう。lightNumberは整数値 1を持つ変数であることを宣言しています
から、配列へのアクセス時にLScriptはこの値を読み込みます。つまりこの行は下記のように読み
込まれていることになります。

trafficColor[1];

宣言リストからtrafficColor[1]の値を探してみると、文字列"red"が入っているのがわ
かりますね。

今はまだ少し混乱してしまうかもしれませんが、このメソッドはプログラミングにおいて一般的
な実践法であり、これについてはさらに解説する例が他にももっとたくさんあります。今の時点で
は基本コンセプトを理解しておくようにしましょう。

文字演算
変数を扱う一方で、少し文字演算についても勉強してみましょう。この宣言グループは何が行わ

れると思いますか？

text1 = "Hello";

text2 = " World!";

text3 = text1 + text2;

text3の結果は"Hello World!"となります。text1は文字列 "Hello"、text2は
文字列 " World!"と宣言しているため、text3はこの値 "Hello"+" World!"をともに取得し、
"Hello World!"となるのです。これが文字演算です。

変数演算
文字演算と同様、変数演算も数学を使って他の変数を新たに作成したり、既存の値を修正したり

します。ここにコードの一部を紹介します。

val1 = 1000;

val2 = 10;

val3 = 5;

val4 = val1/val2 * val3;

val4の結果は500となります。1000を 10で割り導き出された結果 100に対し 5を掛け合わせ
ます。

val[1] = 1;

val[2] = 10;

val[3] = val[1] + val[2];

val[3]の結果は11となります。変数演算はプログラミングにおいて頻繁に使用されます。幸
運なことにたくさんの実習例があります。

定数
忘れることの出来ない重要な変数タイプがもう一つあります。定数は他のタイプとはかなり異な

りますので、この章の最後で解説することにしました。定数はかなり違いがあるものの、非常に簡
単なタイプです。

定数は名前どおり一定の値として処理されます。値を変更することは出来ません。定数は
LScriptが内部で使用する変数を定義していますので、変更するのは不可能なのです。例えばライト
種を見てください。

ライト種：

平行ライトはライトタイプ 0

点ライトはライトタイプ 1

スポットライトはライトタイプ 2

これらの値を事前定義することで覚えやすく、また使いやすくしています。

LWLIGHT_DISTANT = 0

LWLIGHT_POINT = 1

LWLIGHT_SPOT = 2

これでコード内に変数LWLIGHT_DISTANTを書き込めるようになります。平行ライトはタイ
プの値0をもっていると覚えるよりもはるかに簡単ですね！

定数は覚えやすく読み込むのが簡単ですので、LightWaveはこれらの値を保護しています。前述
したとおり、値は変更することは出来ません。

注意
定数は他の変数とは著しく異なっており、また厳しい制限を持っているため、大文字で宣言されてい
ます。定数値を大文字化することでコード内部において即座に認識出来やすいようにしています。

他にも定数の利点として、LScriptにおける内部変更に対しコードを保護する役目もあります。
仮にNewTekのプログラマが、コードの修正が必要な点を見つけ内部変数をいくつか修正した場合
でも、LIGHT_DISTANTはスクリプトに対し常に同じものを指すようにしています。

2.8 LScript 日本語ユーザーガイド



第３章：関数 3.1

第3章 : 関数

関数とはごく小規模でありながら、反復的なプログラミングのタスク（処理）を扱う自己内蔵型
のコードセクションです。関数はメインプログラムとは別に設定し、独自の変数やメインスクリプ
トとは独立したコードを含んでいます。プログラムは関数で使用する特定のデータを関数側へと送
信することで関数を使用することが出来ます。この処理を"関数呼び出し"といいます。

一度関数を呼び出すと、そのデータとコード行が通常のスクリプトのように処理されます。関数
は呼び出し側のコードからは完全に切り離されます。特別にコードに送信されたデータだけがメイ
ンスクリプトから使用可能です。関数には"フォーカス"があります。

関数内のコードが実行終了すると、処理した値を、関数を呼び出したスクリプトへと送信するこ
とが可能です。処理した値が関数内に残されたままであれば、これらの値は何の意味もなくなって
しまいます。その後フォーカスは関数を呼び出したメインスクリプトの場所へと戻り、中断されて
いた箇所から処理を続行します。

LScript内部にある関数だけに限りません。必要に応じてカスタマイズされた独自のユーザー定
義関数（UDF）を簡単に作成することが出来るのです。実際、使用頻度の高い関数を幾つか一緒に
集めてグループ化し、たくさんのスクリプトから呼び出せる共有可能なツールライブラリを作成す
ることも出来ます。スクリプトを組んでいるとすぐに、非常に強力な関数群をいくつも持てるのだ
とわかってくるでしょう。

現実世界からの例
以下、実世界からの例を使用して関数がどのように動作するのかをよりわかりやすく説明してい

きます。

スクリプトを大きな自動車製造工場だと考えてみてください。工場は月に何千台もの車を生産し
なければならないのですが、同じ工場で車の全部品を作ることはしません。例えば、タイヤは別の
国で作られ、ガラスはまた別の国、点火プラグはまた別のという形になります。これらの部品全て
が集まったときに、車を作ることが出来るのです。



この考えをもっと細かくみてみましょう。車製造工場は車を組み立てはじめると、タイヤが必要
であることがわかってきます。タイヤ会社を呼び出し、価格 10,000ドルで 1000個のタイヤを注文し
ます。タイヤ会社はお金を受け取って、タイヤを製造し、それからタイヤを車工場へ納品します。
タイヤを製造するということが、車組み立てという大きな処理の中の一つの関数です。この考えを
プログラムと関数に置き換えてみましょう。

プログラム：“車の製造”

フレームを製作

エンジンを製作

タイヤ会社から 10,000ドルで 1000個のタイヤを買う

...他のアクション...

車が完成

関数: "タイヤを買う"

10,000ドルで 1000個のタイヤを買う

タイヤを製造するための原料を買う

タイヤを集める

1000個のタイヤを工場へ送る

関数はLScript内にある関数とは違ったように見えます（上記は"擬似コード"と呼ばれます）が、
必要なコンポーネントは持っています。より詳細に見ていきましょう。

まずメインのプログラムでは、"車の製造" はフレームを製作し始めてから、エンジンを製作し、
その後で 10,000ドルで 1000個のタイヤを買う必要が出てきました。この時点でプログラムは "タイ
ヤを買う" と言う名前の関数があることを理解しており、これを呼び出します。タイヤ会社にとって
タイヤを納品するためには、何個のタイヤを送るのか、いくら支払いを受け取るのかを把握してお
く必要があります。

関数のこの部分を引数といいます。引数は、仕事（この場合はタイヤの製造）を完了するために
関数が必要とするデータです。タイヤ会社は適切な情報を受け取り、タイヤを製造してから車工場
へとタイヤを送ります。メインプログラム"車の製造"では、"タイヤを買う" 関数を呼び出し、情報
（1000個のタイヤと 10,000ドル）を送ります。このデータ送信の処理を、関数へ引数を渡すといい
ます。

関数の最後の部分は戻り値といいます。一旦関数の処理が終われば、メインプログラムへと返す
ことを同意したものが戻り値となり、単にそこに置くだけでよいのです。この場合、戻り値はタイ
ヤです。

プログラムと関数の作成
今回の例は実際のLScript関数ではありそうもない話なのですが、どのように関数が動くのかを

はっきりと示しています。実際のLScript用語を使わずに、この例をLScript用にフォーマットして
みましょう。

メインプログラム：車の製造

main

{

buildFrame();

buildEngine();

tires = getTires(1000,10000);

...車の完成...

}

そして関数部分：

getTires: numberOfTires, payment

{

buyMaterials();

assembleTires();

...車製造のコード...

return(tires);

}

関数のパーツ
上記の例は正確ではありませんが、その前の例とは異なる点がたくさんあります。この相違点に

ついて見ていきましょう。

最初の違いは'main'という語句にあります。最初のスクリプトはモデラー用のLScriptでしたの
で、モデラースクリプト独自のルールに焦点を当てています。言葉が示す通り、main()関数は
LScriptがコードを実行を開始し始める箇所です。どのモデラーLScriptもmain()を持たねばなり
ません。

ボディマーカー

次に、この波括弧は一体何でしょうか？これはボディマーカーというもので、LScriptにコード
の始端（{）と終端（}）を表しています。スクリプト内にある他のコード全てを除去すると、
main()関数はこのようになります。

main

{

}

このボディマーカー内部にあるコード行、またはボディマーカーでくくられている行はmain()
関数に属しています。関数名称の後にくる開括弧（{）は、閉括弧（}）に到達するまでの存在する
全てのコードを所有していると関数に対して示しているからです。関数用の閉括弧が偶然にも残っ
てしまった場合、LScriptはmain()関数がどこで終わるのかがわからなくなってしまい、嫌なエ
ラーを引き起こしてしまいます。

注意
二つのマーカーの間にあるコードもコードブロックとして参照されます。

セミコロン

この例ではコードはmain()関数に属しています。コードの各行の終端にセミコロン（;）がつ
いているのに注目してください。このセミコロンは各文の終端を示しています。文の終端にあるピ
リオドと同じような動作をするのです。

テキストエディタではENTERキーを押さないと新規行には移れませんよね。LScriptはセミコロ
ン（;）に到達するまでコード行の終了だとはみなされないのです。不適当な句読点は、エラーの
原因となります。例えば次のコードを見てください。

main
{
buildFrame();
buildEngine();
tires = getTires(1000,10000);

}

これは以下と同義です：

main
{
buildFrame();
buildEngine();
tires
=
getTires
(1000,10000);

}

大変不恰好でこのように書くべきではないのですが、コードとしては正当です。スクリプト開発
者はコードの終端にセミコロン（;）を使用する限り、どのように整列させフォーマットしようと
も構わないのです。最後に、以下のコードもまた同義であり正当なコードです。

main

{

buildFrame(); buildEngine(); tires = getTires(1000,10000);

}

注意
最もよくありがちな誤りは、セミコロンを忘れてしまうことです。ここはイージーミスとなる箇所です。

例へと戻りましょう・

tires = getTires(1000,10000);

さて、等記号の後のデータのみを見ていくことにしましょう。

getTires(1000,10000);

LScriptはコマンドや文のデータベースを通してgetTires()という語句を内部的に探しにい
きます。getTires()という名称のコマンドや命令文が見つからなかった場合は、ユーザー定義
関数（UDF）であると仮定し、スクリプト全体を検索して関数を探しに行きます。getTires()
関数は宣言していますので、LScriptは遠くまで探しに行く必要はありません。しかし関数を見つけ
られなかった場合にはエラーが発生します。

二個のデータ、1000と10000（コンマで区切られている）がgetTires()関数へと渡されます。
これは丸括弧内に含まれる二つの値を示します。LScriptでは関数内で定義したのと同数の引数を渡
さなくてはなりません。getTires()関数の中では、numberOfTiresとpaymentという
二つの引数を宣言しました。

思い出してください：

getTires: numberOfTires, payment

{

...

}

この宣言文において、コロン（:）の後に続くものが、関数で必要となる引数です。
getTires()関数本体の中のどこかで、引数numberOfTiresとpaymentを使用して、こ
の処理を実行しています。

getTires()関数の最終部分はリターン文です。

…

return(tires);

…

リターン文はgetTires()関数に、main()関数にデータを送り戻すように指示します。こ
の場合、タイヤの個数を返します。戻り値は括弧の中に記述します。後で戻り値の取得法について
学んでいきましょう（ヒント：等記号の左側に変数があるということを覚えておいてください）。
これで戻されたデータがmain()関数で使用出来るようになります。

関数への引数渡し
それではモデラー上で実際に動作する例を見ていきます。このサンプルはコードの断片でたいし

た処理は行いませんが、関数呼び出しの概念をよりよく理解できるように解説しています。

main

{

a = 10;

b = 15;

c = addAB(a, b);

info(c);

}

addAB: a, b

{

tmpVal = a + b;

return(tmpVal);

}

ここには二つの関数、main()とaddAB()があります。まず最初に、main()関数はそれぞ
れ整数値10と15を持つ変数aとbを宣言します。この変数aとbを渡すことで、フォーカスは
main()関数からaddAB()関数へと移動します。

addAB()では、変数tmpValを宣言しています。ここには a+bの値、すなわち 10+15の値を
入れます。このため変数tmpValの値は25となります。

次の行で

return(tmpVal);

関数に対し変数の値（25）をmain()関数へと送り返すように指示します。メイン関数では変
数cでこの値を受け取ります。これを検証するため、info()コマンドで画面上に変数cの値を表
示するようにします。勿論25と表示されるはずです。

新しい関数へ値を送り、その関数で足し算を行い、main()関数へと値を戻しました。これが
全ての関数の基本的な手続きなのです。

第４章：条件文 4.1

第4章 : 条件文

理想の世界では、スクリプトは常にシーン内またはモデルに対し何が行われているのかを正確に
把握していることでしょう。ユーザーが加える修正や調整も認識し、偶然出くわしたタスクの処理
法についても既に把握しているかもしれませんね。一つのシーンやオブジェクトに対してだけ実行
するような一度限りのスクリプトを作るのであれば、状況を全て把握した上でプログラムを組むこ
とが可能です。しかし勿論このようなやり方では、多様なシーンやオブジェクトに対して動作する
洗練されたスクリプトは期待出来ません。

オブジェクトやシーンにいくつかの条件が存在している場合、その条件を比較し、状況によって
スクリプトに異なる動作をさせることがあります。この判定を条件文といいます。

例えば、モデラースクリプトを使用して二つの選択したポイントを統合するとしましょう。統合
する前に、ユーザーがポイントを二つ選択しているかどうか確認したいですよね。もしポイントが
選択されていれば、スクリプトは二つのポイントを統合できます。しかし選択されていない場合に
は、まず統合する二つのポイントを選択するよう、ユーザーに通知する必要があります。

このような状況はプログラミングの最中には頻繁に起こるもので、様々な文が条件に基づいて処
理されます。条件文の役割は他の文を検証することにあります。文が正しければ（true）、
LScriptはコードのある部分を実行し、正しくなければ（false）無視します。これらの条件文で
最も一般的なものとして、if-then文があります。



If-then 条件文
if-then条件文は式が正しいかどうかをチェックします。式が正しければ（true）スクリプ

トは条件文のすぐ後にあるボディマーカーに囲まれたコードブロックを実行します。式が正しくな
い（false）場合にはコードブロックは無視され、閉括弧（}）の後に続く次のコード行を実行し
ます。

構文：

if(…式…)

{

… 式が trueの場合に実行されるコード…

}

例：

if(pointsSelected == 2)

{

mergePoints();

}

…次の文…

この例を一通り眺めてください。if-then条件文はpointsSelected変数の値が 2である
かどうかを評価しています。2の場合には mergePoints()コマンドが実行されます。
pointsSelectedの値が 2ではない場合、mergePoitns()コマンドは飛ばされ、次の行が
実行されます。

この例では、コードブロックは実行用のコードを 1行（mergePoints文）だけ含んでいます。
このような場合、if-then条件文は実際には以下のコードで示す通り、ボディマーカー無しで書
くことが出来ます。

if(pointsSelected == 2)

mergePoints();

…次の文…

式が正しい場合に単一行のコードだけが実行されるのであれば、このようにコードを書くことが
可能なのです。

演算子
if-then条件文の中にpointsSelected==2という式があるのがわかりますね。これは誤植ではなく、

この式の中で二つの等記号は演算子となっています。LScript内で使用可能な様々な演算子について
見ていきましょう。

以下に基本的な演算子を記します。

演算子 言葉で表すと

=i 値 iの代入

! 非

== 等しい

> 大なり

< 小なり

>= 大なり もしくは 等しい

<= 小なり もしくは 等しい

!= 等しくない

前述のif-then条件文を使用して、各演算子の適切な使用法を解説していきます。

a = b aに値 bを代入

if(a == b)aが bと等しい場合

if(a != b)aが bと等しくない場合

if(a > b) aが bより大きい場合

if(a >= b)aが bより大きい、または等しい場合

if(a < b) aが bより小さい場合

if(a <= b)aが bより小さい、または等しい場合

If-then-else 文
if-then-else文は、先ほどのif-then文とほとんど同じですが、一つだけ大きな違いが

あります。式が正しくない（false）場合にコードを完全に飛ばすのではなく、elseブロック
へと移り、そこにあるコードを実行するのです。

構文：

if(…式…)
{
…式が trueの場合に実行されるコード…

}
else
{
…式が falseの場合に実行されるコード…

}
… 次の文…

else文はtrue文の閉括弧（}）の後に直接続いており、その後にはセミコロンはありません。
これはelse文もif-then文の一部だからです。

例：

if(pointsSelected == 2)
{
mergePoints();

}
else
{
error(“2 or more points must be selected.”);

}

前述の文は以下のように書くことも可能です。

if(pointsSelected == 2)
mergePoints();

else
error(“2 or more points must be selected.”);

実際にはこちらの記述法が前述の形式よりも正しいのです。条件のボディマーカーは文が単一行
のコードを使用している場合には不必要となります。見てわかるとおり、このフォーマットでは長
いスクリプトよりも空間を節約できるのです。

error()文は単に画面上にエラーメッセージを表示し、スクリプトの実行を中断します。

ブール値と if-then-else
条件式の中でif-then-else文とブール値の変数を組み合わせると、面白い状況が生まれま

す。変数のブール値が trueの場合には条件式が trueであるとみなされます。同様に変数が
falseの場合には、式がfalseであるとみなされます。

例 1：

objSaved = false;

if(objSaved)

info(“Object was saved.”);

else

saveObject();

… 次の文…

例 2：

done = false;

if(!done)

info(“The procedure IS NOT done.”);

else

info(“The procedure IS done.”);

…次の文…

Switch 条件文
switch()条件文は非常に強力な機能です。if-then条件と同様、式を解くのですが、その

結果をリストされている値と比較します。値がリストアイテムと一致したら、case文の後ろに直
接書かれているコードを実行します。breakコマンドはswitch()文を全てそのまま残します。

構文：

switch(…式…)

{

case /値 1/:

...文...

break;

case /値 2/:

...文...

break;

default:

...文...

break;

}

… 次の文…

例：

valueA = 3;

valueB = 1;

switch(valueA - valueB)

{

case 1:

info(“A - B = 1”);

break;

case 2:

info(“A - B = 2”);

break;

default:

info(“No Matches found!”);

break;

}

…次の文…

上記例では、変数 valueAと valueBには整数値が与えられます。式の結果 valueA-
valueB（この値は 2）がswitch()コマンドへと渡されます。switchコマンドはこの値を
caseの値リストと比較します。一致すればcaseコマンドにつけられているコードが実行されま
す。

最初のcase文の値は1です。式の値はこの値と一致していないので、LScriptはコードを飛ば
して次のcase文と比較します。次のcase値は式の値と一致するので、LScriptはinfo()コマ
ンドを実行します。このコマンドはinfo()ボックスの中に "A - B = 2"という文字を表示し
ます。次の文はbreak文です。このbreak文により、LScriptはswitch()文の閉マーカー
（}）を直接実行するように指示されます。

switch()文を使用すれば、数種類の異なる値に対し式を比較する場合などにif-thenや
if-then-else文をいくつも使用することなく実行できます。

第５章：ループ文 5.1

第5章 : ループ文

LScriptの最も簡単なものとして、レイアウトコマンド、またはモデラーコマンドがあります。
これらのコマンドはスクリプトの終端まで一度に一行ずつを順に実行していきます。

しかし x回もコマンド文を繰り返したいと思うでしょうか？何度も何度も正確に文を入力するこ
とは可能かもしれません。でもそれは馬鹿げたことですし、非常に非効率的です。そればかりでは
なく、何回繰り返せばよいのか把握出来るでしょうか？値はユーザー入力値に基づく値かもしれま
せんし、他のコマンドから設定される変数かもしれません。ループ文を使用せずにこれらの状況を
処理するLScriptを作成するのは、まず不可能です。

ループ文は、条件がtrueである限りコードブロックを実行します。コードブロック内部の最
初の行が実行され、閉括弧（}）に遭遇した時点で、開括弧（{）文まで戻りコードを繰り返します。
式が falseになるまで処理を繰り返した後、次の文へと移動します。最も一般的な文は for-
loop文です。

for-loop 文
for-loop文は複雑ではありますが、ループ文の中では最も一般的なものです。この文はループを

繰り返します。つまりループ内のコードブロックが処理された時点で、変数を変更し、式内の値を
比較してもう一度処理を反復するかを決定します。式の値が falseになるまで処理を繰り返します。

構文：

for(変数宣言; …式…; 反復)
{
…ループする文…

}

例：

for(i = 0; i < 10; i++)
{
info(i);

}

LScriptが文を実行している時、4つの事柄が起きています。一つは変数が0で宣言されます。



for(i = 0; i < 10; i++)

次にLScriptは、二番目の文で定義されている式がtrueか否かをチェックします。この場合、
LScriptは変数iが 10よりも小さいかどうかをチェックします（現在iは0であり、0は 10よりも
小さな値ですから、式はtrueとなります）。

for(i = 0; i < 10; i++)

式がtrueの場合、LScriptは次に続くコードブロック内の文を直接実行します。実行するよう
に設定しておいた文は、簡単なinfo()ボックスだけです。

for(i = 0; i < 10; i++)
{
info(i);

}

しかしこのinfo()は表示テキストではありません。このinfo()文は、反復変数である変数
iの現在の値を表示しています。これにより実際に変数iがどのように変化しているのかがわかる
ようになります。

for文はまた終わっていません。ブロック内にある全ての文を実行したら反復変数が修正され
ます。どのように変数が修正されるかは、for文の3番目の部分に定義されています。

for(i = 0; i < 10; i++)

この場合i++が使用されていますが、これはi=i+1を手っ取り早く表しています。この文は
コードがループされるたびに変数iがどのように変更するのかを定義しています。ですからiは0
の値から始まり、コードブロック内の文がすべて実行された後には値が1、2、3と変化していき
ます。つまりiの値が10より小さい間は、info()ボックスで表示し続けるということです。

見てわかるとおり、for文を使用すると、ループが可能なだけではなくカウントも出来ます。
配列内の値を扱う場合にもこれはとても便利です。以下の配列を見てください。

objects[1] = “Null”;
objects[2] = “cow.lwo”;
objects[3] = “ball.lwo”;

このコードは何をしているのか想像してみてください。

for(i = 1; i <= 3; i++)
{
info(objects[i]);

}

for文の反復がfalseになったとき、LScriptは次のコードへと移動します。

while loop 文
for文は既知の量数だけ文を反復したい場合には最適です。しかしスクリプトが広範囲の状況

においても使用出来るように設計されている場合など、ループを中断する時期を必ず把握できると
は限りません。スクリプトから何らかのアクションがあるまでループを実行させておきたいという
ような場合、どうすればループを中断することが出来るのでしょうか？そんな場合には、while
ループを使うとうまくいきます。

注意
このコードは厄介な側面も持っています。while ループを使うプログラミングでは、かなり慎重に、
集中して記述していかないと、恐ろしい事態に見舞われる可能性があります。

構文：

while(…式…)

{

…文…

}

…次の文…

whileループはif-then文と似ているように見えますね。これが基本的なwhile文の処理
です。ただしコードの次の行に移動する代わりに、式がtrueではない限りwhileループの実行
文は反復します。

注意
このやり方を見てみると、while ループは前述の for文にも似ていますね。

例：

done = false;

while(!done)

{

if(x ==5)

done = true;

else

x++;

}

whileループを入力する前にブール値をfalseに宣言しておかなくてはなりません。この変
数はwhileループが反復を続行するか否かをチェックするためのものです。

done = false;

この変数には、while文につけられているコードブロックに対し、ライトスイッチのような役
割だけが与えられています。変数doneがfalseであればループが実行され、trueであれば
ループは中断します。ですから whileのブロックコードを一度でも実行させるためには、変数
doneには初期値としてfalseが設定されていなくてはなりません。

まずwhile()文は、ブール値の変数がtrueではない値（つまりfalse）かをチェックしま
す。そうであれば設定されているコードを実行します。

while(!done)

{

if(x ==5)

...

if-then文は整数変数値xが 5と等しいかどうかをチェックします。この場合、変数xには 0
の値が入っています。ですからif-then文のelseブロックが実行されます。elseブロックは
このwhile()ループ用の反復処理です。変数xは1となり、ループが繰り返されます。

最終的にxは値が 5となり、if-then文はtrueとなります。ブール変数doneにはtrue
が代入されます。

done = true;

while()の式はdoneがfalseであるかをチェックし、ループは終了します。それからプロ
グラムのフォーカスは次に設定されている文へと移ります。

では、どこが危険な箇所なんでしょうか？この例をしっかりと見て、どこが悪いのかを見つけて
みましょう。

done = false;

while(!done)

{

if(x ==5)

done = true;

}

これは無限ループと呼ばれるものです。while()コマンドはブール値doneの値がfalseか
否かをチェックしますが、while()文に対するコードブロックでは反復が行われていません。x
の値は変化することがなく、決して 5の値になることがないため、doneは永遠にtrueにはなら
ないのです。このため、スクリプトも終了することが出来ません。無限ループに陥ったときには手
動でレイアウトまたはモデラーを強制終了し、LightWaveを再起動させなくてはなりません。

注意
for文の中でも、無限ループが作成されることがあります。

5.4 LScript 日本語ユーザーガイド



forまたはwhile()ループで作業する際には、かなり慎重に行ってください。スクリプトが
終了できるように変数が反復処理を行っているか、値を変更しているのかなどを確かめてください。
スクリプトがハングしてLightWaveを強制終了し、これまでの全ての作業を失わなければならない
破目に陥ったときには、ユーザーは本当にうんざりするものです。

第６章：モデラー LScript 概論 6.1

第6章 : モデラーLScript 概論

ここまでは実際のLightWaveのリファレンスや三次元アニメーションまたはLScriptを使わずに
コンピュータプログラミングの基礎を学んできました。さてここからは、学んできた知識をお馴染
みの概念と理論、つまり三次元モデリングやアニメーションに適用していきましょう。

このセクション（第 6章～第 8章）では、前セクションでカバーした理論と実例を使用しながら、
モデラースクリプトの作成方法を見ていきます。つまり、既に以下の概念については基本的な理解
がなされているものと仮定します。

・変数

・条件式

・関数

よくわからなくなってきたら、もう一度時間を取ってこれらの章を読み返すようにして下さい。
上記の概念について、よく理解しておく必要があるからです。この時点でプログラマー上級者に
なっているとは期待していませんが、この研究を続けていきながら前回の資料に戻るようなことは
したくありません。その代わり、これらのプログラミング概念を基に補強を重ね、既に学んだ知識
を広げていきましょう。

思い出してください。LScriptはプログラミング言語であり、他の言語と同様、基本から始めて
いかなければならないのです。名詞や動詞については学びましたが、今からは文や節をどのように
組み立てていくのかを学んでいきましょう。最終的には短い物語が作れるようになり、ついにはベ
ストセラーも夢ではありません。でもピューリッツァー賞を取る前にまず、第 2段階を生き残らね
ばなりません！

このセクションでは、今までとは違ったアプローチで教えていきます。前セクションで行ったよ
うに概論や理論を検討することで学んでゆくのではなく、解剖実習と呼ばれる形式をとりましょう。
解剖実習では、コードを行ごとに解析することでプログラミング法を学び、どのようにスクリプト
の作業に影響を与えているのかを学んでいきます。ここから先のプログラミングはこの方法で解説
していきますので、慣れていってくださいね。



このセクションでの目標は、モデラー用LScriptの基本を授けるということにあります。モデラー
用とレイアウト用のプログラミング言語は似てはいますが、構造とプログラムの流れが根本的に異
なります。レイアウト実行には必要となるプログラミングオーバーヘッドの大半が、モデラーのス
クリプトにはありません。ただし、モデラースクリプトで技術的に必要とすることが少ないからと
いって、強力さに欠けるということではありません。

モデラーをLightWaveという複雑なマシンの中の巨大な回転歯車だと考えてみてください。実際
スクリプトは一個の歯車を持つ小さなマシンで巨大なモデラー歯車にかみ合わせようとしていま
す。ご想像どおり、既に回転しているギアに対して小さなギアをかみ合わせようとするのは容易な
仕事ではありません。しかし全てが正しく行われれば二つのギアは一致してともに回転し、モデ
ラーのギアはより小さなLScriptマシンが実行するための力を与えてくれるようになります。

レイアウトで実行するスクリプトは、それよりちょっと複雑です。モデラーのように一個のギア
をかみ合わせるのではなく、スクリプトのギアは半ダースものレイアウトのギアに組み合わせなく
てはなりません。注意が必要ですが、このマニュアルのレイアウトの章を読めばわかるとおり、絶
対に大丈夫です。レイアウトと格闘する前に、まずモデラーの LScriptからはじめるのが良いで
しょう。

LScriptプログラミング言語はモデラー内にあるコマンドや関数に完全にアクセスできるように
してあります。LightWaveに付属している利用可能な多くの強力なスクリプトからも、この可能性
をおぼろげながらも感じ取ることが出来るでしょう。モデラーで出来ることはLScriptでも実現可
能なのです。

つまりモデラーのセクションとの対話は他よりもはるかに簡単なのです。このマニュアルの目標
である「スクリプトを組み始めるのに十分な知識を授ける」ために、モデラー関数のほんの少しの
サブセットをカバーしていきます。それらサブセットをどのように使用すればよいのかを示し、残
りの部分を独力で学べるような知識を与えることが目標なのです。

では始めましょう！

第７章：モデラースクリプトの基礎 7.1

第7章 : モデラースクリプトの基礎

では手始めに、モデラー用の簡単なLScriptを作ってみましょう。このコードの断片で、最初の
スクリプト作成処理とモデラーへどのようにしてスクリプトを入れ込むのかなどについて解説して
いきます。それ以上のことは行いません（どこかで必ず始めなければなりませんけどね）。

この処理ではスクリプト記述に使用可能な 2～ 3のツールについて注釈を入れていきます。これ
らのツールの一つが LScript Editorと呼ばれるNewTekから提供されているカスタムメイドのエ
ディターです。LightWaveのProgramsディレクトリの中にLSED.exeというエディタープログラム
があります。LScript Editorはスクリプトの記述とデバッグを行うプログラミング環境です。このマ
ニュアルの章全体がLScriptプログラミング環境を使用することに捧げられていますので、ここで
紹介されたスクリプトを自由に拡張していってください。

Hello World
今までにプログラミング言語を習得しようとした経験があるなら、最初に記述するプログラムの

一つに"Hello World!"というものがあるのをご存知でしょう。これはスクリプトの使い方、また
この場合モデラーでの実行の仕方を解説することだけを目的としたプログラムです。

では始めましょう。

1 LScript Editorを開きます。

2 File > Newを選択して新規スクリプトを作成します。インターフェイスの大部分で新規LScript
ファイルを表示します。ここにコードを打ち込みます。

3 以下のコードをワークスペースに入力してください。

main
{
info(“Hello World”);

}

4 File > Save As...を選択してスクリプトを"hello.ls"という名称で保存します。忘れずにファイル
を保存してください！綺麗にまとめるためスクリプト用の新規フォルダを作成するようにして下
さい。そうすれば手早く簡単にスクリプトを保存したり探し出せるようになります。

注意
初期段階のスクリプトでは、全てLScript のファイルであることを示す拡張子(.ls)がつきます。



5 モデラーでConstruct（構造）> LScriptを選択し、"hello.ls"スクリプトを検索して下さい。スク
リプトを選択するとモデラーはスクリプトを実行します。コードが全て正しく入力されていれば、
infoボックスに"Hello World!"という文字が表示されているはずですね。エラーメッセージ
が出てきたならエディターに戻り、コードをチェックしてみてください。

注意
infoや error メッセージの表示箇所は警告レベル設定によります。初心者レベルであればメッセージ
は画面中央にパネル形式で表示されます。上級者レベルであればメッセージはモデラーのツールチッ
プセクションに表示されます。

これで終了です！今やあなたはプログラマなのです！

では、今作成したスクリプトをもっと詳しく見ていくことにしましょう。

main()関数
モデラースクリプトには全てこのmain()関数が存在します。名前が示す通り、この関数がス

クリプトのメインの関数となります。モデラーがスクリプトを読み込むと、まずLScriptは即座に
main()関数を検索しに行きます。それからボディマーカー内にあるコードを全て実行します。実
際に以下のスクリプトを動かして見ましょう。

main

{

}

このスクリプトは何の動作も起こしませんが、このわずかなコードだけがモデラーでのスクリプ
ト実行に必要となるのです。

関数の章で学んだとおり、関数で変数を使用出来るように引数として渡すことが出来ます。です
がmain()関数はこの規則には当てはまりません。引数として変数を渡すことは出来ないのです。
引数を渡そうとするとエラーメッセージが出てきます。

ボディマーカー内部には1行だけコードが記述されています。

info(“Hello World!”);

見てわかるとおり、info()関数を使用してスクリーン上に様々なメッセージを送信します。
通常は、ユーザーに対して必要な処理や警告のメッセージなどを伝えます。この関数を使い、スク
リーンに送りたいテキストや値を括弧内に挿入することで、メッセージを表示します。文字メッ
セージや文字メッセージを含んだ変数、もしくは直接値を使用することも可能です。実際に何が起
きているのかをよりよく理解するために、ここではプログラミングメッセージを使用しましょう。

"Hello World!"文字列がinfo()関数へ渡されました。

しかし、この行に以下のように入力してみると

info(Hello World!);

LScriptから以下のようなエラーメッセージが表示されてしまうでしょう。

“In main, line 3, found ‘W’, expecting “)” or “~

(Line: info(Hello World!);)”

このメッセージはかなり説明的な記述になっていますね。このメッセージでは 3行目の文字"W"
の個所でエラーが発生し、ここには文字")"または"~"が来ると想定されていましたと伝えています。
基本的に二重引用符がない場合、info()関数は変数が送信されたのだと想定します。これは全く
正しいのですが、変数の間にスペースが入っていると正しくありません。そうなんです、この習性
をすっかり見落としていました。変数を送っているつもりなど全くなかったのです。しかし簡単な
メッセージでもどれほどLScriptを迷わせてしまうかがわかりましたね。このようなエラーはこれ
からも起きるでしょうし、先に進むことにしましょう。

この変数の考えを続けるため、以下のように入力してください。

info(HelloWorld);

スクリプトは、今度はエラーメッセージ無しに実行します。しかし実行しても変数 Hello
Worldには何も入っていないということを示す"(nil)"という値が表示されます。

ではどのようにしてHelloWorld変数を取得すればよいのでしょうか？次のコードを試して
みましょう。

main

{

HelloWorld = “Hello World!”;

info(HelloWorld);

}

3行目でHelloWorld変数が文字列値"Hello World!"を持つ変数として定義しています。
従ってこの変数info()関数へ渡したときには、info()はこの値"Hello World!"を表示し
ます。

文字列を定義するためには引用符("")を忘れずにつけるようにして下さい。この括弧をつけない
と、LScriptは変数だとみなしてしまいます。

もう少し見栄えの良いプログラムにしてみましょう。

main

{

text = “Hello World!”;

info(text);

}

変数名をより一般的な名称にすることで、より広い範囲で値を記述することが可能です。今おこ
なった修正は、コードをよりよく見せるためのものです。変数名には何をつけても構いません。以
下のような名称をつけることも可能です。

tfaGxCw = “Hello World!”;

名称は事実上有効ではありますが、ちっとも説明的ではありませんね。他の人があなたのコード
を見なくてはならなくなった場合、もっとまずいことにあなた自身が何年か後にこのコードを見る
必要が出てきた場合のことを想像してみて下さい。変数が何の動作をしているのかが即座に理解で
きるように、より説明的で読みやすい変数名が良いでしょう。

ここではスクリプトをより読みやすくするための話をしていますので、コメントについても解説
していきましょう。

コメント
コメントはスクリプト内に注釈を差し込む方法です。注釈はスクリプトで何をしようとしている

のかを、読む側に対し気づかせてくれるものです。ですからコメントにより情報を入力することで、
読む側はより高くあなたの努力を評価することでしょう。コメントはLScriptからは完全に無視さ
れます。コメントはモデラーがスクリプトを実行させる前に内部的に全て除去されるのです。構文
という点から言えば、コメントには何を差し挟んでも間違いになることはないのです。

新しいHello.lsスクリプトにコメント文を挿入してみましょう。

main

{

// ここはコメントです

text = “Hello World!”;

info(text);

}

注意
このマニュアルでは、分かりやすく説明するためにコメントに日本語を使用していますが、実際の
LScript では英数字のみしか使用できません。

ダブルスラッシュ文字（//）を打つことでコメントが作成されます。この文字（//）の後に続
くテキストがコメントであるとみなされ、LScriptからは無視されます。

次のコメント設定は悪い例です。

main

{

コメント //

}

何行にもわたってコメントしたい場合にはどうすればよいのでしょうか？二つの方法が挙げられ
ます。まずは既に学習した方法です。

main

{

// このスクリプトは infoボックスを画面上に表示します

// 変数テキストの値は "Hello World!"です

…

}

二番目の方法では新しい文字を使用します。

main

{

/* このスクリプトは infoボックスを画面上に表示します

変数テキストの値は "Hello World!"です。 */

…

}

/*と*/文字に囲まれた文字は、コメントとなります。注釈が長くなる場合、または不具合が
発生するコードの大部分をコメント化したい場合には、この方法を使用して下さい。より効果的で
見た目も良いです。

注意
/*と */を使用してコメントを作っても、//を使用してコメントを作ったコードには影響を及ぼしませ
ん。どのコメントもコメントアウトされたままとなります。

コメントは完全にオプションですが、出来る限りたくさん差し挟むようにして下さい。後でこの
忠告が役に立つときが来ます。

さて、ここまでが基本編です。次の章では、ほんの少し中身について、ポイントやポリゴンの動
作法について触れていきます。

第８章：モデラー：ポイントとポリゴン 8.1

第8章 : モデラー：ポイントとポリゴン

モデラーにおける素晴らしさのまさに核となるのが、簡単にポイントやポリゴンの作成が出来る
という点です。モデラーでは一歩先に進めて、このジオメトリを編集するために信じられないほど
強力なツールをいくつも提供してくれています。どの 3Dアーティストも知っているとおり、ポイ
ントとポリゴンは三次元モデルで使用される基本となるジオメトリのコンポーネントです。ですか
らLScriptでジオメトリをどのように作成し、編集するのかを知っておくことは、きわめて重要な
ことなのです。

ポイントとポリゴンの作成
始める前に、まだ触れていないLScriptの領域である操作モードについてカバーする必要があり

ます。L S c r i p tコマンドは二つのカテゴリに系統立てられています、一つは Command
Sequence(CS)モード、もう一つはMesh Data Edit(MD)モードです。既にモデリングとスクリプト
双方においてCommand Sequenceモードで操作を行ってきました。このモードのコマンドはジオ
メトリのグループを集合体として処理するコマンド、例えばmove()やrotate()、bevel()
などやextrude()などです。

Mesh Data Editモードは、使用するには少し注意が必要ですが、もっと泥臭いコマンドがたく
さん用意されています。これらのコマンドを使用するとポイントやポリゴンを作成したり、量数な
どの情報を取得したり、サーフェイス名の割り当てなどが可能になります。これらのコマンドを実
行するためには、editbegin()コマンドを使用して、モデラーをMesh Data Editモードへと切
り替える必要があります。モデラーはMesh Data Editモードへと切り替わりましたから、エラーを
出さずに Command Sequenceモードでのコマンド実行は不可能です。これらのコマンドには
addpoint()やaddpolygon()、pointinfo()、polysurfaceなどがあります。

再びCommand Sequenceモードへと切り替えるには、editend()コマンドを実行しなくて
はなりません。このコマンドは二つの処理を行います。Commmand Sequenceコマンドを走らせ
Mesh Data Edit操作モードにより送られた編集を実際に実行することです。この時点までMesh
Data Editモードで実行されたコードは全てバッファに溜め込まれています。editend()関数を
呼び出すまでジオメトリに対し実際の操作は処理されません。editend()はバッファに溜め込ま
れた修正をキャンセルし、バッファ内の操作を無視するブール値を受け取ります。

この概念をわかりやすく説明するため、三つのポイントを作成するスクリプトを作ってみましょ
う。モデラーのスクリプトですから、例のごとくmain()関数から始めます。



main

{

}

ではeditbegin()とeditend()コマンドを追加してみましょう。

main

{

// Command Sequence モード

editbegin(); // Mesh Data Edit モード

editend(); // Command Sequence モード

}

このスクリプトを実行してみても何も起こりません。モデラーはモードを変更する以外何の指示
も出していないからです。全く何も実行されません。

addpoint()コマンドは、引数として渡された三つの座標値を使用してポイントを 1個作成し
ます。このコマンドをeditbegin()とeditend()コマンドの間に差し込んでください。

main

{

// Command Sequence モード

editbegin(); // Mesh Data Edit モード

addpoint(0,0,0);// point 1 の作成

addpoint(0,1,0);// point 2 の作成

addpoint(1,0,0);// point 3 の作成

editend(); // Command Sequence モード

}

スクリプトを実行してみると、モデラーの正面ウィンドウに三つのポイントが作成されているの
がわかりますね。作成したポイントを基にaddpolygon()を使用してポリゴンを一つ作成して
みましょう。まずはこの新しい機能を扱えるようにオリジナルのスクリプトを変更しなくてはなり
ません。

もともと以下のコードを使用してポイントを作成していました。

addpoint(0,0,0);

このコードではaddpoint()関数へ値0,0,0を送りました。addpoint()はポイントを作
成し、LScriptは自動的に次の行へと移っていました。しかしポイントが作成された後に起こる処理
を見逃していました。関数について解説した内容を覚えていますか？関数の中には処理が終了する
と呼び出したスクリプトへと値を戻すものがあると説明しましたね。

8.2 LScript 日本語ユーザーガイド



今回の場合でも値が返されました。値を取得していなかっただけです。addpoint()コマンド
から返される値は作成したポイントのポイント IDです。この IDは新規ポイントやポリゴンが作成
されたときに作られる内部的な値です。ポイント IDを使用してどのポイントを選択しているのか、
またどのポイントを編集したいのかを識別します。ではスクリプトを修正してみましょう。

main

{

//Command Sequence モード

editbegin(); // Mesh Data Edit モード

point[1] = addpoint(0,0,0); // point 1 の作成

point[2] = addpoint(0,1,0); // point 2 の作成

point[3] = addpoint(1,0,0); // point 3 の作成

editend(); // Command Sequence モード

}

このスクリプトを実行してみると、前と全く同じように実行されているのがわかりますね。差と
いえば、addpoint()関数が返す値にあります。point[1]から始まる変数配列には、
addpoint()コマンドから返される値が入ることになります。これを証明するために、変数配列
を使用してもう一つの新規コマンドaddpolygon()を使用してみます。

三角形を作りましょう。このスクリプトでは上記スクリプトと全く同じですが、コメントが取り
除かれMeshDataEditブロックに1行追加されています。

...

main

{

editbegin();

point[1] = addpoint(0,0,0);

point[2] = addpoint(0,1,0);

point[3] = addpoint(1,0,0);

polygon = addpolygon(point);

editend();

}

...

見てわかるとおり、addpolygon()コマンドは作成されたポイントからポリゴンを作成しま
す。変数配列point[1], point[2]そしてpoint[3]に保存されたポイント IDを渡し、ポ
リゴンを作成するのです。モデラーでポリゴンを作成するときと同様、addpolygon()コマンド
ではポイントの順番が非常に大切になります。作成した順に値を渡すことでポリゴンの法線の向き
を決定します。配列を逆順に並べると、ポリゴンの法線は反転します。

ポイント作成時における実例で示したとおり、addpolygon()コマンドも同様の関数ですか
ら、保存用の変数へとポリゴンの IDを返します。この値を他のLScriptコマンドで使用してみると、
三角形ジオメトリの編集が可能になります。ここではpolypointcount()関数へとポリゴン
IDを渡してみましょう。この関数はポリゴン内のポイント数を返すだけの簡単な関数です。既にポ
イント数が何個あるかは知っていますが、ポリゴン IDを簡単にテストできます。

でも一つ見落としていることがあります。polypointcount()関数はMeshDataEditモード
でのみ動作するという点です。MeshDataEditモードでは全ての編集はバッファに溜め込まれ、
editend()コマンドが呼び出されるまで編集は処理されないんでしたね。これは難問です。ポリ
ゴンはeditend()関数が呼び出されるまで作成されないわけですから、当然ポリゴンも作成さ
れておらず、このMeshDataEditセッション内ではポリゴン IDを取得できなくなります。

editend()を呼び出しポリゴン IDを作成してから、もう一度editbegin()関数を使用し
MeshDataEditモードへと切り替える必要があります。ですから、関数を呼び出して戻り値を保存
しておいた上でCommand Sequenceモードへと切り替えるようにしましょう。

main

{

editbegin();

point[1] = addpoint(0,0,0);

point[2] = addpoint(0,1,0);

point[3] = addpoint(1,0,0);

polygon = addpolygon(point);

editend();

editbegin();

pntCnt = polypointcount(polygon);

editend();

info(pntCnt);

}

注意
このコードは他のスクリプトには使用しないほうが良いでしょう。ポイント数の情報は最初の
editbegin()コマンドを呼び出すよりも前に既に把握しているからです。

ポイントとポリゴンの編集
ここまではポイントやポリゴンの作成法について解説してきましたが、既に存在しているポイン

トやポリゴンを編集したい場合には、どうすればよいのでしょうか？通常はカレントレイヤー上に
あるジオメトリを調整したいでしょうから、このデータを取得できなければなりません。処理を行
うジオメトリがポイントであろうとポリゴンであろうと、データへのアクセスや修正はLScriptで
簡単に行えます。これを実演するために、ポイントを統合するスクリプトを作っていくことにしま
しょう。

例： weldfast

問題：統合後に現れるinfo()ボックスをどうにかして取り除くことにより、統合処理を効率
的に行わなければなりません。モデラーのポイントの統合ツールはこの目的にぴったりなのですが、
何千回も一緒にポイントを統合するとなれば、話は別です。

調査：まずはドキュメントとリリースノートにさっと目を通して、この問題にアプローチする最
良の手段を考えてみて下さい。またLightWaveのCD-ROMやオンライン上にある動作可能な既存
のスクリプトから研究してみることも出来ますね。これらのスクリプトは私たちの求めているもの
そのものではありませんが、どのように問題へとアプローチしていけばよいのかについて洞察力が
養われるでしょう。注目すべきは以下の点です。

1 ポイントの位置情報を取得する方法

2 三次元空間でポイントを動かす方法

3 それらを統合する方法

解決法：モデラーの統合ツールを作り直し、こうるさいinfo()ボックスを省略します。

さらに調査を重ねると、スクリプト作成の際に従う全体的な攻略法が作成できます。

1 選択されているポイントのリストを集めます

2 最後に選択されたポイントの座標値を確定します

3 各ポイントを最終ポイントの座標位置へと移動します

4 ポイントを統合します

コード：

いつもどおり、main()関数から始めていきましょう。

main

{

// 選択されているポイントのリストを集めます

// 最後に選択されたポイントの座標値を確定します

// 各ポイントを最終ポイントの座標位置へと移動します

// ポイントを統合します

}

攻略法のテキストをそのままスクリプト内部へコメントとして入力しているのがわかりますね。
こうすることで計画に沿って予定通りにスクリプトを組んでいくことが出来ます。またどこまで進
んだのか、どのくらい作業が残っているのかも一目でわかります。

weldfast.lsとしてコードを保存します。

さて、調査したいポイントやポリゴンはどれなのかを決定しましょう。全てのジオメトリに対し
てなのか、それとも現在選択されているジオメトリに対してだけ処理を行うのかが選択できます。
今回の目的では権限をユーザーの手に委ねたいので、レイヤーで現在選択されているポイントに対
してのみ処理を行うようにします。selmode()コマンドを使用すると、モデラーに対し選択して
いるジオメトリだけを編集するように指示できます。

main

{

// 選択されているコンポーネントに対してのみ編集します

selmode(DIRECT);

// 選択されているポイントのリストを集めます

// 最後に選択されたポイントの座標値を確定します

// 各ポイントを最終ポイントの座標位置へと移動します

// ポイントを統合します

}

このコマンドには三つのオプション値があります。選択しているジオメトリに対してだけなら
DIRECT、暗黙の選択（例えば Include/Exclude選択を仮定）に対してならUSER、レイヤー内の
全てのジオメトリに対してなら GLOBAL という値を用意しています。このコード行では
selmode()関数へ実際に値を送信していますが、大文字変数が定数を表しているということを思
い出してください。LScriptの中心部ではハードコード化された変数GLOBAL=1, USER=2,そして
DIRECT=3という値を保持しています。各アイテムがどの数値で表されているのかを覚えるより
も、これらの変数名称の方がはるかに覚えやすいので、この定数を使うことにしましょう。

次に実際にポイントが選択されているのかを確認し、エラー発生の可能性を取り除きたいと思い
ます。pointcount()コマンドを使用して、現在選択されているポイント数を確定します。

main

{

// 選択されているコンポーネントに対してのみ編集します

selmode(DIRECT);

// 現在選択されているポイント数を取得します

pntCnt = pointcount();

// 選択されているポイントのリストを集めます

// 最後に選択されたポイントの座標値を確定します

// 各ポイントを最終ポイントの座標位置へと移動します

// ポイントを統合します

}

コマンドをテストするため、info()ボックスに値を入れ、pntCntの値をチェックします。
次にモデラーにいくつかジオメトリを作成し、ポイントを選択してからスクリプトを実行します。
選択されているポイント数が正確に出力されているはずです。

またポイントが何も選択されていない場合、pntCnt変数の値がゼロになるかもテストしてみ
たいですね。値がゼロの場合、ユーザーに対しエラーメッセージを表示させましょう。

if(pntCnt == 0)

error(“No points selected.”);

error()関数はinfo()とよく似た動作を行います。info()関数がダイアログボックスを
開いてスクリプトを実行し続けるのに対して、error()はエラーメッセージを表示しスクリプト
の実行を中断します。ユーザーエラーの可能性に十分注意しながら、スクリプトの中身を作成して
いきましょう。

現在選択されているポイントのリストが欲しいですね。幸運なことに、今までに使用したことの
あるコマンドがここでも使用出来ます。editbegin()関数はMeshDataEdit操作モードに切り替
えると同時に、points[]とpolygons[]変数配列も作成します。スクリプトがMeshDataEdit
操作モードに切り替える時に、LScriptはこれらの配列を自動的に作成し、その配列に現在選択され
ているコンポーネントを入れるので、スクリプトで大変役に立つのです。

...

// 選択されているポイントのリストを集めます

editbegin();

// 最後に選択されたポイントの座標値を確定します

// 各ポイントを最終ポイントの座標位置へと移動します

editend();

// ポイントを統合します

...

このコードではpoints[]と呼ばれる変数配列を作成し構成します。三つのポイントを選択し
ている場合には、変数は以下のようになります。

points[1] = ポイント ID #1

points[2] = ポイント ID #2

points[3] = ポイント ID #3

都合の良いことに、配列のインデックス（括弧内の番号）は各ポイント番号を表しています。で
すからpoints[1]は実際にポイント1を表します。これら値の扱い方がうまく処理されます。

次のステップでは最後に選択されたポイントの座標値を取得しましょう。この座標値は選択され
ている他のポイント全てを移動させる座標値となります。ポイントの座標値データを取得するため
にMeshDataEditコマンドであるpointinfo()を使わなくてはなりません。

...

// 最後に選択されたポイントの座標値を確定します

lastPntPos = pointinfo(points[pntCnt]);

...

引数としてポイント IDを指定すると、pointinfo()はx, y, z座標値を表すベクトルの値を返
します。今回の場合、関数にはpoint[]配列のポイント IDを渡します。3点が選択されているた
め、変数pntCntに保存されている値のインデックス番号を選択し、変数pntCntには選択され
たポイント数が設定されます。配列内の最後のインデックスがサイゴン選択されたポイントである
と仮定しても差し支えなくなります。その結果、pntCntも最後に選択されたポイントと等しくな
ります。

最後に選択されたポイントのベクトル座標値を見るため、pointinfo()行の後にinfo()
ボックスを挿入します。以下のような表示が出てくるでしょう。

<0.12, 4.53, 2.3>

次に、選択された各ポイントを通し値を取得して新しい座標位置へと移動できるようにしたいで
すね。"各ポイントを通して"という部分からはループ処理が必要になることがわかります。ループ
は中断処理がかけられるまでコード行を繰り返すコントロール構造でしたね。このコマンドを
MeshDataEditブロックに挿入する必要があります。

ではこのループを作成するにはどうすればよいのでしょうか？ループ構造を正しく動作させるた
めには、データの4つの部分が必要です。

1 始点。ループがカウント処理を始める起点である変数を初期化する必要があります。今回の場合
には配列の最初のポイントが起点となります。

2 終点。ループ処理を終了する終点を把握しておかねばなりません。これは選択されているポイン
ト数となりますのでループを制御する条件式を構築することになります。

3 変数修正。ループはどのようにカウントしていくのでしょうか？大半のループでは一度に変数を
1, 2, 3といった具合にカウントしていきます。

4 実行させるためのコード。この作業は後で行います。今はこのループ処理のことが心配ですね。
info()式を使用して、正しくカウントされているかを監視することにしましょう。

for文の構造は以下のようになっています。

for( 変数の初期化 ; 条件式 ; 変数の修正 )

値を当てはめてみましょう。開始点はポイント 1であることがわかっていますから、カウント変
数currPntを1からはじめます。

for( currPnt = 1; 条件式 ; 変数の修正 )

pointinfo()文における最終ポイントは変数pntCntを使用して表していますので、この
変数pntCntを使用して選択されたポイント総数を表すことが出来るといえるでしょう。そこで
この値を"ループの回数"用の値として使用することにします。

for(currPnt = 1; currPnt <= pntCnt; 変数の修正 )

このforループ文は、"変数currPntの値が変数pntCntの値以下であれば、forループ文
につけられているコードを実行する"と解釈されます。ではなぜ"<"ではなく"<="なのでしょうか？
条件式で何が行われているのかを正確に説明する例を以下に挙げてみましょう。

currPnt = 1かつ pntCnt = 3の場合には、currPnt < pntCnt がtrueとなり
ます。ですからコードを実行します。

currPnt = 2かつ pntCnt = 3の場合には、currPnt < pntCntがtrueとなり
ます。ですからコードを実行します。

しかし、currPnt = 3かつ pntCnt = 3の場合にはcurrPnt < pntCntがfalse
になります。ですからコードは実行しません。

おわかりのとおり、このコードでは 3番目のポイントに対して処理が行われませんが、処理を実
行したいのです。pntCntはcurrPntより大きくなく等しくなるため、条件式ではfalseに
なります。しかし<の代わりに<=を使用すれば、forループ文は実際 2回ではなく 3回実行され
ることになります。これが今回やりたいことなのです。最初になぜループの回数だけinfo()コマ
ンドを作成するかがわかりましたね。こうすれば何が行われているのか正確に把握することが出来
るのです。

各ループ内の処理が終了した後に変数 currPntの値を 1ずつ追加するように指定し、for
ループ文は終了です。この増加処理がないと、変数currPntに保存されている値は決して変更さ
れません。そうすると無限ループを作成することになってしまい、モデラーはフリーズしてしまい
ます。増加処理を追加しましょう。

for(currPnt = 1; currPnt <= pntCnt; currPnt++)

currPnt++は、実際にはcurrPnt=cuttPnt+1という処理に解釈されます。つまりコー
ドブロックが処理されるたびにカウントが 1ずつ増えるようになっているのです。これで関数は完
成です。

...

editbegin();

// 最後に選択されたポイントの座標値を確定します

lastPntPos = pointinfo(points[pntCnt]);

for(currPnt = 1; currPnt <= pntCnt;

currPnt++)

{

// 各ポイントを最終ポイントの座標位置へと移動します

info(currPnt);

}

editend();

...

スクリプトを実行してみると計画していた通り、スクリプトは選択しているポイント番号を数え
上げていますね。これで後2～3行を残して大部分が完成です！

forループのカウント処理は正しく動いていますので、ポイントの位置を移動させましょう。
forループ文の中のinfo()行を、以下のコード行に書き換えてください。

pointmove(points[currPnt],lastPntPos);

pointmove()コマンドは二つの引数を受け取ります。まず最初に移動させたいポイントの ID
をpointmove()へ渡します。points[]配列内の IDを使用することも出来ますが、ここで少
し、forループマジックといえるような技術を使ってみましょう。配列のインデックス値は整数値
を使用する代わりに変数currPntを参照しているのがわかりますね。この戦術はforループ文
の中では常時使用できます。ご紹介しましょう。

currPnt = 1;

point[currPnt]; // 1番目のポイント ID

currPnt = 2;

point[currPnt]; // 2番目のポイント ID

currPnt = 3;

point[currPnt]; // 3番目のポイント ID

と続きます。

forループは実行し続けると変数currPntの値が増加し、1行のコードで配列内の全ポイン
トが扱えるようになります。

pointmove()へ渡す 2番目のパラメータはポイントの新規座標位置です。今回の場合は、最
後に選択されたポイントから保存された座標位置となります。

スクリプトを実行して全てうまくいっているか確認してください。一つだけやり残したことがあ
ります。現在、ポイントは新規座標位置に移動しただけに過ぎません。このポイントを統合しなく
てはいけません。最後の行では実際にこの処理を行います。

// ポイントを統合します

mergepoints();

mergepoints()コマンドで何の引数も指定しない場合には、このスクリプトでも設定してい
る選択モードで定義されたポイント全てを統合します。統合されるポイントはお互いがぴったいり
と重なり合っている状態にあります。

これで全て完了です！最終的なスクリプトを見てみましょう。

weldfast.ls

main

{

// 選択されているコンポーネントのみを編集します

selmode(DIRECT);

// 現在選択されているポイント数を取得します。

pntCnt = pointcount();

if(pntCnt == 0)

{

error(“No points selected”);

}

// 選択されているポイントのリストを集めます

editbegin();

// 最後に選択されたポイントの座標地を確定します

lastPntPos = pointinfo(points[pntCnt]);

for(currPnt = 1; currPnt <= pntCnt; currPnt++)

{

// 最終ポイントの座標位置に移動します

pointmove(points[currPnt],lastPntPos);

}

editend();

// ポイントを統合します

mergepoints();

}

weldfastスクリプトに基づき、他にも二つ簡単なスクリプトを作成することが出来ます。オリジ
ナルのweldfast.lsと異なる箇所にはコメントを追加しました。

WeldAvg.ls

main

{

//スコープ外で変数を宣言します

totPntVal = < 0, 0, 0>;

selmode(USER);

pntCnt = pointcount();

if(pntCnt == 0)

{

error(“No points selected”);

}

editbegin();

/* 座標値を加算し選択されているポイント数で

除算を行うことでポイント座標値の平均値を計算します */

for(currPnt = 1; currPnt <= pntCnt; currPnt++)

{

currPntVal = pointinfo(points[currPnt]);

totPntVal = totPntVal + currPntVal;

}

pntAvgPos = totPntVal / pntCnt;

// 選択されている全ポイントを移動

for(currPnt = 1; currPnt <= pntCnt; currPnt++)

{

pointmove(points[currPnt],pntAvgPos);

}

editend();

mergepoints();

}

また、次のスクリプトではポイントのグループを統合しています。

main

{

selmode(USER);

pntCnt = pointcount();

if (pntCnt == 0)

{

error(“No points selected”);

}

moveId = floor(pntCnt/2);

editbegin();

for(currPnt = 1; currPnt < (moveId + moveId); currPnt++)

{

info(currPnt);

pntPos = pointinfo(points[currPnt+1]);

pointmove(points[currPnt],pntPos);

/* ポイントのグループを扱うので

変数 currPntは二度値を上げないといけません */

currPnt++;

}

editend();

mergepoints();

}

第９章：VMap  9.1

第9章 : VMAP（頂点マップ）

この章ではVMap（頂点マップ）を扱うための基本的な関数をいくつかカバーしています。サブ
パッチサーフェイスをコントロールするためウェイトマップを作成するにしろ、UVマッピングで
テクスチャマップを使用するにしろ、頂点マップは巨大な力を与えてくれます。LScriptも同様に、
これらの頂点マップを修正するための同等の強力なツールセットを提供します。ですがこの新しい
コードを探求していく前に、LScriptのいくつか新しい領域について解説しなければなりません。

今回作成する例は、単にウェイトマップの値を 50%にするだけのスクリプトですが、途中で新し
いコマンドや関数などを広範囲にわたってカバーしていきます。それでは始めましょう。

通常どおり、main()関数から始めますが、以下のコードではいくつかひねりを加えています。

// プリプロセッサコンパイラ指示子

@script modeler

main

{

// VMap Object Agent を作成します

// 選択されているポイント数を取得します

// 選択されている VMapの値を 50%の重みにします

}



プリプロセッサとは？
プリプロセッサという言葉を解析してみると、"処理の前に起こる"ということを意味しており、

これはそのまま、プリプロセッサコマンドの処理行動を表しています。main()関数のコンテンツ
が実行される前に、コンパイラはプリプロセッサ指示子を検索します。この場合、@script
modelerはLScriptコンパイラに対し、これはモデラーで実行するスクリプトであると指示して
います。スクリプトがプラグインとして追加されるときに、LScriptはプリプロセッサでスクリプト
の種類を確定します。では実際に、プリプロセッサ指示子をいくつか追加してみましょう。

@name “tmp”

@version 2.3

推測どおり、@nameプラグマを使用してスクリプトに名称をつけることが出来ます。この名称
はモデラープラグインのドロップリストとコンフィグ画面に表示されます。@versionプラグマ
はスクリプトが動作するよう設計されたLScriptの最小バージョンを指定します。

バージョンは重要です。なぜならLightWaveのシステムは拡張を続けており、LScriptもまた同
じだからです。現在動作するスクリプトが、前バージョンのLScriptのでは動かない可能性があり
ます。@versionプラグマはユーザーがどのバージョンを持っているかをチェックし、そのバー
ジョンでスクリプトが動作するかを検証します。古いバージョンのLScriptではサポートしていな
いコマンドを使用したスクリプトを実行させようとすると、警告が発せられます。

Object Agentsとは？
Object Agentsはスーパー変数であるとみなされます。Object Agentはデータメンバとメソッド

と呼ばれる二種類の異なるツールを持っています。このツールで自身の変数のようなコンテナ内部
に保存されているデータにアクセスします。Object Agentデータメンバには保存されているデータ
に関する適切な情報が入っています。この情報は名称であったりタイプやフラグ、値などの情報で
す。またObject Agentにはタスクを処理したり値を返すためのメソッドと呼ばれる関数があります。
例としてメソッドにはcount()、getvalue()、setvalue()やsetcolor()などがあり
ます。

main()関数内で試してみましょう。

// VMap Object Agent を作成します

vmapObj = VMap(VMWEIGHT);

// マップが存在しない場合はユーザーに通知します

if(vmapObj == nil)

error(“No weight maps in the mesh!”);

前述のコードには二つの異なるObject Agentsを適用しています。最初の例では特定のVMapの
タイプを参照するためのObject Agentを作成しています。このコードをもう少し詳しく見ていきま
しょう。

vmapObj = VMap(VMWEIGHT);

vmapObjと呼ばれるObject Agentは、VMap()コンストラクタから作成されます。コンスト
ラクタは単に関数的なコードの一部であり、Object Agentを作成して使用出来るようにします。こ
の特殊なコンストラクタが、オブジェクト内の VMapを参照するObject Agentを作成します。
VMap()は引数を一つ受け取ることが出来ます。この引数は参照するVMapの種類です。引数は文
字列もしくは定数どちらを渡しても構いません。以下のリストではVMap()に渡せる定数値とその
値を示しています。

VMSELECT “select”

VMWEIGHT “weight”

VMSUBPATCH “subpatch”

VMTEXTURE “texture”

VMMORPH “morph”

VMSPOT “spot”

VMRGB “rgb”

VMRGBA “rgba”

定数値VMWEIGHTを渡してジオメトリ内のウェイトマップを参照するように指示します。

以下の例のように、このObject Agent内にあるデータメンバにアクセスするには等記号を使用出
来ます。

nameOfVMap = vmapObj.name;

numOfValues = vmapObj.dimensions;

typeOfVMap = vmapObj.type;

ピリオド（.）はObject Agentの名称とそのデータメンバおよびメソッドを区切ります。現在、
作成されたObject Agent vmapObjにはオブジェクト内で作成されている 1番目のウェイトマップ
用データが保存されています。

では以下のコードを追加することで、オブジェクト内に少なくとも一つのウェイトマップが存在
しているかどうかをテストしてみましょう。

if(vmapObj == nil)

error(“No weight maps in the mesh!”);

vmapObjが(nil)の場合、オブジェクト内にはVMapが存在しないということなので、その
状況を通知するエラーメッセージをユーザーに送信します。しかしvmapObjが少なくとも一つの
ウェイトマップを探知した場合には、次のコード行を続行します。1番目のVMapを参照したい場
合には、ifブロックの後に以下のコードを続けます。

info(vmapObj);

しかし、スクリプトをテストする前に、実験用のテストオブジェクトを作成しなくてはなりませ
ん。

テストオブジェクトの設定
LScript作成時には、非常に基本的なオブジェクトでコードをテストしてみたくなる場合が頻繁

に出てくることでしょう。コードが動いているかどうかを示すという、必要最低限の要求に応じら
れるオブジェクトでなくてはなりません。スクリプトが大きくなり、その処理に対する信頼度も増
すにつれ、テストはさらに複雑なものとなり、ついには本物のオブジェクトでスクリプトをテスト
することになります。

今回の場合、必要となるのはポイントに対し"testMap"という名称の1個のウェイトマップが割り
当てられている一枚のポリゴンだけです。作成するスクリプトはそれほど動的なものではありませ
んので、一回限りのコードであると考えられます。つまりこのテストオブジェクト以外では決して
適用されないスクリプトなのです。この章の後の方ではインターフェイスが解説してあります。そ
こでも今回のように一回限りのシンプルなスクリプトを作りますが、複数使用が可能となる強力な
ツールへと変換できます。

基本となるオブジェクトを作成しディスクに保存します（後で再読み込みしたくなるかもしれま
せんから）。例を続けて、以下のコードを入力します。

// VMapのスケールを 50%に設定します

scaleAmt = .5;

// 選択されているジオメトリのみに対し処理を行います

selmode(DIRECT);

// MeshDataEditモードに切り替えます

editbegin();

editend();

ここでは新しいことは何一つ行っていません。変数を作成し、VMapの値をスケールするために
使用するパーセンテージ（小数）を代入しました。それから選択モードを DIRECTに設定し
MeshDataEditモードに入りました。この編集ブロックに以下のコードを挿入します。

// 選択されているポイント数を取得します

pntCnt = pointcount();

// 選択されているポイントを通して反復処理を行うために

// forループ文を設定します

for(pnt = 1; pnt < pntCnt + 1; pnt++)

{

// ループをテストします

info(“point” + pnt);

}

今回はinfo()関数を少し違った形式で使用しました。大規模なスクリプトにおいては多数の
値を出力させたいでしょうが、これらの区別がつけにくくなる場合があります。そこで、文字演算
を使用して、多少なりとも意味のある形式で出力するようにしました。テストオブジェクトを読み
込み、ポリゴンの4点を選択し、スクリプトを実行してください。

ポイントを数え上げる(point1、point2...)4個のinfo()ボックスが出てきますね。

次にinfo()関数を次のコードで置き換えてください。

pntCnt = pointcount();

// カレントのポイントが VMapの中でマッピングされて

// いるかどうかをチェックします

if(vmapObj.isMapped(points[pnt]))

{

// マッピングされていれば VMapの値を取得します

values = vmapObj.getValue(points[pnt]);

// 出力をテストします

info(values);

}

vmapObj Object Agentを使用してウェイトマップがカレントポイント上にマッピングされてい
るかをチェックします。ポイント IDをisMapped()メソッドに渡してチェックします。メソッ
ドはif条件式を解くブール値trueまたはfalseを返します。

次の行ではgetValue()メソッドにポイント IDを渡します。このメソッドはカレントポイン
トに対しVMapで保存されている値を返します。それから info()関数で値をチェックします。
スクリプトを実行すると、0.0～1.0の範囲内にある値が4個表示されます。

あとはVMapの値を現在の値の 50%に設定するだけです。info()行を次のコードで置き換え
てください。

// 値に対する次元数全てにループします

for(x = 1;x <= vmapObj.dimensions; x++)

values[x] *= scaleAmt;

forループは値に対する全次元数に反復処理を行います。今回の場合、ウェイトマップには一
つの値だけが割り当てられています。ループブロックは値の各要素（ここでは一つだけ）に対し 0.5
（50パーセント）の値を持つscaleAmtを掛け合わせます。

values[x] *= scaleAmt;

これは、実際には以下の処理を行っています。

values[x] = values[x] * scaleAmt;

変数values[x]の値にscaleAmtが掛け合わされ、その結果が再びvalues[x]へと保存
されることになります。これはforループ内にある式と同様の簡単な増分です。

最後の行はこうなります。

// 選択した VMapの値に 50%掛け合わせます

vmapObj.setValue(points[pnt],values);

この行では、setValue()メソッドに対し二つの引数points[pnt]とvaluesが渡され
ます。valuesという変数の値が、points[pnt]で表されるポイントのVMapの値と置き換わ
ります。

それでは、処理がよく見えるようにモデラーのOpenGLシェードモードをウェイトマップに変え
てスクリプトを実行してください。値が半分になってのがわかりますね。素晴らしいです。

これが最終コードとなります。

// プリプロセッサコンパイラ指示子

@script modeler

@name “tmp”

@version 2.3

main

{

// VMap Object Agent を作成します

vmapObj = VMap(VMWEIGHT);

// マップが存在しない場合はユーザーに通知します

if(vmapObj == nil)

error(“No weight maps in the mesh!”);

// VMapの値に 50%を掛け合わせます

scaleAmt = .5;

// 選択されているジオメトリのみを扱います

selmode(DIRECT);

// MeshDataEdit モードに切り替えます

editbegin();

// 選択されているポイント数を取得します

pntCnt = pointcount();

// 選択しているポイントを通して反復処理を行う

第９章：VMap  9.7



// forループを設定します

for(pnt = 1; pnt <= pntCnt; pnt++)

{

// VMapがカレントのポイントにマッピング

// されているかどうかをチェックします

if(vmapObj.isMapped(points[pnt]))

{

// マッピングされていれば VMapの値を取得します

values = vmapObj.getValue(points[pnt]);

}

// 値の次元数全てに反復処理を行います

for(x = 1;x <= vmapObj.dimensions; x++)

values[x] *= scaleAmt;

// 選択している VMapの値を 50%にします

vmapObj.setValue(points[pnt],values);

}

editend();

}

あとほんの少しのコードとインターフェイスを追加して、何が出来るか見てみましょう。

// プリプロセッサコンパイラ指示子

@script modeler

@name “tmp”

@version 2.3

main

{

// VMap Object Agent を作成します

vmapObj = VMap(VMWEIGHT);

// マップが存在しない場合はユーザーに通知します

if(vmapObj == nil)

error(“No weight maps in the mesh!”);

/* while文を使用することで、VMapの値が存在し、

VMapのタイプが VMWEIGHTである場合には

変数に名称を付け加えます。これはオブジェクト内の

複数の VMapを扱います */

while(vmapObj && vmapObj.type == VMWEIGHT)

{

vmapnames += vmapObj.name;

// .next()メソッドを使用して次の VMapに行きます

vmapObj = vmapObj.next();

}

/* これはインターフェイス部分のコードであり

後ほど詳細を説明します */

reqbegin(“Scale Weight VMap”);

c1 = ctlpopup(“VMap”, 1, vmapnames);

c2 = ctlnumber(“Scale by (%)”,50.0);

return if !reqpost();

vndx = getvalue(c1);

scaleAmt = getvalue(c2)/100;

reqend();

/* ドロップリストからユーザーが選択した VMapの名称で

再び Object Agentを作成します */

vmapObj = VMap(vmapnames[vndx]);

// この場合は Object Agentは作れません

if(vmapObj == nil)

error(“Could not instance_VMap`”, vmapnames[vndx],
“`!”);

// 選択されているジオメトリのみを扱います

selmode(DIRECT);

// MeshDataEdit モードに切り替えます

editbegin();

// 選択されているポイント数を取得します

pntCnt = pointcount();

// 選択されているポイントを通して反復処理を行う

// forループを設定します

for(pnt = 1; pnt <= pntCnt; pnt++)

{

// カレントのポイントに対し VMapがマッピングされているかを

// チェックします

if(vmapObj.isMapped(points[pnt]))

{

// マッピングされていれば VMapの値を取得します

values = vmapObj.getValue(points[pnt]);

// 値の次元数全てに対し反復処理を行います

for(x = 1; x <= vmapObj.dimensions; x++)

values[x] *= scaleAmt;

// VMapの値に 50%を掛け合わせます

vmapObj.setValue(points[pnt],values);

}

}

editend();

}

第１０章：サーフェイスとレイヤー 10.1

第10章 : サーフェイスとレイヤー

モデラーにおけるサーフェイスの処理はLScriptにとっては簡単な仕事です。サーフェイスとは
基本的に名称を表す文字列であり、ポリゴンはサーフェイスに属しています。この情報を編集する
ためのたくさんのツールが使用できます。また少数ではありますが、比較的苦労せずにサーフェイ
スを作成したり編集するための、自由に使える強力なコマンドが用意されています。

レイヤーはLightWaveのモデリングにおける重要な側面です。レイヤーのコントロール法を把握
していれば、モデラーやLScriptの能力を十分に活用することが出来るのです。一般的にLScriptに
おけるレイヤーはモデラーにおいての動作とほぼ同じです。どのレイヤーが前景レイヤーなのか、
背景レイヤーなのか、ジオメトリはどのレイヤーに属しているのかなどを確定します。

この章ではポリゴンのサーフェイス名称に基づいてオブジェクトをばらばらにするスクリプトを
作成し、LScriptの二つの側面を解説していきます。ポリゴンを小さな断片に分離させた後、新規オ
ブジェクトを別個のレイヤーへと配置します。テストオブジェクトとして、NewTekのお気に入り
である牛のオブジェクトを使っていきましょう。牛のオブジェクトには程よい量のジオメトリがあ
り、サーフェイスも数種類用意されているからです。

では始めましょう。

@script modeler

@version 2.3

@name “FissureLite”

main

{

// サーフェイス名称を取得します

// サーフェイスが付けられているポリゴンを選択します

// 選択しているジオメトリをカットします

// 空レイヤーを検索します

// 新しいレイヤーにジオメトリを貼り付けます

}

まずは予備データをいくつか取得しましょう。main()関数に次のコードを入力します。



// USERモードに切り替えます

selmode(USER);

/* 現在の前景レイヤーを取得します

これがオブジェクトが入っているレイヤーとなります */

workLayer = lyrfg();

// サーフェイスリストから 1番目のサーフェイスを取得します

currSurf = nextsurface();

lyrfg()関数は、どのレイヤーがカレントの前景レイヤーかを示す単数もしくは複数の整数値
を返します。この値を変数workLayerに保存しておき、オリジナルのオブジェクトが入ってい
たレイヤーを保存しておきます。

サーフェイスは、モデラーで管理されている内部リスト内に整列化されています。next
surface()コマンドを使用することで、このリストを通して見ることが出来ます。コマンドに対
し引数が何も与えられない場合は、リスト内の 1番目のサーフェイスを返します。しかしコマンド
にサーフェイス名称が与えられると、この引数を参照点として使用し、リスト内にある次のサー
フェイスを返します。上記のコードでは1番目のサーフェイス名称を検索しています。

while()コマンドは、スクリプト組み立て時に使用するコマンドの中で最も危険なコマンドの
一つです。不正に使用すると無限ループに落ちいってしまい、手動でモデラーを強制終了する破目
になり、データを全て失ってしまいます。このような結果を避けるため、while()コマンドは非
常に慎重に設定します。重要：必要となるコードを全て配置し安全に使用出来るまで、このコード
はテストしません。

// サーフェイスの最後までループを繰り返します

while(currSurf != nil)

{

}

ループ処理が終了した段階で、モデラーの内部リストにある全サーフェイスを通して実行したこ
とになります。変数currSurf内の文字列値がnilになった段階でループを終了します。です
から変数currSurfに対し、ある時点でnilを代入しないと無限ループとなってしまいます。
while()コマンドの中でfalseとなるような条件式を一時的に設定することにより、この問題
を解決し、スクリプトを終了させます。

// サーフェイスの最後までループを繰り返します

while(currSurf != nil)

{

// ループを終了させます

currSurf = nil;

}

ループ処理は強制的に終了させましたから、スクリプトを安全にテストすることが出来ます。ス
クリプト実行時には何の処理も行いませんが、スクリプトは無事終了し、モデラーがコントロール
できるようになります。ループから抜け出すためにbreakコマンドを使用することも出来ますが、
どんな値が条件式をfalseにするかをテストしたかったのです。currSurfの値にnilを設定
出来ることがわかりましたので、ループ処理は終了します。

この条件式はそのように動作するよう設計されています。サーフェイス名称に対するリスト内に
おいて、検索を中断する時期を把握する方法が必要でした。幸運なことに、リスト内の最後のサー
フェイス名は名称に対し常にnilの値を取ります。ですからリストを通し最終的に終端まで検索
すると、nilの値が示されることが保証されています。ここで終端まで来たかどうかを確認する必
要があります。

// サーフェイスの最後までループを繰り返します

while(currSurf != nil)

{

// サーフェイスに属するポリゴンを選択します

selpolygon(SET, SURFACE, currSurf);

// ループを終了させます

currSurf = nil;

}

selpolygon()関数は、非常に複雑なコマンドです。使用時に最もわかりにくいコマンドの
一つです。今回の簡単な例においてこのコマンドには三つの値を渡しています。最初の二つの値は
メジャーまたはマイナーモードを表す定数値です。SETはポリゴンを選択し、CLEARは選択を解
除します。SURFACEはマイナーモードを表す引数で、選択ポリゴンに対するパラメータとして
サーフェイス名を使用することを指定しています。3番目の引数は、検索する実際のサーフェイス
名称です。

cowオブジェクトにスクリプトを実行してみましょう。スクリプトが終了した時点で、目が選択
されているはずです。リストの中においてCowEyesサーフェイスが 1番目のサーフェイスだから
であり、selpolygon()コマンドを使用して目のポリゴンが選択されたのです。

リストには、ジオメトリを何も保持していないサーフェイス名称が含まれている場合があります。
そのためリストをスキャンするときに、実際にポリゴンが選択されているのかを確認しなくてはな
りません。このチェック処理を行わないと、スクリプトはジオメトリを何も選択せずcut()コマ
ンドを使用してしまい、オブジェクト全体が除去される場合があるのです。これは避けたいので
selpolygon()コマンドの後に以下のコードを入力します。

...

pntCnt = pointcount();

if(pntCnt)

{

// 条件式をテストします

info(“Got Some!”);

}

...

edit ブロックは前章で何度も処理してきましたから、もうお馴染みですね。point
count()を使用して、選択されているジオメトリのポイント数を取得します。その後、if文で
値が実際に変数pntCntに保存されているのかを判定します。この条件文はあまりに簡単すぎて
何も出来ないのではないかと思われるかもしれませんが、変数や条件式に関する規則を思い出して
ください。

まず、条件式はifブロック内にある値がtrueかfalseかにより、コードを実行するのかし
ないのかを依存しています。内部ブール値trueとfalseはそれぞれ値1と0を持っています。
実際にはtrueの値はゼロ以外の値ならどの値でもtrueとなります。ですからpntCntが0で
ない限りはtrueであるとみなされ、ifブロック内のコードを実行します。これはまさに今回機
能してほしいことですね。テストしてみると、info()ボックスに“Got Some!”という文字が表
示されるでしょう。つまりスクリプトは予測どおりに動いたことになります。

次にinfo()行を以下のコードで置き換えてください。

// 選択されているジオメトリをクリップボードに移します

cut();

// 空レイヤーを検索します

emptyLyr = lyrempty();

// そのレイヤーへと切り替えます

lyrsetfg(emptyLyr[1]);

// クリップボードからレイヤーへとジオメトリを貼り付けます

paste();

まずはcut()コマンドでオブジェクトからポリゴンを削除し、モデラーのクリップボードへと
配置します。それから次に利用できる空レイヤーを検索します。これはlyrempty()関数を使用
して処理します。この関数（とlyrfg()）は、安全に使用出来る空レイヤーを表す整数配列を返
します。空レイヤーに切り替えるためlyrsetfg()コマンドを使用し、引数として整数値を指定
します。emptyLyr配列変数における 1番目のインデックスを使用すると、使用可能な 1番目のレ
イヤーが表示されます。最後にpaste()コマンドを使用し、クリップボードからジオメトリを取
得して、空レイヤーへと挿入します。

このスクリプトを実行すると、牛の目が切り取られ、他のレイヤーへと貼り付けられましたね。

さてループ処理を終了させるためのコードを、より論理的なものへと置き換えなくてはなりませ
ん。条件式にfalseを設定しているコード行を削除し、while()ループの一番下の箇所に以下
のコードを挿入します。

// リファレンスとしてカレントのサーフェイスを使用し

// 次のサーフェイスを取得します

currSurf = nextsurface(currSurf);

/* ループが続行するようであればジオメトリの残りの部分がある

オリジナルのレイヤーへと戻ります

オリジナルのレイヤーは変数 workLayerに記憶しておきましたね */

lyrsetfg(workLayer);

現在、変数currSurfに保存されているサーフェイスをnextsurface()関数のリファレ
ンスとして使用することが可能です。この 2行のコードはwhileループ文と結びつけられており、
サーフェイスの名称がnilになるまで反復処理を行います。nilになった時点でスクリプトは終
了します。

これで完成です！以下が最終的なコードとなります。

@script modeler

@version 2.3

@name “FissureLite”

main

{

// USER モードに切り替えます

selmode(USER);

/* 現在の前景レイヤーを取得します

これがオブジェクトが入っているレイヤーとなります */

workLayer = lyrfg();

// サーフェイスリストから 1番目のサーフェイスを取得します

currSurf = nextsurface();

// サーフェイスの最後までループを繰り返します

while(currSurf != nil)

{

// サーフェイスに属しているポリゴンを選択します

selpolygon(SET, SURFACE, currSurf);

pntCnt = pointcount();

if(pntCnt)

{

// 選択されているジオメトリをクリップボードに移します

cut();

// 空レイヤーを検索します

emptyLyr = lyrempty();

// そのレイヤーへと切り替えます

lyrsetfg(emptyLyr[1]);

// クリップボードからレイヤーへとジオメトリを貼り付けます

into layer.

paste();

}

// リファレンスとしてカレントのサーフェイスを使用し

// 次のサーフェイスを取得します

currSurf = nextsurface(currSurf);

/* ループが続行するようであればジオメトリの残りの部分が

おいてあるオリジナルのレイヤーへと戻ります

オリジナルのレイヤーは変数 workLayerに記憶しておきましたね */

lyrsetfg(workLayer);

}

}

第１１章：変位マップ 11.1

第11章 : 変位マップ

Displacement Mapスクリプトでは時間が経過するとともにオブジェクトのポイント座標位置を
修正することが出来ます。Displacement Map Object Agent用のメソッドを使用すると、オブジェ
クトの外観を修正するため、ポイント座標値データの読み書きが可能になります。Displacement
Mapスクリプトはオブジェクトが溶け出しているような外観を作ったり、慣性の力に従ったり、ま
たは他の複雑なアニメーションを操作することが出来ます。

LScriptにあるLayoutスクリプト構造の大半と同様、Displacement Mapもこのスクリプトクラ
スで使用可能なオプションを利用するため事前定義されている関数群を使用します。スクリプトが
適切に動作するためにこれらの関数がいくつか必要となる一方で、その他は単にオプションの機能
を提供するだけとなります。

create() と destroy()

create()とdestroy()関数は、変数の初期化とスクリプトに対する関数のクリーンアップ
を行います。create()関数はオプション引数としてMesh Object Agentを受け取ります。Object
Agentはスクリプトが適用されているオブジェクトから作成されます。

newtime()

newtime()関数は、シーンのカレントフレームが修正されるたびに呼び出されます。この関
数はレイアウトから三つの引数、Mesh Object Agent（id）とカレントフレーム（frame）、それ
にカレントタイム（time）を受け取ります。Mesh Object Agent（id）はスクリプトが適用され
ているオブジェクトから作成されます。

一般的に、これらの引数は複数の大域変数を通して他の関数でも利用可能になります。

flags()

flags()関数は、レイアウトに対しスクリプトがポイントデータをどのように処理するのかを
指示します。このスクリプトが関数に正しく要求するのはreturn()文を含む単一行だけという
のが一般的です。return()文から渡される二つの定数のうち一つは、スクリプトがWORLD座
標のポイントを処理するのか、それともLOCAL座標のポイントを処理するのかを指示します。定
数が省略されている場合、ポイントはWORLD座標で処理されます。



process()

process()関数は、ポイント修正コードが記述されている最も重要な関数です。この関数は
レイアウトから呼び出されたときにDisplacement Map Object Agent（da）を引数として受け取り
ます。process()関数はフレームが変化するたびにレイアウトにより呼び出されます。

Displacement Map Object Agentには、二つのデータメンバoPos[]とsource[]があります。
どちらのデータメンバもポイント座標値 x, y, zにアクセスします。しかしこれらのデータメンバに
は大きな二つの違いがあります。一つはoPos[]は読取専用であり、source[]はそうでないと
いう点です。もう一つは、oPos[]はオブジェクトの起点（ローカル）からの位置を返し、
source[]はワールド座標における座標値を返すという点です。

load() と save()

load()と save()関数を使用すると、レイアウトはこのスクリプト操作に付属する特定の
データの読み書きが可能になります。選択されたオプションや設定はスクリプトをどのように実行
するのかを決定するパラメータであり、シーンファイル内部に保存されます。シーンが保存または
読み込まれるときに、スクリプトが適用されているアイテムのチャンネル情報が確実に破棄されな
いようにします。

レイアウトがシーンを読み込みスクリプト用のエントリが現れると、load()関数が呼び出さ
れます。この関数内では、想定どおりのスクリプト処理を可能にする設定や値を読み込み、割り当
てます。同様にsave()関数はユーザーがシーンを保存するときに呼び出されます。この関数には、
シーンファイルにスクリプトのデータを系統立てて保存するときに必要なコードが全て含まれてい
ます。

options()

options()関数は、プラグインリストでスクリプトのプロパティをダブルクリックしたり、
編集を選択するたびに呼び出されます。そこにはスクリプトのインターフェイスを設定し実行する
ために必要なコードが全て含まれています。

例: Splat!
まず最初にヘッダー情報を取得しましょう。このスクリプトは最新バージョンから特別なことを

要求してはいないので、version2.3にしておきましょう。Displacement Mapスクリプトであ
ることを指定し、"Splat"という名称にします。このスクリプトをプラグインとして追加すると、
リストに現れます。

@warnings

@version 2.3

@script displace

@name Splat

このスクリプトに対して4つの関数を使用します。ここで設定を行います。

create

{

}

newtime: id, frame, time

{

}

process: da

{

}

それに

options

{

}

後でインターフェイスも扱いますが、いくつかテストできるように初期値を設定しておく必要が
あります。複数の関数にまたがり、これらの値を共有できるよう大域変数として設定しましょう。

@script displace

@name Splat

// 大域変数

splatValue = 0;

create

{

変数 splatValue=0と設定しました。後々、インターフェイスを設定してユーザーから
splatの割合を設定してもらうようにしますが、今はこうしておきましょう。次に、プラグイン
リスト内で名称を表示するためのスクリプトの解説文を簡単に設定することにします。

create

{

setdesc(“Splat!”);

}

さて次に、newtime()やprocess()関数が呼び出されるときにレイアウトから受け取る値
を学ぶため、これらの関数の中にinfo()関数をいくつか挿入してみましょう。

newtime: id, frame, time

{

info(“newTime: id: “, id, “ frame:“,frame, “ time: “,
time);

}

process: da

{

// daはレイアウトから渡される Displacement Map Object Agent

info(“process: “, da.oPos[1]);

}

そろそろスクリプトを保存しても良いでしょう。splat.lsとして保存してください。でもスクリ
プトを実行する前にテストオブジェクトを作成する必要があります。Displacement Mapはオブ
ジェクトにある多くの（あるいは全ての）ポイントの移動処理を行うため、テストオブジェクトは
ジオメトリ数が出来るだけ少ないほうが良いですね。そうすればスクリプトに何か正しくないこと
が発生しても、何百回も発生するわけではないですから。

モデラーで、中心点をワールド座標（0, 0, 0）に持つ簡単なボックスを作成してください。アク
セスするポイントデータは、三次元空間におけるボックスの位置によって劇的に変化します。

オブジェクトをBox.lwoとして保存してください。いろいろなモデリング状況をカバーするため
に、利用可能なテストオブジェクトをいくつも用意しておくほうが良いでしょう。スクリプトをテ
ストする状況が増えるほど、設計ミスやバグを見逃してしまう可能性は少なくなるのです。

レイアウトにBox.lwoを読みこむ

オブジェクトBox.lwoを選択し、オブジェクトの Item Properties（アイテムプロパティ）パネル
> Deform（変形）タブ > Add Displacement Plugin（変位プラグイン追加）ポップアップメニュー
から splat.lsを適用します。

すぐにinfo()ボックスリクエスタが開き、データを表示します。これはエラーメッセージで
はなく、newtime()関数内に設定したinfo()関数の結果です。うっかりリクエスタを閉じて
しまったら、タイムスライダを前後に動かすだけで同じような結果を表示させることが出来ます。
最初のinfo()パネルには、スクリプトが適用されているオブジェクト名称、現在のフレーム番号、
タイムが表示されます。

二番目に表示されるinfo()パネルは、process()関数に挿入したinfo()関数の結果です。
画面上に送られるデータはオブジェクトの最初のポイントの x位置を表しています。その後に続く
info()パネルは、残りのポイントそれぞれの x座標位置を表示します。ボックスは 8個のポイン
トを持っていますから合計 8回パネルが表示されます。このためにも多くのポイントを持つオブ
ジェクトを作成するのは避けたのです。

現在のテストの要求に対し各ポイントの x座標値を見るのはちょっとやり過ぎています。本当に
必要としているのは、スクリプトをテストするためにポイントデータを一つだけ見ることです。現
在オブジェクトのどのポイントに対して処理を行っているのかを把握するため、カウンター変数を
設定しましょう。まずはカウンターを大域変数として設定します。

// 大域変数
splatValue = 0;
pntCounter = 0;
newtime: id, frame, time

もう画面上にポイント情報を送る必要はなくなりましたので、newtime()とprocess()関
数からinfo()関数を削除しましょう。オブジェクトのポイントデータを個別にハンドルするため
の新しいinfo()関数を追加します。以下のコードを追加してください。

process: da
{
pntCounter++;
if(pntCounter == 1)
info("process: point: ", pntCounter, " ", da.oPos[1]);

}

このコードを追加すると、処理されているカレントのポイント番号を追いかけることが出来ます。
スクリプトを実行してみると、最初のポイントだけが表示されているのがわかりますね。しかしこ
の理論の抜け穴を調べるために、フレームを変更してみてください。今までに得たDisplacement
Mapスクリプトについての知識から、イベント全体において全ポイントを通し最初のポイントのX
座標値だけを表示していくだろうと想定することでしょう。しかしそのようには動作しません。ど
こが悪いのかわかりますか？

エラー箇所はここです。カウンターは正しく処理されているものの、スクリプトに対しフレーム
が変更されたときにカウンターをリセットするようスクリプトに指示していないために、エラーが
発生するのです。最初にスクリプトが実行されたときには、1から 8までカウントされます。しか
し、二回目に実行されたときには 9から 16までカウントされます。この処理が続いていくため
pntCounterは再び1になることが二度とないのです。

この問題を修正するためには、フレームが変更された時点でカウンターをリセットするように、
スクリプトに指示してあげるだけでよいのです。newtime()関数に次のコード行を挿入してくだ
さい。

newtime: id, frame, time

{

pntCounter = 0;

}

スクリプトを保存しオブジェクトから古いスクリプトを除去した上で、もう一度新しいスクリプ
トを試してみてください。何度かテストしてみると、タイムスライダが変更されるたびにカウン
ターがゼロへ戻っているのがわかるでしょう。この結果、フレームが更新されるたびに一個の
info()パネルが表示されることになります。

では次に、アニメーションがポイントのX位置にどのような影響を及ぼすのかを見ていきましょ
う。まず始めにこの例を通して使用する簡単なテストアニメーションを設定していきましょう。ア
ニメーションをより簡単に作成するため、プラグインリスト内にあるスクリプトをオフにします。
そうすればスクリプトはスクリーンにメッセージを送信するのを一時的に中断します。次に二つの
アニメーションキーフレームを作成します。0フレームでは 0, 0, 0の位置に、フレーム 10では 2, 0,
0の位置にキーを作成します。これでシーンを保存すれば、シーンがおかしくなってしまっても
シーンを簡単にリセット出来るようになります。

シーンに対しアニメーションをつけたら、スクリプトをもう一度オンにし、毎フレームのアニ
メーションを通して見てみましょう。シーンではポイントのX座標値は変わっているにもかかわら
ず、メッセージに表示されるX座標値は変化していませんね。これはオブジェクトのローカル座標
値を使用した例です。Object AgentのoPos[1]データメンバを使用することで、ポイント座標値
はオブジェクトがシーン内のどこに位置しているかなどを気にせずに、ローカルに決定することが
出来るのです。レイアウトでピボットポイント位置を変更した場合でも同じです。

ポイントのワールド座標へアクセスするにはどうすればよいのでしょうか？Displacement Map
Object AgetのoPosデータメンバにアクセスする代わりに、sourceデータメンバを使用してみ
てください。info()コードの行を以下のように修正します。

info("process: point: ", pntCounter, " ", da.source[1]);

スクリプトを更新し再実行しましょう。タイムスライダを変更してみると、sourceデータメ
ンバとの違いがわかりますね。オブジェクトのポイントをX軸を横切るように移動すると、ワール
ド座標も変更されます。必要となるため、オブジェクトのワールド座標値を使用し、カレントの
process()関数を以下のコードへと置き換えます。

process: da

{

pntCounter++;

if(pntCounter == 1)

{

// ポイントの WORLD座標値を取得します

xPos = da.source[1];

yPos = da.source[2];

zPos = da.source[3];

info(xPos, " ", yPos, " ", zPos);

}

}

新しく追加されたprocess()関数でスクリプトを実行してみると、まだinfo()関数は表示
されるものの、ポイントの x, y, z座標値が表示されていますね。この値が私たちが処理を加える値
となります。

oPos[]とsource[]データメンバには、もう一つ大きな違いがあります。oPos[]を通し
てポイントの座標値は取得できるものの、値を設定することは出来ません。例えば以下の式はエ
ラーが発生します。

da.oPos[1] = 1;

da.oPos[2] = 1;

da.oPos[3] = 1;

ところが、以下のコードは正当です。

da.source[1] = 1;

da.source[2] = 1;

da.source[3] = 1;

source[]データメンバはデータの読み書きどちらも可能ですから、手動でポイントのワール
ド座標値を設定することが出来ます。LScriptではオブジェクトの元のポイント座標値を修正するこ
とは出来ません。

この Splatスクリプトでは、単にオブジェクトジオメトリの最小のY座標値を強制的に配置しま
す。このためにprocess()関数にもう少しラインを追加します。

info(xPos, " ", yPos, " ", zPos);

// ポイントの値が最低許容値よりも

// 小さいかどうかを判定します

if(yPos <= splatValue)

da.source[2] = splatValue;

技術的にはこれで完了です。しかし、最初のポイント値だけをテストしてきましたので、pnt
Counterリファレンス、info()関数や条件式を省いていきましょう。また、コード内でもはや
意味をなさない関数は除去しましょう。

最終的なスクリプトは以下のようになります。

@warnings
@script displace
@name splat
// 大域変数
splatValue = 0;

create
{
setdesc("Splat!");

}

process: da
{
// ポイントの World座標値を取得します
xPos = da.source[1];
yPos = da.source[2];
zPos = da.source[3];
// ポイントの値が最低許容値よりも
// 小さいかどうかを判定します
if(yPos <= splatValue)
da.source[2] = splatValue;

}

このスクリプトの全ての効果をさらに解説するために、フレーム0の値が（0, 2, 0）、10フレーム
の値が（0, 0, 0）となるようにキーフレームを変更しましょう。XZ平面にボックスのオブジェクト
が衝突するさまを見てください。

約束どおり、ここにスクリプト用のインターフェイスコードがあります。process()関数の
ボディマーカーが閉じられた後にこのコードの断片を追加するだけです。

options

{

reqbegin("Splat!");

c1 = ctlnumber("minimum y value: ", splatValue);

return if !reqpost();

splatValue = c1.value;

reqend();

}

変数値やその変数がどこから来るのかなどをうまく説明するため、長たらしく不必要なコードを
いくつかprocess()関数内に残しておくことにします。どうすればprocess()関数をより効
果的に出来るのかを考えてみましょう。以下は極めて効率的なprocess()関数です。

process: da

{

// ポイントの値が最低許容値よりも

// 小さいかどうかを判定します

if(da.source[2] <= splatValue)

da.source[2] = splatValue;

}

例: dynSplat!
インターフェイスでデータを取得するのではなく、他のオブジェクトのワールド座標値を用いて

前回の例でも使用した衝突面を定義していきます。こうすればスクリプトはより動的なものとなり、
アニメーション効果のオプションを提供するようになります。新規スクリプト用に、前回の splat.ls
スクリプトを起点として使用しましょう。

まずスクリプトの名前を変更します。

@name pragma to read:

@script displace

@name dynSplat

// 大域変数

次に、使用する大域変数をいくつか変更する必要があります。変数splatValueを削除し、
以下のコードを追加します。

@name dynSplat

obj;

objName = “Null”;

currTime;

create

{

これらの値には参照しているオブジェクトから作成されたObject Agentやオブジェクトの名称、
さらにシーンの現在時刻が入ります。スクリプトの新しい名称を考慮して、スクリプトの
create()関数を変更しましょう。

create

{

setdesc("dynSplat!");

}

前回のスクリプトではnewtime()関数を全く使用していませんでしたので省いていました。
しかし今回はnewtime()関数を使用してシーンの現在時刻用の新しい変数currTimeを設定
します。currTimeは大域変数ですので、スクリプト内にある関数は全てその値にアクセス可能
です。

newtime: id, frame, time

{

currTime = time;

}

また、参照オブジェクトの名称から作成されたObject Agentを大域変数objへと代入します。
参照オブジェクトの名称は変数objNameに保存されています。現在この値は"Null"となっていま
すが、このスクリプトがシーン内で正しく動作するためには"Null"という名称のオブジェクトが必
要となります。最終的にはこれをカバーするためのインターフェイスを追加しますが、今は手動で
"Null"を設定するようにしましょう。

newtime: id, frame, time

{

currTime = time;

if(objName)

obj = Mesh(objName);

}

お気づきの通り、Object Agentを作成する前に変数objNameに実際に値が入っているのかを
確認する時点でエラー判定を作成しました。速度の関係上、process()関数内ではなく、ここで
代入を行うようにしてあります。process()関数内でObject Agentを作成すると、絶えず呼び出
されることになってしまい、ユーザーにうるさくなってしまいます。newtime()関数はカレント
フレームが変更した場合にのみ呼び出されますから、操作はそれほど目立たなくなります。

最後にprocess()関数を作成しましょう。修正の大半はここで行われます。実際に一から始
めるのが多分良いでしょう。

process: da

{

}

アニメーションしているオブジェクトの位置データにアクセスし、そのY座標値をポイントの最
小値として使用したいのです。まずはMesh Object Agentであるobjが作成されているかどうかを
確認しましょう。

process: da

{

if(obj)

{

}

}

Object Agentの作成に成功していれば、その位置情報を取得できます。

if(obj)

{

pos = obj.getPosition(currTime);

}

getPosition()メソッドは、シーン内の任意の時刻 currTime（この大域変数は
newtime()関数内で設定されています）におけるオブジェクトの位置情報をベクトルで返します。
さて、あとは splat.lsと同様、値を比較するだけです。

{

pos = obj.getPosition(currTime);

if(da.source[2] < pos.y)

da.source[2] = pos.y;

}

シーン内に"Null"という名称のオブジェクトがある限り、このオブジェクトのY座標値を使用し
て衝突面をどこに配置するのかを決定することが出来ます。以下、新規インターフェイスコードに
適応するように二つ変更点を加えた最終スクリプトを記します。

@warnings

@script displace

@name dynSplat

// 大域変数

obj;

objName;

currTime;

create

{

setdesc("dynSplat!");

}

newtime: id, frame, time

{

currTime = time;

if(objName)

obj = Mesh(objName);

}

process: da

{

if(obj)

{

pos = obj.getPosition(currTime);

if(da.source[2] < pos.y)

da.source[2] = pos.y;

}

}

options

{

reqbegin("dynSplat!");

c1 = ctlmeshitems("objects: ", "Null");

if(!reqpost())

return;

obj = c1.value;

if(obj)

objName = obj.name;

reqend();

}

第１２章：オブジェクト置き換え 12.1

第 12 章 : オブジェクト置き換え

Object Replacement（LS-OR）用のLScriptでは、アニメーション中にどのオブジェクトをいつ
使用するのかを制御することが出来ます。オブジェクトは特定のフレームや時刻に基づいて置き換
えることも可能ですし、位置によって置き換えることも出来ます。実際、ほとんど全てのものをオ
ブジェクト置き換えの引き金として使用することが出来ます。プログラマに全て委ねられているの
です。

LS-OR メソッドとメンバ
LScript Object Replacementのコマンドの中心となるのは、 process()関数です。

process()関数では引数Replacement Object Agentを一つだけ受け取ります。Replacement
Object Agentはアニメーションに関する重要な情報をスクリプトに与えるデータメンバ群を提供し
ます。このデータメンバの大半は読取専用であるため、情報を取得することは出来ても変更するこ
とは出来ません。

以下に利用可能なメンバを記します。

読取専用

objIDは置き換えるジオメトリのオブジェクトを表すObject Agentです。objIDによって共
通のObject Agentメソッド全てにアクセスできるようになります。

curFrameはカレントジオメトリに対するフレーム番号を表す整数値です。

curTimeはカレントジオメトリに対するタイムインデックスを表す整数値です。

newFrameは次のフレーム番号を表す整数値です。オブジェクトがこの新規フレームにおいて
異なる外観とならねばならない場合に、新規ジオメトリが読み込まれます。

newTimeは次のタイムインデックスを表す数値です。オブジェクトがこの新規フレームにおい
て異なる外観とならねばならない場合に新規ジオメトリが読み込まれます。ネットワークレンダリ
ングなどでは非連続の時刻間を移動する場合もありますので、curTimeとnewTimeが連続して
いない場合もあります。

タイムインデックスは各フレームで生成される浮動小数点数値です。フレーム番号とは異なり、
タイムインデックスはシーン中のフレーム/秒の設定に依存します。



例：

シーン中で 24フレーム/秒に設定されている場合、フレームに対するタイムインデックスは以下
のようになります。

フレームタイムインデックス(24フレーム/秒)

1/24 もしくは .042

2/24 もしくは .083

3/24 もしくは .125

シーン中で 30フレーム/秒に設定されている場合、フレームに対するタイムインデックスは以下
のようになります。

フレームタイムインデックス(30フレーム/秒)

1/30 もしくは .033

2/30 もしくは .066

3/30 もしくは .1

curTypeは現在のレンダリングの種類を指示する定数値です。スクリプトはこの値を調べ、イ
ンタラクティブなプレビューや実際のレンダリング用として異なるジオメトリを提供することが可
能です。curTypeには以下のタイプがあります。

NONEはカレントのタイムインデックスに対しジオメトリを何も読み込まないことを示していま
す。

PREVIEWはレイアウトプレビューが生成されていることを示しています。

RENDERは完全なレンダリングが処理されていることを示しています。

newTypeは次のフレーム/タイムインデックスのタイプを示す定数値です。NONE, PREVIEW
もしくはRENDERです。

curFilenameはカレントのオブジェクトジオメトリファイルを表す文字列であり、ジオメト
リが読み込まれていない場合には'nil'になります。ディスク上に存在するオブジェクトのフルパ
ス付き名称となります。

書き込み可能
newFilename

newFilenameは、カレントオブジェクトに置き換えるオブジェクトの名称です。LScriptは
この変数を使用して新規ジオメトリをレイアウトへ読み込むので、オブジェクトのフルパス付き名
称を指定してください。

注意
カレントジオメトリを置き換えるときにのみnewFilenameに値を代入してください。カレントオブ
ジェクトを代入しないようにして下さい。

ではObject Replacementのサンプルスクリプトを見てみましょう。このスクリプトではアニメー
ション中にある特定したフレームにおいて二つの異なるオブジェクトを入れ替えるようにします。
スクリプトを管理しやすいように複数のセクションに分けてどのように処理しているのかを見てい
きましょう。

CHANGER.LS

//-----------------------------------------

// Changer

//

どのスクリプトタイプでも、'//'文字はユーザーコメントの領域を表します。コメントに関して
の詳細は107ページ14章をご覧下さい。

@version 2.3

@warnings

@script replace

@name Changer

プラグマ指示子はスクリプトラベルの後に記述します。まず最初にこれらの指示子が処理されま
すので、スクリプト内のどの場所に置いても自由です。しかしスクリプトをより読みやすくするた
めにはスクリプトの冒頭にこの指示子を置くのが良いでしょう。

次に大域変数の宣言領域となります。LScriptには二種類の変数宣言、グローバル（大域）と
ローカルがあります。ローカル変数はスクリプトのある特定の領域に属する変数です。関数内で宣
言されている変数はローカルであると解釈され、変数が終了すると同時に破棄されます。カウン
ター用に変数を 2～ 3使用したい場合には大変便利で、スクリプト内のどこかにある変数を参照す
る必要もありません。

しかし大域変数は全ての関数から利用可能であり、この変数を使用した関数が終了した時点でも
破棄されることはありません。

changeobj1 = "";

changeobj2 = "";

prog = "Changer v1.01";

swapframe1 = 1;

swapframe2 = 30;

大域変数をより詳しく見て、スクリプト実行中に各変数で何を行っているのかを解明していきま
しょう。

changeobj1には置き換える 1番目のオブジェクト名称が入っています。スクリプトのポイン
トは既存のオブジェクトを異なるオブジェクトに置き換えるところにあるのですから、置き換える
オブジェクトの名称を保存しておく場所が必要なのです。

changeobj2は 2番目に置き換えるオブジェクトのファイル名称を保存しておくもう一つの変
数です。

progはスクリプトラベル用の便利なショートカットです。これを使えば一箇所でリクエスタ用
のスクリプト名称を変更することが出来、名称を載せるたびにスクリプト全体を検索しなくてもす
みます。

swapframe1はオブジェクトchangeobj1に対応するフレーム番号です。ユーザーからは
フレーム番号を二つ取得します。これらはオブジェクトを切り替えるフレームです。

swapframe2はchangeobj2に対応する2番目の切り替えフレームです。

それではcreate()関数から始めましょう。

create

{

setdesc(prog);

}

create()関数はスクリプトが読み込まれた最初の時点で呼び出されます。この関数はスクリ
プトが最初に追加された時、もしくはシーンファイルが読み込まれた時に呼び出されます。
create()関数の中ではスクリプト作成時に処理したい変数やその他のアクションなどを準備し
ます。

今回の場合、setdesc()関数を使用して、アクティブなスクリプトを識別させます。関数
setdesc()は一つの文字列変数を受け取り、この文字列がプラグイン行に表示されることになり
ます。変数progのコンテンツを渡すと以下のようになります。

図 12-1. インターフェイスにおけるスクリプト名称

process: ra

{

thisframe = ra.newFrame;

}

実行早々、process()関数でReplacement Object Agentとオブジェクトの全データメンバに
アクセスできるようになります。Replacement Objectは変数raを通してスクリプトに渡されます。
変数raは好きなように呼び出せますが、LScript Editorから提供されるテンプレートの一部ですの
で、ここで使用しましょう。

ローカルとグローバル変数の違いについて先ほど解説しました。process()関数の二行目に
変数thisframeがありますが、この変数はここまで宣言されていません。process()内部以
外では必要としませんので、関数内部において動的に作成します。このthisframeはスクリプ
トの他のセクションからは見えませんが、スクリプトを実行する箇所でフレームの跡を追うためだ
けに使用するので大丈夫なのです。

変数thisframeはReplacement ObjectのデータメンバnewFrameを通して現在のフレーム
を取得します。Object Agentのデータメンバを取得するためには正しい構文を使用します。Object
Agentを入れてある変数名称の後にピリオドをつけ、データメンバの名称を付け加えます。以下、
サンプルを示します。

ObjectAgent.datamember

このスクリプトの正しい構文は以下のとおりです。

ra.newFrame

newFrameは読取専用のデータメンバですから、変数に記入するためだけに使用されます。以
下、代入可能なデータメンバの扱い方を示します。

switch(thisframe)

{

case swapframe1:

ra.newFilename = changeobj1;

break;

case swapframe2:

ra.newFilename = changeobj2;

break;

}

ra.newFrameからの情報を使用して、オブジェクトを変更するべきかどうかをチェックしま
す。

switch()文を使えば、読みやすく簡単に拡張可能なコードで選択を系統立てることが出来ま
す。switch()は変数（今回の場合thisframe）を受け取り、定義しているcaseの一連の値
と比較します。今回は各フレームにおいて検討するため二つの case 文を用意しました。
thisframeの値をswapframe1とswapframe2の両方に対して比較し、値が一致した場合
にはそれぞれのコマンドを処理します。switch()関数はswap変数に対する一連のif()文と
似ていますが、より論理的な形式でグループ化しています。

スクリプトを実行しているうちに、switch()で変数swapframeのどちらかの値と一致し
た場合、ra.newFilenameに変数changeobjのどちらか一方を代入します。これでレイア
ウトでは変数changeobjのどちらかのオブジェクトファイルが読み込まれることになります。

下記のoptions()関数内では、変数swapframe、changeobjのグループ双方に対し値
を割り当てます。

load: what,io

{

if(what == SCENEMODE)

{

changeobj1 = io.read();

swapframe1 = integer(io.read());

changeobj2 = io.read();

swapframe2 = integer(io.read());

}

setdesc(prog);

}

save: what,io

{

if(what == SCENEMODE)

{

io.writeln(changeobj1);

io.writeln(swapframe1);

io.writeln(changeobj2);

io.writeln(swapframe2);

}

}

load()とsave()関数を使用すると、スクリプト用の大切な情報を保存したり取得出来ます。
双方の関数を走らせるためにOBJECTMODEとSCENEMODEという二つのモードがあります。置
き換え情報は特定のオブジェクトに対してではなくシーンファイルのみに保存されるため、どちら
の関数に対してもSCENEMODEを使用します。

SCENEMODEではシーンファイル保存時にsave()関数内の情報が.lwsファイルに書きこまれ
以下になります。

Plugin ObjReplacementHandler 1 Changer

Script I:\PAPERWORK\LSCRIPTDOCS\LSOR\SCRIPTS\CHANGER.LS

X:\Obj02.lwo

1

X:\Obj03.lwo

30

EndPlugin

レイアウトはプラグイン再読み込みの際に必要となる情報を全て取得し、記憶用の値に記入しま
す。保存した順番どおり情報をシーケンシャルに記憶させる方法に注意してください。

また、シーンファイルに書き込まれている情報は全てテキスト（文字）とみなされます。ですか
ら整数変数swapframe1をコマンドwriteln()でシーンファイルに書き込むと、テキストへ
と変換されます。シーンファイルを再び読み込む場合は文字から整数へと変換する必要があります
からよく覚えておいて下さい。integer()コマンドはload()関数内を見てもわかる通り、変
換処理を行います。

options

{

reqbegin(prog);

c1 = ctlfilename("Swap Object",changeobj1,20,true);

c2 = ctlinteger("Swap Frame",swapframe1);

c3 = ctlfilename("Swap Object",changeobj2,20,true);

c4 = ctlinteger("Swap Frame",swapframe2);

return if !reqpost();

changeobj1 = getvalue(c1);

swapframe1 = getvalue(c2);

changeobj2 = getvalue(c3);

swapframe2 = getvalue(c4);

reqend();

}

options()関数内でユーザー用のインターフェイスを作成します。レイアウトでオプション
ボタンを選択するとインターフェイスが表示されます。インターフェイス作成はこのマニュアルの
あらゆる箇所でカバーされていますが、スクリプトを実行しレイアウトにてどのような動作をする
のか見てみましょう。

スクリプトの実行
レイアウトを開く前に、使用する三つのオブジェクトを選択しなければなりません。一つはベー

スとなるオブジェクトであり、後の二つは置き換え用のオブジェクトです。手早くテストできるよ
うに、オブジェクトは小さくて簡単なものにして下さい。テスト用にお気に入りの何万ポリゴンも
ある代表作を選んでしまうと、切り替え時において読み込みに時間がかかってしまいます。

それではレイアウトを開き、Edit Plug-ins（プラグイン編集）パネルでスクリプトをLightWave
に読み込みます。スクリプトが読み込まれたらEdit Plug-ins（プラグイン編集）パネルを閉じ、最
初のオブジェクトを読み込みます。

オブジェクトを読み込んだら、オブジェクトの Item Properties（アイテムプロパティ）パネル
を開き、Geometry（ジオメトリ）タブのObject Replacement（オブジェクト置き換え）プルダウ
ンメニューからChangerを選択します。

図 12-2. オブジェクト置き換えのプルダウンでハイライトになっている

Object Replacement（オブジェクト置き換え）プルダウンのとなりにあるオプションを選択し、
Changerのインターフェイスを開きます。

図 12-3. Changer インターフェイス

パネル右側にある小さな三角形でファイルリクエスタを開き、置き換え用オブジェクトを選択し
ます。

図 12-4. 三角を押してファイルリクエスタを開く

選択したオブジェクトがシーンに挿入されるフレームを数値フィールドに入力します。

図 12-5. オブジェクトが挿入されるフレームを表示するフィールド

オブジェクトを選択し、フレーム番号を修正して'OK'ボタンをクリックします。アニメーション
をプレビューすると、オブジェクトが置き換わっているが確認出来ます。

注意
Object Replacementはキーフレームをスライドさせただけでは処理されません。オブジェクト置
き換えを確認するにはプレビューを作成するかシーンをレンダリングして下さい。

例に沿ってスクリプトを書いてきましたが、以下に完全なコードリストを用意しています。

フルリスト(REPLACER.LS)
@version 2.3

@warnings

@script replace

@name Changer

changeobj1 = "";

changeobj2 = "";

prog = "Changer v1.01";

swapframe1 = 1;

swapframe2 = 30;

create

{

setdesc(prog);

}

process: ra

{

thisframe = ra.newFrame;

switch(thisframe)

{

case swapframe1:

ra.newFilename = changeobj1;

break;

case swapframe2:

ra.newFilename = changeobj2;

break;

}

}

load: what,io

{

if(what == SCENEMODE)

{

changeobj1 = io.read();

swapframe1 = integer(io.read());

changeobj2 = io.read();

swapframe2 = integer(io.read());

}

setdesc(prog);

}

save: what,io

{

if(what == SCENEMODE)

{

io.writeln(changeobj1);

io.writeln(swapframe1);

io.writeln(changeobj2);

io.writeln(swapframe2);

}

}

options

{

reqbegin(prog);

c1 = ctlfilename("Swap Object",changeobj1,20,true);

c2 = ctlinteger("Swap Frame",swapframe1);

c3 = ctlfilename("Swap Object",changeobj2,20,true);

c4 = ctlinteger("Swap Frame",swapframe2);

return if !reqpost();

changeobj1 = getvalue(c1);

swapframe1 = getvalue(c2);

changeobj2 = getvalue(c3);

swapframe2 = getvalue(c4);

reqend();

}

第１３章：カスタムオブジェクト 13.1

第13章 : カスタムオブジェクト

LightWaveのCustom Objectでは、レイアウトに読み込まれている他のメッシュアイテムに対し
レンダリングされない特別なオブジェクトをユーザーが割り当てることが出来ます。このオブジェ
クトはユーザーに対し情報を表示したり、ツール特有のデータを提供したり、LightWaveで事前定
義されていない視覚効果を提供するために使用出来ます。この機能を使うと、簡単なヌルオブジェ
クトを文字列で表示したり、効果範囲を示したり、さらにはより複雑なヘッドアップディスプレイ
にすることも可能なのです。これらは全てユーザーにとっては信じられないほど強力な視覚化ツー
ルですが、さらに重要なのは、LightWaveのレンダラーからは不可視の状態にあるという点です。

カスタムオブジェクトは、スクリプトにより提供されるポイント座標値とエッジのリストによっ
て構築されます。このリストはカスタムオブジェクトになるワイヤーフレームメッシュを定義しま
す。エッジは実線や点線、ダッシュ、ロングダッシュなど様々なメソッドを使用して描画すること
が出来ます。

Custom Objectスクリプトは、以下のレイアウト関数を使用します。

create() と destroy()

create()とdestroy()関数はスクリプトに対する変数の初期化と関数のクリーンアップを
扱います。Custom Objectスクリプトタイプ用として、create()関数はスクリプトが適用される
オブジェクトから作成されたMesh Object Agentをオプションとして受け取ることが可能です。

init() と cleanup()

init()とcleanup()関数は、レンダーパスの開始（init()）と終了（end()）時に呼
び出されます。

newtime()

newtime()関数はシーン内でカレントのタイムインデックスが変更されるたびに呼び出され
ます。Custom Objectスクリプト内において、newtime()関数はレイアウトから二つの引数
frameとtimeを受け取ります。これらの値は大域変数に割り当てるのが一般的で、サポートし
ている複数の関数内で使用出来るようにします。



flags()

flag()関数はレイアウトにフラグの値を返すことでCustom Objectスクリプトクラスに利用可
能なオプションを設定します。現在フラグの値は、SCHEMAだけが定義されています。このフラグ
はレイアウトに対し、スケマティックビューにおける描画をサポートするか否かを指示します。ス
クリプトが描画をサポートしていない場合には、この関数は定義しない方が良いでしょう。一般的
に一つのコマンドだけ、つまりreturn()文だけがこの関数には付加されます。

process()

process()関数には、重要な描画コードの大半が含まれます。オブジェクトに再描画の必要
ありとレイアウトが判断するたびにこの関数が呼び出され、引数を一つ、Custom Object Agentで
あるcaを受け取ります。このObject Agentには、開発者がカスタムオブジェクトの作成/描画を
行うための多数のデータメンバとメソッドが含まれています。

注意
Custom Object Agentで利用可能なデータメンバとメソッドの完全なリストについては第2巻第
29章239ページを参照してください。このマニュアルのリファレンスセクションにあるCustom
Object Agentsを参照してください。

load() と save()

load()とsave()関数を使用すると、このスクリプトの操作に関する特定のデータを読み込
んだり保存することが出来ます。選択されている任意のオプションやスクリプトをどのように実行
するのかを決定する設定は、シーンファイル内部に保存されなくてはなりません。シーンが保存ま
たは読み込まれるときに、スクリプトが適用されているアイテムのチャンネル情報が確実に破棄さ
れないようにします。

レイアウトがシーンを読み込みスクリプト用のエントリが現れると、load()関数が呼び出さ
れます。この関数内では、想定どおりのスクリプト処理を可能にする設定や値を読み込み割り当て
ます。同様にsave()関数はユーザーがシーンを保存するときに呼び出されます。この関数はシー
ンファイルにスクリプトのデータを系統立て保存するときに必要なコードが全て含まれています。

options()

options()関数は、プラグインリストでスクリプトのプロパティをダブルクリックしたり、
編集を選択するたびに呼び出されます。そこにはスクリプトのインターフェイスを設定し実行する
ために必要なコードが全て含まれています。

例：custText.ls
まず最初の例では、スクリプトが適用されているオブジェクトの名称を表示する簡単なカスタム

オブジェクトを作成してみましょう。手始めに、ヘッダー情報を取得し、二つの大域変数を設定し
ます。

@warnings

@version 2.4

@script custom

@name custText

// 大域変数

obj;

currTime = 0;

必要とする情報の一番目の部分は、スクリプトのオブジェクトの名称です。オブジェクトの名称
を表示するために使用するラベルを作成するため、この情報が必要となります。create()関数
から送られるMesh Object Agent引数からこのデータを取得することが可能です。

create: ma

{

obj = ma;

setdesc(“custText”);

}

作成しておいた大域変数maにMesh Object Agentを割り当てることで、スクリプト内の関数全
てからObject Agentのデータメンバとメソッドを利用することが出来ます。このヘッダー部分では
スクリプトの解説文も定義しておきます。

レイアウトにおける他のアニメーションオブジェクト同様、カスタムオブジェクトはXYZ座標
値システムを使用して、三次元空間のどの箇所に描画すればよいのかを決定します。今回のスクリ
プトに対しては、メッシュオブジェクトの現在位置を文字描画の始点として使用します。こうする
と名称テキストが探しやすく、属しているメッシュオブジェクトがわかりやすくなります。

以前使用したグローバルMesh Object Agent用のメソッドを通してメッシュオブジェクトの位置
情報を取得することが可能です。しかし、getPosition()メソッドは関数がMesh Objectのポ
イント情報を取得する時刻を指示する引数を必要とします。シーンの現在時刻はスクリプトの
newtime()関数へと渡されますので、この関数を追加して、引数timeへ大域変数currTime
を代入すればよいのです。

newtime: frame, time

{

currTime = time;

}

大域変数を設定すると同時に、process()は動作し始めます。まずメッシュオブジェクトの
位置を取得したいですね。

process: ca

{

// Mesh Objectから情報をいくつか取得します

objPos = obj.getPostion(currTime);

}

これらの変数をなぜ大域変数にするのかはお分かりでしょう。create()からMesh Object
Agentであるobjを使用するだけではなく、process()からcurrTimeも使用します。次に
実際に文字を描画しましょう。

process: ca

{

// Mesh Objectから情報をいくつか取得します

objPos = obj.getPostion(currTime);

// Object Agentである caを使用してカスタムオブジェクトを作成します

ca.setColor(<100,100,100>);

ca.drawText(objPos, obj.name, X , CENTER);

}

まずはカスタムオブジェクトの描画色を灰色（<100,100,100>）に定義します。最終的には
drawtext()メソッドを使用して、実際にオブジェクトの名称をスクリーン上に描画します。変
数objPosを渡し、データメンバobj.nameを使用してラベルの位置と表示したいテキストを
定義します。それから定数値XとCENTERを渡すことで、ラベル上のテキストの位置合わせを定
義します。

スクリプトを保存しヌルオブジェクトに適用してください。オブジェクトのピボットポイントに
メッシュオブジェクトの名称が表示され、ラベルが描画されているのがわかりますね。またメッ
シュオブジェクトの位置を移動させたりキーフレームをつけてみると、ヌルオブジェクトがアニ
メーション中にどのような動作を取るのかもわかるでしょう。

最終コードは以下のとおりです。

@warnings

@version 2.4

@script custom

@name custText

// 大域変数

obj;

currTime = 0;

create: ma

{

obj = ma;

setdesc(“custText”);

}

newtime: frame, time

{

currTime = time;

}

process: ca

{

// Mesh Objectから情報をいくつか取得します

objPos = obj.getPostion(currTime);

// Object Agentである caを使用してカスタムオブジェクトを作成します

ca.setColor(<100,100,100>);

ca.drawText(objPos, obj.name, X , CENTER);

}

例：Barn.ls
このサンプルでは非常に簡単な家の形をしたカスタムオブジェクトを作成してみましょう。まず

はスクリプトのヘッダー情報を作成するところから始めましょう。

@warnings

@version 2.4

@script custom

@name custBarn

次にカスタムオブジェクトをどのように描画するのかを定義しなくてはなりません。二つの大域
変数vertとedgeを宣言することで定義します。

それぞれオブジェクトを描画するために使用する値のリストを保持します。

vert = @ <0.0, 0.0, 0.0>, <1.0, 0.0, 0.0>,

<1.0, 1.0, 0.0>, <0.5, 1.5, 0.0>,

<0.0, 1.0, 0.0>, <0.0, 0.0, -1.0>,

<1.0, 0.0, -1.0>, <1.0, 1.0, -1.0>,

<0.5, 1.5, -1.0>, <0.0, 1.0, -1.0> @;

edge = @ 1, 2, 2, 3, 3, 4, 4, 5, 5, 1, 6, 7,

7, 8, 8, 9, 9, 10, 10, 6, 1, 6, 2, 7, 3,

8, 4, 9, 5, 10, 2, 5, 1, 3 @;

変数vertには始点ポイントをvert[1]、終点ポイントをvert[10]としたオブジェクト内
の各ポイントの位置を記しています。変数edgeリストでは始点と終点ポイントのインデックスを
リストすることでワイヤーフレームのエッジを定義しています。例えば（上記vertとedgeリス
トを使用すると）

vert[1] = <0.0, 0.0, 0.0>

vert[2] = <1.0, 0.0, 0.0>

edge 1 = 1, 2

これはつまり

edge 1 = vert[1], vert[2]

もしくは

edge 1 = <0.0, 0.0, 0.0>, <1.0, 0.0, 0.0>

オブジェクトが定義されましたので、process()関数を設定し、正しく描画させなければな
りません。総数 17本のエッジが作成されます。注意深くforループ文を設計すれば、1行だけで
簡単に描画することが可能です。

process:

{

// 直線を描画します

for(x = 1;x < 35;x += 2)

ca.drawLine(vert[edge[x]],vert[edge[x+1]]);

}

drawline()メソッドは二つの引数、始点ポイントと終端ポイントを受け取ります。これら二
つの値を提供すると、LScriptはレイアウトに対し二点を結ぶ直線を描画するように指示します。
forループ文では単にverts[]とそれに関連しているedge[]のリストを回していき、正しく
描画されるようにフォーマットしています。ループするたびに変数xの値がどのように増加してい
るのかに注目してください。カウンターxを使用することで最初の頂点にアクセスし、x+1とすれ
ば 2番目の頂点にアクセスが可能です。ループの各回でこのような処理を行い、drawline()メ
ソッドでエッジを一本描画します。

少し複雑になってきましたので、process()関数を修正します。

process:

{

// 直線を描画します

for(x = 1;x < 31;x += 2)

ca.drawLine(vert[edge[x]],vert[edge[x+1]]);

ca.setPattern("dot");

for(x = 31; x < 35;x += 2)

ca.drawLine(vert[edge[x]],vert[edge[x+1]]);

}

再びループを通して描画をコントロールしていますが、17本のエッジが 15本しか描画していま
せん。次に直線のパターンをドットに設定し、デフォルトである実線の代わりにドット状の直線を
描画するように指定します。最後の二つのエッジはその前の 15本のエッジと同じようなforルー
プ分を使用して描画します。

スクリプトを保存しヌルオブジェクトにスクリプトを適用してください。17本で出来た簡単な
家が描画されましたね。

サンプルコード:
以下のコードは、既存のLightWaveモデルを既成のカスタムオブジェクトスクリプトに変換して

くれるモデラーのスクリプトです。コードを入力するのに多少時間がかかるかもしれませんが、こ
のスクリプトはカスタムオブジェクトのスクリプト作成処理を非常に簡単にしてくれます。

//-----------------------------------------------

// Mesh2Custom by Bob Hood

//

// 現在可視状態にあるレイヤー(複数可)にあるメッシュを

// Layout Custom Objectに変換する LScript

//

@warnings

@version 2.3

main

{

vectors = nil;

indices = nil;

pointcount() || error("I need some mesh to work with!");

reqbegin("Mesh2Custom by Bob Hood");

c1 = ctlfilename("Script file","mycustobj.ls");

return if !reqpost();

filename = getvalue(c1);

reqend();

// 配列を構築します

editbegin();

foreach(poly,polygons)

{

pnts = polyinfo(poly);

m = pnts.count();

for(x = 2;x < m;x++)

{

vec1 = pointinfo(pnts[x]);

vec2 = pointinfo(pnts[x + 1]);

ndx1 = 0;

ndx2 = 0;

if(!vectors)

{

// 配列の初期化を行います

vectors += vec1;

vectors += vec2;

ndx1 = 1;

ndx2 = 2;

}

else

{

// 配列内における各ベクトル用のインデックスを検索します

for(y = 1;y <= vectors.size();y++)

{

if(vectors[y] == vec1)

ndx1 = y;

if(vectors[y] == vec2)

ndx2 = y;

}

if(!ndx1)

{

vectors += vec1;

ndx1 = vectors.size();

}

if(!ndx2)

{

vectors += vec2;

ndx2 = vectors.size();

}

}

indices += ndx1;

indices += ndx2;

}

}

editend();

// 重複インデックス値を最適化します

lines

x = 1;

while(x <= indices.size())

{

if(indices[x])

{

y = x + 2;

while(y <= indices.size())

{

if(indices[y])

{

if((indices[x] == indices[y] && indices[x+1] ==
indices[y+1]) || (indices[x] == indices[y+1] && indices[x+1]
== indices[y]))

{

indices[y] = nil;

indices[y+1] = nil;

}

}

y += 2;

}

}

x += 2;

}

indices.pack();

indices.trunc();

// 配列データからカスタムスクリプトを生成します

file = File(filename,"w") || error("Couldn't open output
file!");

file.writeln("@warnings");

file.writeln("@version 2.3");

file.writeln("@script custom"); file.nl();

file.write("vert = @ ");

for(x = 1;x <= vectors.size();x++)

{

if(x > 1)

13.10 LScript 日本語ユーザーガイド



{

file.writeln(",");

file.write(" ");

}

file.write("<",vectors[x].x,",",vectors[x].y,",",vectors[x].z,"
>");

}

file.writeln(" @;"); file.nl();

file.write("edge = @ ");

for(x = 1;x <= indices.size();x++)

{

if(x > 1)

{

file.write(",");

if(x & 1)

{

file.nl();

file.write(" ");

}

}

file.write(indices[x]);

}

file.writeln(" @;"); file.nl();

file.writeln("process: ca");

file.writeln("{");

file.writeln(" for(x = 1;x < ",indices.size() + 1,";x +=
2)");

file.writeln("ca.drawLine(vert[edge[x]],vert[edge[x+1]]);");

file.writeln("}");

file.close();

}

第１４章：プロシージャルテクスチャ 14.1

第14章 : プロシージャルテクスチャ

プロシージャルテクスチャもしくはシェーダーは、オブジェクトのサーフェイス上に現れますが、
実際にはオブジェクトに存在していません。LightWaveはレンダリング時においてプロシージャル
テクスチャを計算するため、最終画像において表示されます。

LScriptの Procedural Texture（LS-PT）スクリプトを使うと、LightWaveがサーフェイスの
レンダリングをコントロールできるようになります。

シェーダー関数
他のレイアウトスクリプトと同様、LS-PTのために特別に設計されている関数が 2～ 3あります。

これらの関数は幾つかの使い道があります。シェーダーを初期化したりスクリプトが現在動作して
いるフレームを決定したり、また flags() UDFの場合には、スクリプトで必要となる特定の
バッファだけをリクエストします。

init()

init()は各レンダーシーケンスの最初に一度だけ呼び出されます。レンダー処理が始まる前
に設定が必要となるスクリプトの値は、ここで初期化しなくてはなりません。

cleanup()

cleanup()はレンダーシーケンスの終了時に呼び出されます。変数の破棄処理を行います。

newtime(frame, time)

newtime()はカレントのレンダーシーケンスにおいて各新規時刻の開始時に呼び出されます。
整数値frameはカレントのフレーム番号を、数値timeはカレントのタイムインデックスを表し
ます。フレーム番号とは違い、タイムインデックスは各フレームから生成される浮動小数点数値で
す。タイムインデックスはシーンに対するフレーム/秒設定に依存しています。



例：

シーン中で 24フレーム/秒に設定されている場合、フレームに対するタイムインデックスは以下
のようになります。

フレームタイムインデックス(24フレーム/秒)

1/24 もしくは .042

2/24 もしくは .083

3/24 もしくは .125

シーン中で 30フレーム/秒に設定されている場合、フレームに対するタイムインデックスは以下
のようになります。

フレームタイムインデックス(30フレーム/秒)

1/30 もしくは .033

2/30 もしくは .066

3/30 もしくは .1

flags()

LS-IFと同様、LS-PTスクリプトもレイアウトに対しサーフェイステクスチャのどの属性に対し
修正を加えるのかを指定しなくてはなりません。スクリプトはflags()関数から以下の値を一つ
もしくは複数返します。

NORMAL

COLOR

LUMINOUS

DIFFUSE

SPECULAR

MIRROR

TRANSPARENT

ETA

ROUGHNESS

RAYTRACE

LS-PT 関数とデータメンバ
LS-PTの主な処理の核はprocess()関数です。プロシージャルテクスチャではprocess()

関数はピクセル単位に呼び出されます。

LS-PTはprocess()関数に対し引数を一つ提供します。この引数はShaderと呼ばれるもう
一つのObject Agentです。Shader Object AgentにはレイアウトLScriptにおけるシーン用Object
Agentのように、シェーダーに関する情報を教えてくれるメソッドやデータメンバが含まれていま
す。

Shader Object Agentsは影響を与えるサーフェイスに関する必要な情報全てを提供します。これ
らの値のうち、制御したりサーフェイスを修正出来るものも幾つかあります。その他は情報を与え
るだけで修正は出来ません。

まずは読み取り専用の情報を見ていきましょう。

sxは最終画像におけるピクセル座標値のスポットのX座標値を表す数値です。(0,0)は左上隅の
座標値となります。

syは最終画像におけるピクセル座標値のスポットのY座標値を表す数値です。(0,0)は左上隅の
座標値となります。

oPos[3]はオブジェクト座標におけるスポットの座標位置を表す浮動小数点数値です。座標値
ですからX,Y,Zを表す三つの数値を返します。

wPos[3]はワールド座標におけるスポットの座標位置を表す浮動小数点数値です。座標値です
からX,Y,Zを表す三つの数値を返します。

gNorm[3]はワールド座標におけるスポットのジオメトリ法線を表す浮動小数点数値です。こ
れはスポットにおける無修正のポリゴンの法線であり、スムージングやバンプマッピングなどに
よって修正される前の法線情報です。座標値ですからX,Y,Zを表す三つの数値を返します。

spotSizeはスポットの直径の近似値をあらわす数値です。エッジ上に見えるサーフェイス上
のスポットは長く薄くなっているので近似値となっています。テクスチャアンチエリアシングを計
算する場合などに使用可能です。

raycast()は、本来raytrace()と同じパラメータを受け取る速度向上版の関数ですが、
レイの長さしか返しません。シェーディングは評価されることはなく、レイトレースの再起も行わ
れません。

raySource[3]はワールド座標における視線レイが発生する起点を表す浮動小数点数値です。
カメラになる場合が多いでしょうが、カメラである必要はありません。座標値ですからX,Y,Zを表
す三つの数値を返します。

rayLengthは視線レイが空間を通ってこのスポットに到達するまでの距離を表す値です。

cosineはスポットにおいて視線レイとサーフェイス法線との角度の余弦を表す値です。これ
はビューの跳ね返りを示し、スポットの近似サイズを測定します。

oXfrm[9]はオブジェクトからワールド座標変換行列を表す浮動小数点数値です。この行列は
他の方法でも計算可能ですが、ここでは速度用にメソッドであり、主として方向ベクトル用に使用
するために用意されています。座標値ですからX,Y,Zを表す三つの数値を返します。

wXfrm[9]はワールドからオブジェクト座標変換用の行列です。この行列は他の方法でも計算
可能ですが、ここでは速度用にメソッドであり、主として方向ベクトル用に使用するために用意さ
れています。

objIDはシェーディングされるオブジェクトを表すObject Agentへのポインタです。

polNumはシェーディングされるオブジェクトのポリゴン数を表す整数値です。これは通常の
メッシュオブジェクトの場合にはポリゴン数を指しますが、非メッシュオブジェクトの場合には他
のサブオブジェクト情報を表現しています。

illuminate(light,position) -> array[6]

illuminate関数は、カレントインスタンスにおいて指定したライトから指定した位置に当
たるライトのレイを表す 6個の数値の配列（色[1-3]と方向[4-6]）を返します。指定したワールド座
標値にライトが全く照射されていない場合にはゼロが返されます。色には（もし存在するならば）
陰やフォールオフ、スポットライトコーン、ライト、それにポイントの間にある透明なオブジェク
トの効果が含まれます。

raytrace(position,direction) -> array[4]

raytrace関数は指定した位置から指定した方向に向かう（ワールド座標で）レイを追跡する
ために呼び出されます。関数はレイの長さを表す要素[1]とその方向からくる色[2-4]を表す 4つの要
素を含む配列を返します。レイの長さが無限であれば-1.0を返します。使用されるdirectionは
出て行くレイの方向であり、単位ベクトルに正規化されている必要があります。

これらの値を読み込み使用することは出来ますが、変更することは出来ません。読み込むだけで
は面白くないので、LS-PTでは好きなようにサーフェイスの値を修正することもできるようにして
います。修正可能な値は下記に限定されています。アイテムの大半の名称はレイアウトにおけるコ
ントロール名称に対応していますので、レイアウトにおいて色・質感編集になれていれば、ここで
も名称をすぐに探し出せるでしょう。

wNorm[3]はワールド座標におけるスポットの新規ジオメトリ法線です。この値を修正するこ
とで、ジオメトリを修正することなくサーフェイスをバンプ状に見せることが出来ます（バンプ
マッピング）。シェーダーは修正後にベクトルを正規化しておく必要があります。

color[3]はサーフェイスカラーを表す赤[1]、緑[2]、青[3]のパーセンテージを表す数値です。
値1.0は100%に相当します。

luminousはサーフェイスの自己発光度を表す数値です（1.0は100%）。

diffuseはサーフェイスの拡散レベルを表す数値です（1.0は100%）。

specularはサーフェイスの反射光を表す数値です（1.0は100%）。

mirrorはサーフェイスの鏡面反射率を表す数値です（1.0は100%）。

transparencyはサーフェイスの透明度を表す数値です（1.0は100%）。

etaはサーフェイスの屈折率を表す数値です（1.0は100%）。

roughnessはサーフェイスの粗さを表す数値です（1.0は100%）。

例：(PULSE2.LS)
LS-PTスクリプトを通してどのようにサーフェイスの値を修正するかを示した基本的な例を見て

いきましょう。

これはPulse2.lsというスクリプトです。サーフェイスの自己発光度と拡散レベルの値を受け取
り、車のウィンカーのように規則的に明滅を起こします。わずかなインターフェイスで明滅の長さ
を調節し、ユーザーはサーフェイスの拡散レベルチャンネルを保護するかどうかを選択できます。

// Pulse2

//

// サーフェイスの自己発光度と拡散レベルを(交互に)反復することで

// サーフェイスを"波動"のように見せます

//

// 更新日 04.21.98 Bob Hood

// 更新日 07.20.01 Scott Wheeler

スクリプトのトップ部分はスクリプト名称や重要な情報を置いておくコメントの領域です。'//'
文字はその行がコメント行であると識別し、スクリプト実行時においてライン上のテキストは全て
無視されます。これらの文字はスクリプトから 2～ 3行を完全に削除するのではなく省きたい場合
にとても便利です。その部分をコメント文にしてしまえばよいのですから。

@version 2.3

'@'文字で始まるコマンドはコンパイラ指示子です。コンパイラ指示子はコンパイラに対しスクリ
プトのパフォーマンスをあげたり、スクリプト実行に必要となるある特定の条件のコンパイラを修
正する場合に指示します。

'@version'指示子はスクリプトを適正に実行させるために必要となるLScriptの最低バージョ
ンをコンパイラに指示します。新規コマンドセットを使用しているスクリプトにはこの指示子を必
ず入れるようにして下さい。指示子はユーザに対しスクリプトが機能するためにはLScriptをアッ
プグレードする必要があるかどうかをユーザに伝えるからです。

コンパイラ指示子はスクリプトのどの箇所でも配置することも可能なのですが、組織化し探しや
すくするためにもスクリプトの一番上にコンパイラ指示子を全て配置しておくのが良いでしょう。

thisFrame;

pulserange = 2;

prog = "Pulse v1.01";

keepdiffuse = true;

次に大域変数とする変数を宣言します。大域変数はまるで関数の一部であるかのように全スクリ
プト関数から可視状態になります。この変数はローカル変数とは対照的であり、ローカル変数は関
数内部においてのみ宣言されます。ローカル変数は宣言されている関数以外の関数からは見ること
は出来ません。ループ処理はどのようにローカル変数が動作するのかを示す良い例です。関数内部
において小さなループ処理を行っている場合、ループに使用される変数は関数外部から見える必要
はありません。この目的はループ自身を越えて拡張することではないからです。

各大域変数の目的を理解する必要がありますから、それぞれ一つずつ見ていきましょう。

thisFrameにはカレントのフレーム番号が入っています。

pulserangeは各パルスの長さを割り当てる初期値です。後ほど見ますが、スクリプト実行
前に他の値を代入します。しかしスクリプトの最初に配置しておいて、スクリプトを通して検索し
追跡しなくてもよいように中央に集めておけば修正が可能になります。

progは現在のスクリプトのバージョンを保持する文字変数です。スクリプトにおいて特に重要
な部分ではありませんが解説文の設定（後ほど確認）がより容易になります。

keepdiffuseはユーザによって設定された拡散レベルの値の情報を保護しておくかどうかを
指定するブール値です。スクリプトの後の方で、サーフェイスの拡散レベルの値を保持または上書
きすることになります。

create

{

scene = Scene();

pulserange = scene.fps;

setdesc(prog+" is Pulsing on "+ string(pulserange)+" Frame
Intervals");

}

create()関数はスクリプトが最初に読み込まれた時点で呼び出される関数なので、初期設定
を行うには最適の場所です。create()関数はinit()関数とは異なります。init()は各レン
ダーシーケンスの最初に呼び出されますが、create()関数はスクリプトが読み込まれたとき、
またはLightWaveが起動されたときに呼び出されます。

create()関数の第 1行は大切な行です。このたった一行の中にたくさんの情報が変数
'scene'に渡されます。Scene()関数を使用することで Scene Object Agentを作成しています。
これでカレントシーンを処理するいくつものデータメンバやメソッドにアクセスできるようになり
ます。

注意
Scene()の動作法や利用可能なタイプについては、リファレンスマニュアルのScene Object
Agentsの章を参照してください。

Scene Object Agentの作成時には、カレントシーンに対し利用可能な情報全てが変数'scene'に
保存されています。アクセス可能な情報の一つとしてシーンのフレーム/秒（fps）があります。で
すからフレーム/秒（fps）データメンバを代入することで、変数pulserangeの設定が可能にな
ります。

create()関数内の最後の行はsetdesc()関数の呼び出しです。この関数ではシェーダーリ
スト内にスクリプトの値を表示した解説行を作成できます。ここではスクリプトの名称を記入して
ある変数'prog'を使用しましょう。オプションパネルを開かずにスクリプトがどのように動作する
のかについてフィードバックを与えられるよう、setdesc()関数を使用することが頻繁にありま
す。スクリプトが始めて実行されると、シーンの fpsが 24に設定されている場合、解説文は以下の
ように表示されます。

図 14-1. インターフェイス上のスクリプトの解説文

newtime: frame, time
{
thisFrame = frame;

}

newtime()関数は新しいフレーム毎に一回ずつ呼び出され、開発者はフレームに基づき情報
を取得したり変更することが出来ます。今回はカレントのフレーム情報を取得し、大域変数
'thisFrame'に代入します。

flags
{
return(LUMINOUS,DIFFUSE);

}

flags()関数では、LScriptに対しサーフェイスのどの属性に対し影響を及ぼすのかを正確に
指示します。flags()関数を追加しなくてもスクリプトは実行されますが、サーフェイスのどの
属性の値を修正するのかコンパイラにきちんと伝える方がメモリにそのバッファだけを読み込みま
すので、はるかに効率的です。

今回の例ではサーフェイスに対しLUMINOUSとDIFFUSEの値を修正します。ですからこの二
つのサーフェイスの属性だけが利用可能になるように指示します。

次の関数はprocess()関数です。process()関数はスクリプト実行を全て処理する場所で
す。LScriptのProcedural Textureでは、process()関数はイメージの各ピクセルに一回ずつ呼び
出されます。process()関数はnewtime()の後に呼び出されるため、フレームが変更される
ことにより、集められた情報を使用出来ます。

process: sa

{

if(thisFrame == 0)

return;

process()関数でShader Object Agentにアクセスすることが出来ます。このAgentは関数へ
と渡される変数'sa'です。ですからサーフェイスに関する情報を得たいとき、またサーフェイスの
値を変更したいときには毎回、'sa'変数を通して処理することになります。これは関数内のより下
のところで意味をなしてきます。

ゼロで除算してしまうなどのエラーを避けるために、このスクリプト処理では 0フレームを除外
しています。カレントフレームを判定し、0であればreturnコマンドを呼び出してprocess()
関数から抜け出します。

j = thisFrame;

if(thisFrame > pulserange)

{

i = integer(thisFrame / pulserange);

if(i != 0)

{

j = thisFrame - (i * pulserange);

}

k = pulserange / 2;

if(j <= k)

{

// 明滅をオンにします

sa.luminous = j / k;

// 逆の値を生成します

if (!keepdiffuse) sa.diffuse = 1.0 - (j / k);

}

else

{

// 明滅をオフにします

// パーセンテージが正しくあるように折り曲げます

j = j - k;

sa.luminous = 1.0 - (j / k);

if (!keepdiffuse) sa.diffuse = j / k;

}

}

上記コード部分は一見したところでは複雑に見えますが、全体を見て、それから細かく分けてい
きましょう。このコードでは現在のフレームとパルスの長さを読み込むことで、時間が経つにつれ
てサーフェイスの自己発光度と拡散レベルの値を変更していきます。では処理を二つの部分、パル
ス総時間の半分である明るい状態と、再び暗くなった状態とに分けて見ていきましょう。

拡散レベルの値はブール変数'keepdiffuse'によって保護されるかどうかに気をつけてくださ
い。これは自己発光度が増したときに、サーフェイスの拡散レベルを下げるのか、それともそのま
まの状態にしておくのかをコントロールするユーザー定義の値です。

コードセクションがどのように動作するのかを見るには、関係する変数全てによる表を構築し、
時間の経過と共にそれらの変数がどのように変化していくのかを見てみるのが良いでしょう。下記
の例においてサンプルフレームの表を作りました。この表ではパルスのレートを 20フレームと仮定
し（10フレームアップ、10フレームダウン）、自己発光度と同様にサーフェイスの拡散レベルも修
正しましょう。

このように 2～ 3個のフレームで実行してみれば、最終的にどのようになるのかがわかるでしょ
う。実行する時には条件式を忘れずに全て処理するようにしてください。if文を誤ると、途中で
値が変更してしまうからです。

save: what, io

{

if (what == OBJECTMODE)

{

io.writeNumber(pulserange);

io.writeNumber(keepdiffuse);

}

}

save()関数を使用するとオブジェクトファイルの中にシェーダー情報を保存することが出来
ます。今回の場合、変数'pulserange'と'keepdiffuse'の値を保存しましょう。

save()関数には二つの個別のモードがあります。SCENEMODEではプラグイン情報をシーン
ファイルへと保存し、OBJECTMODEではプラグイン情報をオブジェクト自身へと保存します。
シェーダー情報は直接LightWaveのオブジェクトファイルへと保存されますので、このスクリプト
ではOBJECTMODEを使用します。

'io' Agentを使用して、情報をオブジェクトファイルへと追加します。ファイル書き込みのコマ
ンドに関してはリファレンスマニュアル第１章：共通コマンドの章を参照してください。

load: what,io

{

if (what == OBJECTMODE)

{

pulserange = io.readNumber();

keepdiffuse = io.readNumber();

}

setdesc(prog + " is Pulsing on "+string(pulserange) + "
Frame Intervals");

}

load()関数はsave()関数とよく似ています。PULSE2.LSが適用されているオブジェクトが
読み込まれると、load()関数が呼び出され、save()関数を使用して保存しておいた情報を取得
します。今回の場合、この特定のサーフェイスに対し保存されていた pulserange と
keepdiffuseの値を読み戻します。

また、ユーザーにフィードバックを与えるための処理も加えましょう。create()関数内で
setdesc()が呼び出されるため、ユーザにはデフォルト値が表示されることになります。特定の
サーフェイスに対する値がデフォルト値と異なる場合、読み込み時には表示されません。オブジェ
クトからサーフェイスの値を読み出した後でsetdesc()を読み出せば、この問題を解決します。
ユーザーがインターフェイスを開かなくても、シェーダープラグインリストの中に適切な値が表示
されるようになります。

コードの最終部分は options()関数です。これはインターフェイス作成部分です。関数
options()は、ユーザーがシェーダー用のオプションポップアップから選択したときに呼び出さ
れます。

ここではインターフェイスの作成法については解説しませんが、2～ 3点指摘しておくことがあ
ります。

options

{

reqbegin(prog);

reqsize(314,69);

reqbegin()コマンドはインターフェイス開始を示し、残りのコントロールに対するステージ
を設定します。reqbegin()に渡す値はウィンドウのタイトル領域に表示する名称です。これが
スクリプト名称変数を作成した置いたもう一つの理由です。コードの一番上の箇所で変数の名称を
変更することでウィンドウの名称も同時に変更することが可能なのです。コード処理という意味で
は特別面白い箇所ではありませんが、更新時間を縮める利点があります。

c1 = ctlminislider("Pulse Range",pulserange,1,100);

ctlposition(c1,13,12);

c2 = ctlcheckbox("Keep Diffuse",keepdiffuse);

ctlposition(c2,201,12);

return if !reqpost();

pulserange = getvalue(c1);

keepdiffuse = getvalue(c2);

reqend();

setdesc(prog + " is Pulsing on "+string(pulserange) + "
Frame Intervals");

}

スクリプトのインターフェイス部分において指摘しておきたい最後の要素は終わり 2行にありま
す。reqpost()関数はユーザーが"OK"ボタンまたは"Cancel"ボタンを押したときにブール値を
返します。reqend()関数は実際にパネルを閉じるときに呼び出されます。

最後にもう一度setdesc()関数を呼び出して新しい値を表示設定しましょう。LightWaveの
CD-ROMにはPulse2.lsスクリプトが入っています。ここまで解説を読みながら少しずつコードを
入力してきた方のために、以下に完全なコードを提供します。間違いがないか、一通りチェックし
てみてください。

PULSE2.LS

// Pulse2

// サーフェイスの自己発光度と拡散レベルを(交互に)反復することで

// サーフェイスを"波動"のように見せます

//

// 更新日 04.21.98 Bob Hood

// 更新日 07.20.01 Scott Wheeler

@version 2.3

thisFrame;

pulserange = 2;

prog = "Pulse v1.01";

keepdiffuse = true;

create

{

scene = Scene();

pulserange = scene.fps;

setdesc(prog + " is Pulsing on " + string(pulserange) + "
Frame Intervals");

}

newtime: frame, time

{

thisFrame = frame;

}

flags

{

return(LUMINOUS,DIFFUSE);

}

process: sa

{

if(thisFrame == 0)

return;

j = thisFrame;

if(thisFrame > pulserange)

{

i = integer(thisFrame / pulserange);

if(i != 0)

j = thisFrame - (i * pulserange);

}

k = pulserange / 2;

if(j <= k)

{

// 明滅オン

sa.luminous = j / k;

if (!keepdiffuse) sa.diffuse = 1.0 - (j / k);

}

else

{

// 明滅オフ

j -= k;

sa.luminous = 1.0 - (j / k);

if (!keepdiffuse) sa.diffuse = j / k;

}

}

save: what, io

{

if (what == OBJECTMODE)

{

io.writeNumber(pulserange);

io.writeNumber(keepdiffuse);

}

}

load: what, io

{

if (what == OBJECTMODE)

{

pulserange = io.readNumber();

keepdiffuse = io.readNumber();

}

setdesc(prog + " is Pulsing on " + string(pulserange) + "
Frame Intervals");

}

options

{

reqbegin(prog);

reqsize(314,69);

c1 = ctlminislider("Pulse Range",pulserange,1,100);

ctlposition(c1,13,12);

c2 = ctlcheckbox("Keep Diffuse",keepdiffuse);

ctlposition(c2,201,12);

return if !reqpost();

pulserange = getvalue(c1);

keepdiffuse = getvalue(c2);

reqend();

setdesc(prog + " is Pulsing on " + string(pulserange) + "
Frame Intervals");

}

第１５章：イメージフィルタスクリプト 15.1

第15章 : イメージフィルタスクリプト

Image Filterスクリプトを使用すると、LightWaveのレンダリング出力に対し素晴らしい後処理
コントロールが出来ます。画像を反転するだけの簡単なスクリプトから、望みどおりの効果を得る
のに必要となる複雑なスクリプトまで作成可能です。

この章での最終目標は、Image Filterシステムの動作についての基本的な理解を深めることにあ
ります。ですから Image Filterコントロールの簡単な側面をより重視します。さらに高度なスクリ
プトはLightWaveのCD-ROMやNewTekのWebサイトにおいてありますので、これらのソースを
利用すると、Image Filterコントロールをより深く掘り下げていくのに役立つでしょう。

イメージバッファ
LightWaveからレンダリングされた画像には割り当てられた豊富なデータがあり、このデータは

バッファという一連のスポットに保存されています。各バッファには画像内の個別のエレメントに
関する情報が入っています。例えば、LUMINOUSバッファ内にあるデータを調べることで、ピク
セルの自己発光度を見ることが出来るのです。

このデータ全てが読取専用であるわけではありません。2～ 3のバッファに対してはデータが読
取専用とは限りません。後処理効果として振舞えるように特定のピクセルをコードで変更すること
が可能です。バッファされている情報を使用して後々合成ソフトなどで操作できるよう自己発光度
のピクセルだけを個別の画像として保存することも可能なのです。

Image Filter Object Agentで利用可能なデータメンバとメソッドの完全なリストについては、リ
ファレンスマニュアル第23章：Object Agent Referenceを参照してください。



Image Filter 関数
Image Filterスクリプトは、大半のLScriptのレイアウトスクリプトタイプと同じように設定しま

す。開発者が機能追加用に使用する事前定義関数がいくつか含まれています。以下、これらの関数
と Image Filterスクリプト内における役割を記していきます。

create() と destroy()

create()関数はスクリプトが最初に起動したときに呼び出されます。ユーザーが最初にスク
リプトを起動したとき、またはシーンから読み込まれた時にも呼び出されます。通常、変数の初期
化とコマンドの設定がここで行われます。

destroy()関数はユーザーによってアクティブな処理からスクリプトが除去されたとき、ま
たスクリプトが割り当てられていたシーンがクリアされた時に呼び出されます。またこの関数の中
で変数のクリーンアップを行う場合もあります。他の関数と同様、関数内に何のコードもなければ、
スクリプトに含める必要はありません。

process()

process()関数はスクリプトの中では馬車馬のような関数です。LScriptは各フレームがレン
ダリングされた後に自動的にこの関数を呼び出します。LightWaveの内部イメージバッファ（上記
参照）を使用することで、開発者はレンダリングされたフレームのイメージにアクセスし、ピクセ
ルを修正するコードを実行し、イメージにもう一度値を戻したり、後で使用するために他へ保存す
ることが容易に可能になります。これらのアクションは全てスクリプトのprocess()関数内で
起こることです。

process()関数は、Image Filter Object Agent内に含まれているデータメンバやメソッドを通
してフレームのバッファ情報にアクセスします。このObject AgentはLScriptから呼び出されると
いつでも自動的にprocess()関数へと渡され、Image Filterスクリプトで必要とされる情報を全
て保持しています。

process: ifo

{

}

引数ifoは Image Filter Object Agentです。このObject Agentを通して、利用可能なバッファ
の一つに情報を集めたり設定したり、またシーン特有のデータメンバを問い合わせたり、データを
新規バッファへコピーペーストが出来たりします。

load() と save()

load()とsave()関数を使うと、後々使用出来るようにシーンファイル内部に情報を保存し
ておくことが出来ます。オプションの状態、ユーザーカラーなど、後で読み込みなおしたい情報は
何でも保存可能です。

例えば、レンダー時にスレートを作成する Image Filterを作成したとします。shotの名称、show
の名称、takeなどを保存したくなるかもしれません。以下で情報を保存できます。

save: what, io

{

io.writeln(shot);

io.writeln(show);

io.writeln(take);

io.writeln(episode);

io.writeln(date);

io.writeln(slateframe);

}

save()関数はレイアウトから渡される二つの引数を受け取ります。最初の引数は、レイアウ
トがこの関数を現在呼び出している時点で二つの保存モードのうちのどちらかのモードを指定しま
す。値が SCENEMODEであれば、ユーザーはシーンの保存を選択したことになります。しかし
OBJECTMODEの場合には、スクリプトが適用されているオブジェクトを保存するように指示する
ことになります。

どのモードで作業するのかを把握するのは重要なことです。SCENEMODEで保存する値は文字
列（ASCII）形式、OBJECTMODEではバイナリの値が要求されます。本来、引数は SCENE
MODEの値を指定するのがほとんどです。OBJECTMODEでデータを保存するのは、シェーダース
クリプトを書くときのみです。他のLScriptクラスではオブジェクト自身に割り当てられるタイプ
はありません。

save()関数へ渡される引数ioはObject Agentです。これはシーンファイルに保存されるスク
リプトのデータ行にアクセスするために使用されます。writeln()メソッドを使用して文字列
データを保存するシーンファイルへと埋め込みます。いったん保存された情報から値を取り出すに
は、load()関数を使用してください。

load: what,io

{

shot = io.read();

show = io.read();

take = io.read();

episode = io.read();

date = io.read();

slateframe = integer(io.read());

}

構造上、この関数は前述のsave()関数と同じになります。load()はsave()で使用されて
いる引数と同じですが、情報を保存するio Object Agentを使用する代わりに、シーンファイルか
らデータを読むためにioのread()メソッドを使用します。

注意
load()と save()関数の双方において、データの割り当てや収集に使用される変数はグローバルス
コープとなります。

options()

options()関数は、プラグイン用のインターフェイスを作成します。このインターフェイスは
ユーザーがプラグインのプロパティを選択した時に表示されるものです。プラグイン用のインター
フェイス作成に関する詳細な情報は、このマニュアルの第24章：LSIDEを参照してください。

例：Black and White

作成する最初のスクリプトは本当に基本的なものです。画像からRGBの値を取得し平均化して
黒白画像を作成するものです。

まずはLScript Editorを開きます。エディターを使えば、スクリプトを簡単に作成できる素晴ら
しいツールをいくつも使用できます。これらのツールの一つはツールメニューの下にあるテンプ
レート機能です。テンプレートを選択し Image Filterをハイライトにします。

ready-to-useな状態にあるテンプレートを読み込み、そのテンプレートを使用して Image Filter
スクリプトを作成します。これでスクリプトの形式を覚えておく必要がなくなります。テンプレー
トが教えてくれるからです。しかし完成するまでにテンプレートからコードをいくつか変更したり
削除する必要があります。それでも基礎は出来上がるのです。

// ---------------------------

// LScript Image Filter template

//

デフォルトのスクリプトにある最初の数行はスクリプトのコメントや名称を挿入するための領域
です。//で始まる行はコメントです。この行の後にあるテキストは全てコマンドとして評価される
ことはありません。

以下の行を作成するスクリプトの名称で置き換えることにしましょう。

// ---------------------------

// Black and White

//

メインコメントの後に一連のコンパイラ指示子が続きます。この時点で私たちに関係する指示子
は@script指示子のみです。このコマンドはスクリプトがLightWaveに追加された時にスクリプ
トの種類をLightWaveが分類できるようにします。

@version 2.5

@warnings

@script image

ご存知の通り、次のグループの最初の行はコメントです。このコメントではユーザー定義関数
（UDF）外で定義されている変数は全てグローバルであり、スクリプト全体で利用可能であること
を示しています。画像の幅と高さの値をここで定義しますので、後々process()コードにおい
てこれらの変数や配列を定義しなくてはなりません。process()関数が呼び出される前に画像の
サイズを割り出すことは出来ないからです。

// ここに大域変数

大域変数を初期化した後は、スクリプトが開始また終了した時点で何をするのかをLScriptに伝
えるビルトイン関数を呼び出さなくてはなりません。

create

{

// ここで一度初期化を行います

}

create()関数はスクリプトが最初に起動されたときに呼び出されます。ユーザーから最初に
起動された時、またはシーンから読み込まれた時にも呼び出されます。ここでスクリプトの名称を
設定するのがよいでしょう。

create

{

setdesc(“Black and White”);

}

setdesc()コマンドは Image Filterプラグインリスト内のタイトル領域に配置するテキスト値
を受け取ります。デフォルトのLScriptタイトルを独自のタイトルに置き換えることで、よりプロ
フェッショナルなスクリプトに見えます。この例ではスクリプトに"Black and White"という名前を
付けましょう。どんな名前を付けても構いません。

理論上、destroy()はcreate()の後に来ます。スクリプトが処理を終えた後の大掃除をこ
こで行います。今回のスクリプトではこの機能は必要としませんので、スクリプトから以下の行を
削除することが出来ます。

destroy

{

// ここで気をつけて最終クリーンアップ処理を行います

}

前述の通り、process()関数は処理する画像に関する重要なデータをObject Agentの形式で
渡します。まずはじめに見たいのは Image Filter Object Agentのデータメンバwidthとheight
に表されている、レンダリングされた画像のサイズです。これらの値を使用してイメージ全体を処
理を確実にするための変数のループ処理をコントロールします。

process: ifo

{

// 画像の幅と高さの情報を取得します

imgWidth = ifo.width;

imgHeight = ifo.height;

}

集められたこのデータを使って、イメージ内の全ピクセルを通して処理出来ます。二つのfor
ループを作成し、処理を行います。一つは画像のX座標値を、もう一つはY座標値を反復処理しま
す。まずはYから始めましょう。

imgHeight = ifo.height;

// y座標値のループ処理

for(i = 1;i <= imgHeight; ++i)

{

}

...

次に Xの反復処理です。

for(i = 1;i <= imgHeight; ++i)

{

// x座標値のループ処理

for(j = 1; j <= imgWidth; ++j)

{

}

}

この二つのループを通して画像内のピクセル毎にデータを読み込んだり書き込んだりします。白
黒イメージを作成するために使用する数学の方程式は、至って簡単なものです。各ピクセルに対し
まずは赤、緑、青の三つの値を平均化し、ピクセルの合計値を変数avgに代入します。

for(j = 1; j <= imgWidth; ++j)

{

// 平均値を計算します

avg = (ifo.red[j,i] + ifo.green[j,i] + ifo.blue[j,i]) /
3;

}

平均値が決定すれば、avgの値を各チャンネルの値へと置き換えます。

avg = (ifo.red[j,i] +

ifo.green[j,i] + ifo.blue[j,i]) / 3;

// 平均値をカラーチャンネルに代入します

ifo.red[j,i] = avg;

ifo.green[j,i] = avg;

ifo.blue[j,i] = avg;

赤、緑、青の配列位置を返し、グレースケール値を作成します。

第１５章：イメージフィルタスクリプト 15.7



では以下に現在のコードを全て書き出してみましょう。

// ---------------------

// Black and White

//

@version 2.5

@warnings

@script image

create

{

setdesc(“Black and White”);

}

process: ifo

{

// 画像の幅と高さの情報を取得します

imgWidth = ifo.width;

imgHeight = ifo.height;

// y座標値のループ処理

for(i = 1;i <= imgHeight; ++i)

{

// x座標値のループ処理

for(j = 1; j <= imgWidth; ++j)

{

// 平均値を計算します

avg = (ifo.red[j,i] +

ifo.green[j,i] + ifo.blue[j,i]) / 3;

// 平均値をカラーチャンネルに代入します

ifo.red[j,i] = avg;

ifo.green[j,i] = avg;

ifo.blue[j,i] = avg;

}

}

}

進行モニターの追加
上記画像処理の例は極めて簡単な例ですが、新しい画像を計算するのはLScriptにとっては依然、

時間を要するものです。処理コードがより複雑になればなる程、最終画像までレンダリングするの
にさらに長い時間がかかるでしょう。ユーザーは処理がどの辺りまで来ているのか知りたいもので
す。レンダリング状況ウィンドウ上に完成までのパーセンテージを表示するための進行モニターを
表示することで、この情報を提供することが出来ます。

Black and Whiteスクリプトでは画像サイズを取得するすぐ後に、このコードを配置します。

imgHeight = ifo.height;

// 進行モニターを設定します

if(runningUnder() != SCREAMERNET)

moninit(imgHeight);

ここでやるべきことは二つあります。一つはこの画像がレンダーファーム内にレンダリングされ
ているのかどうかをチェックすることです。レンダリングされていれば、進行モニターを使用する
必要はありません。どのみち誰も見ることは出来ないのですから。レンダーファームにレンダリン
グされていない場合、moninit()関数へ画像の高さを渡します。

moninit()関数を呼び出すことにより、レンダー状況スクリーンへ進行状況を通知するスク
リプトを設定します。変数imgHeightをmoninit()へ渡してモニターの最大カウント値を決
定します。こうしてレンダー状況ウィンドウは、処理がどれくらいの割合で完成しているのかを決
定します。

例えば画像が 640x480であれば、変数imgHeightの値は 480となります。imgHeightを
moninit()関数へ渡すことにより、モニターの最大カウント値は 480となります。ですから 480
のうち12行目を通過すると、レンダー状況ウィンドウは正しい割合を表示することになります。

monstep()関数を呼び出すことでモニターの現在のカウント数を増やせます。画像の高さが
100%となるように設定していますので、この関数を高さループの終端に挿入します。そうすれば
高さループが1増えるごとにモニターの値も1ずつ増えることになります。

ifo.blue[j,i] = avg;

}

// 進行モニターの更新

if(runningUnder() != SCREAMERNET)

if(monstep())

return;

この関数は、モニターの現在のカウントを増やすだけでなくブール値も返します。値がtrue
であれば"Cancel"ボタンまたはESCキーが押されたことを示します。これによりユーザーは長時間
の処理をキャンセルできるようになります。

注意
進行モニターは全てに対し適用可能なわけではなく、レイアウトプラグイン構造によっては異なる動
作を行います。

第１６章：アイテムアニメーション 16.1

第16章 : アイテムアニメーション

LScriptの Item Animation（LS-IA）は、アニメーション中にオブジェクトの位置や回転、スケー
ル値を変更し、レイアウトで計算されていた元の値を上書きします。

process()

他の大半のLScriptと同様、Item Animationスクリプトの主な動作箇所はprocess()関数で
す。LS-IA用のprocess()関数は、引数を三つ受け取ります。

第 1引数'ma'はMotionAccessと呼ばれるObject Agentです。これはスクリプト内でユーザー定
義変数として表されます。スクリプト間の整合性を取るため、'ma'を使用することをお勧めします。
このObject Agentではスクリプトを適用するオブジェクトに割り当てられているメソッドやデータ
メンバ全てにアクセスが可能です。利用可能なメソッドの完全なリストについてはこのマニュアル
のメソッドの章を参照してください。

frameにはアニメーションの現在のフレームが入っています。

timeは各フレームに対し生成される浮動小数点数値のタイムインデックスが入っています。こ
のタイムインデックス値はフレーム番号と違いシーンのフレーム/秒設定に依存しています。

例：

シーン中で 24フレーム/秒に設定されている場合、フレームに対するタイムインデックスは以下
のようになります。

フレームタイムインデックス（24フレーム/秒）

1/24 もしくは .042

2/24 もしくは .083

3/24 もしくは .125

シーン中で 30フレーム/秒に設定されている場合、フレームに対するタイムインデックスは以下
のようになります。

フレームタイムインデックス（30フレーム/秒）

1/30 もしくは .033

2/30 もしくは .066

3/30 もしくは .1



LS-IA メソッド
MotionAccess Object Agentでは利用可能な一般的なメソッドに加え、Item Animation構造には

特別に二つのビルトインメソッドが用意されています。この二つのメソッドでは指定した時刻にお
けるモーションの属性を読み込んだり（"get"）、カレント時刻において属性を書き込んだり
（"set"）することが可能です。

get(attribute,time)

get()関数を使用すると、スクリプトが適用されているオブジェクトの位置データなどを取得
することが出来ます。get()関数から取得できる情報の種類は、要求する'attribute'のタイプ
に依存します。属性のタイプとしてPOSITION, ROTATIONそれにSCALINGが指定可能です。

例：

オブジェクトから回転情報を取得したい場合には、以下のようにget()関数を呼び出します。

TheObjectRotation = get(ROTATION,time);

get()関数の呼び出しに対しベクトルタイプを返します。戻り値は選択された属性の値を表す
三つの値<x,y,z>を一組としたものです。上記例では変数'TheObjectRotation'にはオブジェ
クトの回転に対する<h,p,b>なる三つの値を含むベクトルの値が返されます。

また個々の名称を使用することで、ベクトルの一つの値だけにアクセスすることも可能です。オ
ブジェクトのヘディングの情報だけが欲しい場合には、以下のようにアクセスします。

ObjectHeading = TheObjectRotation.h;

get()関数自身にも同じテクニックを適用すれば、特定のチャンネル情報をより早くより読み
やすい形で取得できます。

ObjectHeading = get(ROTATION,time).h;

注意
ROTATIONは度数で値を返します。POSITIONはメートル単位、SCALINGはパーセンテージと
して1.0は100%を表す値を返します。

timeは'attribute'の情報を要求する時点におけるタイムインデックスです。上記例ではカ
レント時刻における値を全て取得しています。しかしタイムインデックスを増減させることで、現
在時刻前後の値を取得することが可能です。以下のコードでは 1秒先のヘディング情報を取得でき
ます。

ObjectHeading = get(ROTATION,time+1.0).h

16.2 LScript 日本語ユーザーガイド



set(attribute,value)

set()関数では'attribute'（POSITION, ROTATION, SCALING）に対し指定したベク
トル値'value'を設定することが出来ます。これは現在時刻におけるオブジェクトのプロパティに
影響を及ぼします。

objID (読取専用)

このプラグインが割り当てられているオブジェクトのObject Agentのポインタです。

スクリプト作成例：

ROTATER.LS

Axis Rotation Control

以下の例では、モーションの特定の軸に沿って移動する距離に従い回転値を割り当てるスクリプ
トを調べていきます。このタイプのコントローラでは、ボルトの回転にナットを合わせるといった
使用法が考えられます。ナットがボルトの長さにあわせ上下に動くにつれ、ネジに沿って回転しま
す。

この例は非常に基本的なタイプのオブジェクトコントロールではありますが、LScriptの Item
Animationでオブジェクトのモーションをいかにたやすく拘束できるかを示すには大変好都合で
す。

//-----------------------------------------

// Axis Rotation Control

//

// Scott Wheeler

// 09/12/01

@version 2.3

@warnings

@script motion

全スクリプトと同様、上記部分は謝辞を述べたりコンパイラ指定用に予約されています。注釈の
指示子の一つに@nameがあります。@nameを使用すると、レイアウトで保持させるスクリプト
の名称を割り当てられます。@nameで指定しない場合には、スクリプトの保存名称が代用として
使用されます。

この場合、@name指示子がありませんから、スクリプトは単にRotaterと呼ばれます。小さな
点ではありますが、LightWaveプラグインをよりプロフェッショナルに見せてくれます。

rotspermeter = 1.0;

controlaxis = 1;

controlpos = 1;

スクリプトで使用する大域変数を宣言するには、コンパイラ指示子のすぐ後が良いでしょう。大
域変数はスクリプト内のどの場所からも見える変数です。この変数と同系の制限バージョンとして
ローカル変数があります。ローカル変数は、特定のUDF内部にあるローカルタスクを処理するよう
定義されます。

注意
ローカル変数はループ処理におけるカウンター変数として、また一時的な記憶値として使用される場
合が最も多いでしょう。

以下のアイテムは使用する大域変数です。

rotspermeterは1メートル動くごとの回転数を入れておく変数です、

controlaxisはオブジェクトが回転する軸、Heading, Pitchもしくは Bankです。
ctlchoice()コマンドを通してスクリプトインターフェイス上からユーザーに軸を選択し確定
される整数値です。ctrlchoice()関数はユーザーが選択した選択リストを表す整数値を返しま
す。

注意
ctrlchoice()についての詳細は、このマニュアルのインターフェイスの章にあります。

controlposはcontrolaxisと同様の変数です。回転の基となる位置移動軸を特定する
XYZを表す整数値です。

変数は出来るだけ実行する処理と似せて名前を付けるようにして下さい。この機能はスクリプト
に三つぐらいの変数しかなければそれほど重要ではありませんが、非常に多くの変数を持つスクリ
プトで作業する場合には大切なことです。後々スクリプトを更新する他の人にとってもスクリプト
がより読みやすくなります。

create

{

setdesc("Axis Rotation Control");

}

次のコード部分はスクリプトが作成時に実行されます。ですから関数名は'create'と名づけら
れているのです。

setdesc()コマンドはモーションオプションパネルでスクリプトの名称を表示することが出
来ます。このステップは絶対必要ということはありませんが、スクリプトをより洗練されたものへ
とするための特別なアイテムのうちの一つです。

setdesc()を使用しない場合、モーションオプションパネルではスクリプト名称を表示し、
LScriptとして識別します。

図 16-1. インターフェイスにおけるデフォルトの記述

setdesc()コマンドを使うと、デフォルトの代わりにウィンドウに読みやすい名称で置き換
えることが出来ます。

図 16-2. setdesc()コマンドでの記述

実をいうとスクリプト記述という大きな計画においてたいしたことではないのですが、こうする
ことでスクリプトには正当性とプロフェッショナリズム的な雰囲気が醸し出されるのです。

create()関数の中では変数へのデフォルト値の割り当てや、スクリプト追加時に一度だけ処
理される一般的な処理も実行できます。

process: ma, frame, time

{

rot = ma.get(ROTATION,time);

roth = rot.h;

rotp = rot.p;

rotb = rot.b;

pos = ma.get(POSITION,time);

大半のスクリプトと同様、process()関数では大量の処理が行われます。Item Animationの
場合、process()関数はレイアウトで時刻が変化するたびに呼び出されます。レンダリング中で
あってもプレビューが再生中であっても同様です。

今回のスクリプトでは、まず現在時刻においてオブジェクトが何をしているのかを見つけ出しま
しょう。変数rotはオブジェクトの回転情報が（ベクトルとして）代入されます。それからロー
カル変数 roth, rotp, rotbには、各々 rotベクトルの一部が代入されます。変数 posは
Motion Access Object Agentからオブジェクトの位置情報を格納します。

お望みの情報を手に入れるためget()メソッドを使用します。回転の個別のチャンネルに対し
影響を与えるようにしたいので、ROTATIONを三つのコンポーネントヘディング（roth）、ピッ
チ（rotp）、バンク（rotb）へと割り振ります。

スクリプトの後半にありますが、ROTATIONチャンネルを分けることで他の二つのチャンネル
の値を保護することが出来ます。そうすれば、影響を受けない二つのチャンネル上に存在するキー
フレーム付きモーションとスクリプトの拘束を受けるチャンネルとを合わせることが可能です。

if (controlpos == 1) rotamount = pos.x*(rotspermeter*360);

else

if (controlpos == 2) rotamount = pos.y*(rotspermeter*360);

else

rotamount = pos.z*(rotspermeter*360);

これで、受け取ったPOSITION情報を使用してオブジェクトの回転を修正することが出来ます。
しかし、まずは回転量をオブジェクトの位置に応じて割り当てることが必要です。以下の方程式を
使用して値を割り出しましょう。

Rotational Amount =  Object Positional Axis

Location * (Rotations Per Meter User Input * 360)

移動軸として使用する軸をユーザーに選択させますので、その選択軸に応じた方程式を計算する
用意をしておかなくてはなりません。上記のif()文ではXに対しては 1、Yに対しては 2、Zに対
しては3でコントロールします。

また変数 rotspermeterは 360が掛け合わされている点に注意してください。これで
"Rotation Per Meter"の入力値が度数へと変換されました。

if (controlaxis == 1) roth = rotamount;

else

if (controlaxis == 2) rotp = rotamount;

else

rotb = rotamount;

位置軸に対するif()文と同様、このif()文も割り当てられてた回転値を取得するのはどの回
転軸なのかを確定します。

ma.set(ROTATION,<roth,rotp,rotb>);

全て計算し割り当て終えたら、変更点をオブジェクトのモーションに再び割り当てなくてはなり
ません。set()メソッドを使用して処理しましょう。オブジェクトのROTATIONのみに影響を
与えたいので、ROTATIONだけにset()を行いましょう。

ROTATION情報はベクトルとして書き直している点に注目してください。つまりヘディングと
ピッチ、バンクの値は一組となっているのです。そのために変数を <と >で囲っているのです。
<>は三つの値を一つのベクトルとして結び付けます。ベクトルを作成する時にはその順番に注意
してください。間違えてもエラーチェックは効きません。もし間違えて変数の位置を変更してしま
うと、一つのチャンネル値が他のチャンネル値となってしまいます。例えば以下のように

ma.set(ROTATION,<rotp,roth,rotb>);

としてしまうと、ヘディングとピッチの情報が入れ替わってしまい、とんでもなく変な結果を引
き起こしてしまいます。

options

{

reqbegin("Axis Rotation Control");

c1 = ctlchoice("Rotation
Axis",controlaxis,@"H","P","B"@);

c3 = ctlchoice("Control Axis",controlpos,@"X","Y","Z"@);

c2 = ctlnumber("Rotations Per Meter",rotspermeter);

return if !reqpost();

controlaxis = getvalue(c1);

controlpos = getvalue(c3);

rotspermeter = getvalue(c2);

reqend();

}

options() UDFは、ユーザーがスクリプトのプロパティを尋ねる際のボタンやコントロール
を全て設定する箇所です。

インターフェイスの設計はこのマニュアルの別の章でカバーしていますが、上記のコードは以下
のようなインターフェイスを生成します。

図 16-3. options()で生成されたインターフェイス

load: what,io

{

if(what == SCENEMODE) // ASCII シーンファイルの処理

{

controlaxis = integer(io.read());

controlpos = integer(io.read());

rotspermeter = number(io.read());

}

}

save: what,io

{

if(what == SCENEMODE)

{

io.writeln(controlaxis);

io.writeln(controlpos);

io.writeln(rotspermeter);

}

}

load()とsave() UDFを使うと、シーン保存時に保持しておきたい変数を保存することが出
来ます。これによりシーンが再び読み込まれたときに、保持しておきたい変数が自動的に復元され
ます。この機能がなければシーンを読み込むたびに値を設定しなおさなければならず、スクリプト
の使い勝手がひどく制限されてしまいます。

以上が Item Animationです。上記の小さなチャンクでは全体として見通しづらい点もあるで
しょうから、以下にスクリプト全部を記しておきます。

ROTATER.LS

///-----------------------------------------
// Axis Rotation Control
//
// Scott Wheeler
// 09/12/01
@version 2.3
@warnings
@script motion
@name AxisRotationControl
rotspermeter = 1.0;
controlaxis = 1;
controlpos = 1;

create
{
setdesc("Axis Rotation Control");

}

process: ma, frame, time
{
roth = ma.get(ROTATION,time).h;
rotp = ma.get(ROTATION,time).p;
rotb = ma.get(ROTATION,time).b;
pos = ma.get(POSITION,time);
if (controlpos == 1) rotamount = pos.x*(rotspermeter*360);
else
if (controlpos == 2) rotamount = pos.y*(rotspermeter*360);
else
rotamount = pos.z*(rotspermeter*360);

if (controlaxis == 1) roth = rotamount;
else
if (controlaxis == 2) rotp = rotamount;
else
rotb = rotamount;

ma.set(ROTATION,<roth,rotp,rotb>);
}

options

{
reqbegin("Axis Rotation Control");
c1 = ctlchoice("Rotation

Axis",controlaxis,@"H","P","B"@);
c3 = ctlchoice("Control Axis",controlpos,@"X","Y","Z"@);
c2 = ctlnumber("Rotations Per Meter",rotspermeter);
return if !reqpost();
controlaxis = getvalue(c1);
controlpos = getvalue(c3);
rotspermeter = getvalue(c2);

reqend();
}

load: what,io
{
if(what == SCENEMODE) // ASCIIシーンファイルの処理
{
controlaxis = integer(io.read());
controlpos = integer(io.read());
rotspermeter = number(io.read());

}
}

save: what,io
{
if(what == SCENEMODE)
{
io.writeln(controlaxis);
io.writeln(controlpos);
io.writeln(rotspermeter);

}
}

第１７章：チャンネルフィルタ 17.1

第17章 : チャンネルフィルタ

Channel Filterスクリプトでは、ユーザーがアイテムモーションの個別のチャンネルに対しアニ
メーションを修正出来るようになります。このタイプのスクリプトはチャンネルの値に対し読み書
き両方においてアクセス可能です。チャンネル内で処理される調整を監視し作用します。

一般的なChannel FilterスクリプトはChannel Object Agentから値を読み取り、収集したデータ
に対しテストと計算を実行し、最終的には値を修正しチャンネルへと値を戻します。

LScriptに見られる大半のレイアウトスクリプト構造と同様、Channel Filterも、このスクリプト
クラスで使用可能なオプションを利用するために事前定義された関数をいくつか使用しています。
スクリプトを適切に動作させるためにこれら関数がいくつかは必要となりますが、他は単にオプ
ションの機能を提供しているだけです。

create() と destroy()

create()とdestroy()関数は全変数の初期化とスクリプトに対するクリーンアップ処理を
扱います。Channel Filterスクリプトタイプ用として、create()関数はスクリプトが適用されて
いるチャンネルから作成されたChannel Object Agentを受け取ります。

process()

process()関数は重要なチャンネル修正コードの大半が書かれている箇所です。この関数は
レイアウトから呼び出された時に三つの変数 ca, frameそれに timeを受け取ります。
process()はレイアウトでフレームデータが修正された時もしくは評価されたときに呼び出され
ます。

Channel Access (ca) Object Agentには、チャンネルに保存される値を認識し編集するために使
用可能なデータメンバ一つとメソッドが二つあります。他の構造特有のObject Agentと同様、
Channel Accessはprocess()関数の外側では作成できません。

Channel Accessのデータメンバnameにはチャンネルの現在の名称が保存されています。例え
ば、アイテムのXチャンネルに対しChannel Filterスクリプトを適用している場合、nameデータ
メンバには文字列"Position.x"が入っています。

get()とset()メソッドで、Channel Access Objectでアイテムのチャンネルに保存されてい
る数値を取得したり編集したり出来るようになります。get()関数は引数 timeを受け取り、
set()では単にチャンネルの現在のフレーム上で動作します。



process()関数への二番目（frame）と三番目（time）のパラメータはチャンネルが評価
される時刻を示しています。これらの値はユーザーインターフェイス内で表示されるシーンの現在
の時刻に一致するかもしれないし、しないかもしれないという点を覚えておいて下さい。

load() と save()

load()とsave()関数を使用すると、レイアウトはこのスクリプト操作に属する特定のデー
タを読み込んだり保存することが出来るようになります。スクリプトをどのように実行するのかを
確定する選択オプションや設定は、シーンファイル内部に保存されなければなりません。シーンが
保存されたり読み込まれる時に、このスクリプトが属するアイテムのチャンネル情報は破棄されま
す。

レイアウトがシーンを読み込み、スクリプト用のエントリが現れると、load()関数が呼び出
されます。この関数内では、想定どおりのスクリプト処理を可能にする設定や値を読み込み、割り
当てます。同様にsave()関数はユーザーがシーンを保存するときに呼び出されます。この関数に
は、シーンファイルにスクリプトのデータを系統立て保存するときに必要なコードが全て含まれて
います。

options()

options()関数はプラグインリストでスクリプトのプロパティをダブルクリックしたり、編
集を選択するたびに呼び出されます。そこにはスクリプトのインターフェイスを設定し実行するた
めに必要なコードが全て含まれています。

Channel Object Agent
create()とイベントコールバックUDF関数の双方が、引数channelを受け取ります。こ

のObject Agentにはキーフレーム情報の取得や設定、キーフレームの作成や削除、キーフレームの
スプライン情報の検索などを行うデータメンバやメソッドがいくつか用意されています。

注意
このObject Agentはアイテムの firstChannel()メソッドを呼び出すことでも作成可能です。

例1: maxValue

まず手始めに、チャンネルの値を強制的に最大値へ設定するスクリプトを作成しましょう。スク
リプトがアイテムチャンネルからオンに設定されると、チャンネルは事前に定義された値を越える
ことが出来なくなります。スクリプトのヘッダー情報から取り除いていきましょう。

@warnings
@version 2.5
@script channel
@name maxValue

ここでは新しく学ぶことは何もありません。プリプロセッサ指示子でLScriptに対しこのスクリ
プトで使用するオプションをいくつか設定しています。@script指示子の中でのchannel設定
を使用して、このスクリプトがChannel Filterであると定義している点に注目してください。

レイアウトにおける大半のスクリプト構造と同様、Channel Filtersは利用可能なお馴染みの関数
をいくつも持っています。まずはcreate()関数です。

create: channel
{
}

create()関数はスクリプトが最初に読み込まれたときに呼び出されます。ユーザーがチャン
ネルに対しプラグインリストからChannel Filterスクリプトを適用した時、またChannel Filterが適
用されているシーンファイル内のオブジェクトが読み込まれた時にも、この関数が呼び出されます。
通常、create()関数の中でスクリプトのキー変数をセットアップしたり、ユーザーエラーの
チェック、スクリプト解説文の設定を行います。ではここでスクリプト解説文の設定を行いましょ
う。

create: channel
{
// プラグインリストの解説文を設定します
setdesc(“maxValue”);

}

見ておわかりのとおり、create()関数は呼び出されるたびに、引数channelを受け取りま
す。この引数はスクリプトが適用されるチャンネルから作成されたChannel Object Agentです。
create()関数に対し簡単なinfo()関数を追加することで、テストしてみることが出来ます。

create: channel

{

info(“We have attached this to the: “,channel.name,“
channel.”);

// プラグインリストの解説文を設定します

setdesc(“maxValue”);

}

ここではChannel Obejct Agentのデータメンバnameをテストしてみました。このスクリプトを
アイテムのXチャンネルに適用している場合には、使用しているチャンネル名を告知するinfo()
リクエスタが表示されるはずです。

注意
Channel Object Agentで利用可能なデータメンバとメソッドの完全なリストについては、リファ
レンスマニュアル第26章：Channel Object Agentsを参照してください。

さて次に、このタイプのスクリプトにおいて最も重要となる関数、process()関数へと移り
ましょう。この関数はフレームのモーションデータが変更された時に呼び出される関数なので、関
数内にはフレーム特有のコードを全て書いておくようにして下さい。そうすればチャンネル情報が
修正された時点で、スクリプトは新しい値へと更新されます。では何が利用出来るのかを見ていき
ましょう。

process: ca, frame, time

{

}

見てわかるとおり、process()関数には三つの引数が渡されます。最初の引数caはチャンネ
ルのChannel Access Agentです。これは単一のデータメンバ（name）と二つのメソッド（set()
とget()）をエクスポートし、アイテムチャンネル内の値をやり取りすることが可能です。2番目
と 3番目の引数frameとtimeは単にユーザーが現在シーンにおいて編集している箇所を示して
います。では見てみましょう。

process: ca, frame, time

{

info(frame);

}

この小さなテストでは二つの作業を行います。まずはこれからLightWaveを無限ループに陥らせ
るので、実行する前に作業を保存するようにして下さい。2番目に、process()関数がどのくら
いの頻度で呼び出されるのかを正確に示します。スクリプト実行時に、関数が各フレームに対し定
期的に繰り返し呼び出されることがわかるはずです。OKボタンを押すことは出来ますが、他は何
も出来ません。スクリプトを中断するにはレイアウトを強制終了するしか手はありません。どのよ
うに動くのかを見た後は、コードからこの行を忘れずに削除するようにして下さい。

info()関数へ送られるデータを修正するだけで、process()関数へと渡される他の引数を
引き続きテストすることが可能です。渡された各引数によって異なる結果を表示しますが、
process()関数が呼び出される頻度は変化しません。いつ、どのくらいの頻度で関数が呼び出さ
れるのかを決定するデータは表示されません。それはChannel Filterスクリプト自身の設計だから
です。

明らかに、process()関数内でデータを表示するためにinfo()リクエスタを使用すること
は出来ません。スクリプトを変更する毎にレイアウトを終了させ再起動させるなんて絶対に嫌で
しょう。このタイプのスクリプトを学ぶ間に、可視化したいデータの表示法を考えつかないといけ
ませんね。常にチャンネルデータの値を取得するのではなく、処理するチャンネルに対し変更が加
えられた場合にのみ、表示するようにすればよいのです。捜し求めているのはマスターのイベント
ハンドラーと同じです。幸運なことに channel Obejct Agentはこのメソッドを持っています。

create: channel

{

channel.event(“newEvent”);

// プラグインリストの解説文を設定します

setdesc(“maxValue”);

}

使用するevent()はパラメータとしてユーザー定義関数（UDF）の名称を受け取ります。こ
の場合、スクリプトのチャンネルが修正された時点でスクリプト内部のnewEvent()関数が呼び
出されます。これをテストする前に、新しいUDFを定義しなくてはなりません。

newEvent: channel, event

{

info(event);

}

これでチャンネルが修正されるたびにnewEvent()関数が呼び出され、info()リクエスタ
は実際に何が起きているのかを表示するようになります。しばらくこの編集に時間をかけてみると、
イベントがいつ生成されたのか、何のイベント生成されたのかが理解できるようになります。また
newEvent()関数は渡されたイベントを取得するだけではなく、同様にChannel Object Agnetを
受け取ることに気が付いたことでしょう。

この例で作成しているスクリプトはチャンネルのevent()コールバック関数を使用しません。
イベントパラメータを試し終えたら、newEvent()関数を完全に除去し、同様にcreate()関
数内部におけるevent()メソッドの呼び出しも除去してください。

最終的にはこのスクリプトに対するインターフェイスが必要となりますが、今はまず望みどおり
に動作をスクリプトが実行出来るかについて心配しなくてはなりません。ただしこのスクリプトが
正しく動作するためには、インターフェイスが提供するデータも必要となります。そこでインター
フェイス作成に多大な時間をかけずに、大域変数を定義し初期値として与えることにしましょう。

@name maxValue

// Global variables:

maxValue = 1.0;

create: channel

{

スクリプトが正しく動作していることが評価され証明されたときには、インターフェイスを作成
しましょう。次にチャンネルの値をモニターするためにprocess()関数を設定していきます。

process: ca, frame, time

{

// Channel Access Object Agentからの値を取得します

currValue = ca.get(time);

}

get()メソッドにタイムインデックスを渡し、その時刻におけるチャンネルの値を取得します。
get()メソッドは単一の浮動小数点数値を返します。チャンネルの値を問い合わせるだけで、
キーフレーム全体の値をとるわけではないという点を覚えておいてください。上記例では返される
値は指定した時刻におけるアイテムのX座標値です。

それでは値が取得できたところで、変数maxValueの値を比較してみましょう。

process: ca, frame, time

{

// Channel Access Object Agentからの値を取得します

currValue = ca.get(time);

// 値を比較します

if(currValue >= maxValue)

ca.set(maxValue);

}

まずチャンネルの現在の値が変数maxValueに格納されている値以上であるかを比較してみま
す。以上であればChannel Accessのset()メソッドを使用して、チャンネルの値をmaxValue
の値に設定します。このようにしてチャンネルの値を編集しますが、maxValueよりも小さな値
であれば何も行いません。この設定は上限値設定を行っているのだと思って下さい。

この時点で実験してみましょう。

1 オブジェクトCowを読み込みます。

2 CowのXチャンネルにスクリプトを追加します。

3 maxValueの値を越える程度に、X軸に沿ってCowを動かしてみてください。

レイアウトでCowを動かすと、チャンネルの値がmaxValue以上の地点では、キーフレーム
を作成しても値を変更したりカレントフレームが更新されるとすぐにmaxValueに固定されるの
がわかるでしょう。

以下はこれまでのコードにmaxValue変数を設定するためのインターフェイスを追加したコー
ドとなります。

@warnings

@script channel

@version 2.5

@name maxValue

// 大域変数

maxValue = 1.0;

create: channel

{

// プラグインリストの解説文を設定します

setdesc("maxValue");

}

process: ca, frame, time

{

// Channel Access Object Agentからの値を取得します

value = ca.get(time);

// 値を比較します

if(value >= maxValue)

ca.set(maxValue);

}

例 2: dynLimiter

このスクリプトでは、先ほど作成したmaxValueスクリプトの動的バージョンを作成していき
ます。上記と同じ例を使用してはじめましょう。しかし変数内にチャンネルの最大値を設定するの
ではなく、他のオブジェクトのチャンネルの値を最大値として使用します。こうすることで時間の
経過とともに最大値をアニメーションさせることが出来ます。

先ほどの大域変数maxValueを、プリファレンスとして使用するオブジェクトの名称を表す
objNameに置き換えましょう。ここでは文字列値"Null"を使用します。この値はインターフェ
イスからも取得可能です。

// 大域変数

objName = “Null”;

create: channel

変数objNameには参照オブジェクトとして使用するオブジェクトの名称が入ります。まずは
process()関数内においてこの変数からMesh Object Agentを作成します。

process: ca, frame, time

{

// Object Agentを作成します。

obj = Mesh(objName);

// Channel Access Object Agentから値を取得します

Object Agentを作成しましたので、可変チャンネルに保存されている値を取得することが出来ま
す。先を続ける前に、Object Agentが実際に作成されたのかどうかを確認します。確認したら、参
照オブジェクト内に存在する最初のチャンネルからChannel Objectを作成することが出来ます。

if(obj)

{

// オブジェクトの最初のチャンネルを取得します

chan = obj.firstChannel();

}

else

error(“No Mesh Object Agent created.);

firstChannel()メソッドはMesh Object Agent内にある最初のチャンネルを返すだけです。
この例では一つのチャンネル、Position.Xだけに関心があります。今回の例では非常に幸運でした。
参照オブジェクト内に最初のチャンネルがPosition.Xである、引数caとしてprocess()関数に
送信されるチャンネル名称と一致するからです。しかし、常にこういう事態になるとは仮定出来ま
せん。

アイテムの全チャンネル名称を通して、process()関数へ渡されたチャンネル名称と一致す
るかどうかを確かめながら検索する必要があります。ですから二つの異なるオブジェクトの同じ
チャンネルを見ることになります。while()文を呼び出します。

// オブジェクトの最初のチャンネルを取得します
chan = obj.firstChannel();
// チャンネル名称 caと chanを比較します
while(chan && chan.name != ca.name)
chan = obj.nextChannel();

while()文を使うと、オブジェクト内の次のチャンネルから繰り返しChannel Object Agentを
作成することが可能です、どのチャンネルが開始地点となるのかを把握するにはどうすればよいの
でしょうか？最初のChannel Object Agentであるchanを作成するのには、firstChannel()
メソッドを使用しました。それ以降Mesh Object AgentのnextChannel()メソッドを呼び出す
と、オブジェクト内の次のチャンネルが返されます。例えば最初のチャンネルの値がPosition.Xで
あれば、次のチャンネルから返される値はPosition.Yといった具合に続いていきます。

評価する双方のチャンネル名称が一致した場合にのみ、while()文はスクリプトの処理続行を
許可します。二つのオブジェクト内部で同じチャンネルを比較するようにします。この場合におい
てChannel Obejct Agentであるchanが作成されたかを確認してみましょう。

if(chan)
{
}

今や、このスクリプトが正しく動作するために必要となる情報の多くを保持しています。リファ
レンスオブジェクトの名称（objName）やスクリプトのチャンネルから作成されたObject Agent
であるca Object Agentの値、リファレンスオブジェクトなどの情報です。あとはリファレンスオ
ブジェクトのチャンネルから値を取得するだけです。

前述の例において、値を比較しチャンネルの値を調節するコードは既に書いています。今回は上
記で開始したif()文内におけるコードを全て変更しましょう。

if(chan)
{
// 参照オブジェクトの Channel Object Agentから値を取得します
maxValue = chan.value(time);
// Channel Access Object Agent(このチャンネルにおける)から値を取得します
value = ca.get(time);
// 値を比較します
if(value >= maxValue)
ca.set(maxValue);

}

処理を多数行いますが、どれもみな非常に簡単なものです。まずvalue()メソッドを使用して
参照オブジェクトのチャンネル値を取得しました。次にca Channel Access Object Agent内に保存
されていた値を取得しました。それから二つの値を比較し、caの値がmaxValueを超えていれ
ば、set()メソッドを使用してcaにmaxValueの値を代入するだけです。

このスクリプトをテストしてみましょう。

1 Cowオブジェクトを読み込んでいるか確認してください。

2 "Null"という名称のヌルオブジェクトを追加します。

3 このスクリプトをオブジェクトCowへ適用します。

4 NullをX軸に沿ってキーフレームをつけます。

5 参照オブジェクトNullを越える箇所にCowオブジェクトをアニメーションさせます。

6 キーフレームを作成し、カレントフレームを変更してみます。Cowオブジェクトが参照オブ
ジェクトNullのX座標位置に固定されているのがわかるますね。

最終スクリプト:
@warnings

@script channel

@version 2.5

@name dynLimiter

// 大域変数

ObjName = “Null”;

create: channel

{

// プラグインリストの解説文を設定します

setdesc(“maxValue”);

}

process: ca, frame, time

{

// Object Agentを作成します

obj = Mesh(objName);

// Channel Access Object Agentから値を取得します

if(obj)

{

// オブジェクトの最初のチャンネルを取得します

chan = obj.firstChannel();

// caと chan Object Agentsのチャンネル名称を比較します

while(chan && chan.name != ca.name)

chan = obj.nextChannel();

if(chan)

{

// 参照オブジェクトの Channel Object Agentから値を取得します

maxValue = chan.value(time);

// Channel Access(このチャンネル) Object Agentからの値を取得します

value = ca.get(time);

// 値を比較します

if(value >= maxValue)

ca.set(maxValue);

}

}

else

error(“No Mesh Object Agent created.);

}

第１８章：ジェネリック（総括）スクリプト 18.1

第18章 : ジェネリック（総括）スクリプト

全スクリプトタイプの中で最も制限のないスクリプトが、LScript Generic（LSGN）です。スク
リプトタイプの大半がレンダー処理のある特定の側面に制限されていますが、Generic LScriptはそ
れら全ての外側に存在しています。スクリプトはレイアウト内部の様々な側面の多くにアクセス出
来るのです。

GenericスクリプトはモデラーのLScriptのような動作を行います。これらのスクリプトは呼び出
された時点で割り当てられる処理全体を完了する"run-once"操作です。LightWave環境内部から実
行するスタンドアローンプログラムのようなものですから、開発者はGeneric LScript内において、
他のレイアウトLScriptタイプでは利用できない機能を自由に実行することが可能です。

ジェネリックスクリプト構造
Genericスクリプトが"run-once（一回だけの実行）"イベントであるならば、プラグインの実行期

間内においてある特定の時間に特定の関数を呼び出すことは出来ません。ですからcreate()や
destroy()、init()関数は利用できません。Genericのスクリプトは任意のシーンやオブジェ
クトに制限されていないため、他のスクリプトタイプが使用可能なsave()やload()関数もま
た利用できないのです。

レイアウトのGenericスクリプトは、generic()関数という単一のエントリーポイントを持っ
ています。スクリプトはこの関数の呼び出しから始まり、関数の処理が完了した段階で終了します。

さらに単純化するため、options()関数はGenericのスクリプトでは利用できません。これは、
Genericではインターフェイスが持てないと言っているのではありません。インターフェイスコン
トロールは全て generic()関数の中に配置されることになるのです。スクリプト処理がこの
コードに到達した時点でリクエスタが表示されます。



LS-GN 関数
Genericのスクリプトには 2～ 3の関数が用意されています。これらの関数を使用して、レイア

ウトのシーンを管理することが出来ます。

loadscene(filename[,title])

loadscene()関数では、指定されたファイル名称をレイアウトのシーンファイルとみなして
読み込みます。オプション引数 titleはいったん読み込まれた後で内部的に保持されるシーン
ファイル名称として使用されます。このオプション名称がシーンの新規ファイル名称となります。
このパラメータが省略されている場合には、引数filenameが代用として使用されます。

savescene(filename)

savescene()関数は、レイアウトにあるカレントのシーンを、指定したファイル名称
filenameで保存します。

例： INCREMENTSCENE.LS

GenericのLScriptでは基本的にスクリプト関数は制限されていませんから、ここで解説していく
例はどれも、その小規模な範囲以上のものをカバーすることは不可能です。それを頭に入れた上で、
Genericのスクリプトタイプを使用してLightWaveに追加可能な機能の種類を表示する短いスクリ
プトを見ていきましょう。

以下のスクリプトはLightWaveに読み込まれたシーンを受け取り、スクリプトを実行するたびに
バージョン番号を増やしながらシーンファイルを保存します。

//-----------------------------------------

// Incremental Scene Saver v1.0

//

// Scott Wheeler

// 08-03-01

@version 2.3

@warnings

@script generic

@name LW_IncrementalSceneSaver

スクリプトの最初の部分は主としてコメント領域です。これについての詳細は、このマニュアル
の第１４章：プロシージャルテクスチャを参照して下さい。

スクリプトラベルの後にはプラグマ指示子が来ます。このスクリプトを読みやすくするため、指
示子は全てスクリプトの一番上に置いておきます。プラグマ指示子は行頭にある文字"@"によって
認識され、その後に指示名称と指定可能な引数が続きます。

読み込まれたシーンの名称を変更する前に、現在のシーンの名称を把握しておく必要があります。
シーン特有の情報にアクセスするためScene Object Agentを使用します。Scene()コンストラク
タを呼び出し、現在のシーンからScene Objectを作成することが出来ます。

scene = Scene();

変数sceneに入っているScene Object Agentには、データメンバがいくつもあります。データ
メンバの一つにfilenameがあります。データメンバfilenameにはカレントシーンの完全な
パス付きファイル名称の情報が入っています。ディスクにシーンを保存するためには、この名称が
必要となるのです。

例：

X:\Scenes\MyProjects\ThisScene.lws

特定のデータメンバを名称で要求することでScene Object Agentである'scene'の外へと情報を
抜き出します。Object Agentコンテナ（この場合はscene）の終端にメンバnameを追加します。

例：

scene.filename // Scene Object 'scene'のファイル名称を返します

scene.polycount // Scene Object 'scene'内にあるポリゴン総数を返します

変数'scene'にMesh Object Agentが含まれている場合には、データメンバpolycountにア
クセスすると、オブジェクト内のポリゴン数を返します。

generic

{

scene = Scene();

if (scene.filename == "(unnamed)")

SaveSceneAs();

SaveSceneAs(setupscene(scene.filename));

generic()関数の開始行において、現在のLightWaveシーンに実際にその名称が割り当てら
れているのかをチェックします。if()文内において Scene Object Agentから直接情報にアクセス
しています。こうするとコード内で必要とされる行数は圧縮出来ますが、スクリプトは理解しづら
くなりますね。

では、if()文が実際にはどのような動作を行うのかを見ていきましょう。LightWaveは保存さ
れていないシーンに対し"(unnamed) "という名称を自動的につけます。ですから名称に
"(unnamed)"と設定されていないかをチェックし、それに応じて処理を行います。

シーンが保存されていない場合、Command SequenceコマンドであるSaveSceneAs()を呼
び出します。Command Sequenceコマンドは直接LightWaveの内部関数とリンクしています。これ
らのコマンドにより、スクリプトはLightWaveのインターフェイスと内部関数双方をさらにコント
ロールできるようになります。ただしCommand SequenceコマンドはGenericとMasterクラ
ススクリプト以外ではアクセス出来ません。

SaveSceneAs()は、保存するシーンのファイル名称を要求します。まだシーン名称を決めて
いないので、名称を空にしたままでコマンドを呼び出します。するとLightWaveはユーザーに対し
"Save As"ファイルダイアログを開きます。その後は最初からシーンにファイル名称が付けられ
ていたかのように処理されます。

今回SaveSceneAs()コマンドへと渡す情報は、シーン名称変更用に作成された関数の戻り
値です。このユーザー定義関数（UDF）はsetupscene()から呼び出され、SaveSceneAs()
へ渡す変更後のシーン名称を返します。SaveSceneAs()内でsetupscene()を呼び出し入
れ子状態にすることで、ここでもコードの行数を削ります。

シーン名称を変更するための別個の関数を使用し、コードを一部切り離すことでより読みやすく
することも出来ます。UDFではある一つのセクション内に共有化する機能の領域を配置することが
可能です。そうすればシーン名称を処理するコードは全てsetupscene()関数内にありますの
で、コードの修正がはるかに容易になります。異なるセクションを探すためにメインコードを通し
て検索する必要がなくなるのです。

これほどまで極端にスクリプトを小さなサイズへとするのは賢いやり方だとは思えないかもしれ
ませんが、あらゆる状況においてコードをより読みやすくするための方法を探し出していくにはと
てもよい実習です。いずれは後々変更を加えないかもしれませんし、ある期間を経た後スクリプト
へと戻っていくかもしれません。どちらにしろコードを読みやすくしておくほうが良いのです。

もう一つ、独自の関数を作成するとコードのローカライズを行うことで何度も使用出来るという
利点があります。コード自体を反復させるのではなく、関数を一つ作成しスクリプト内部で何度も
関数を呼び出すことが出来ます。

setupscene: sceneName

{

}

UDFは処理中に使用するパラメータ群を受け取ることが可能です。今回の場合では、scene
Name というパラメータを一つ渡すことになります。generic()関数内で見たとおり、
setupscene()へはscene.Finenameのコンテンツを渡します。それからこの値はローカ
ルで使用するために変数sceneNameへと代入します。

cropsize = (sceneName.size()) - 4;

sceneName = strsub(sceneName,1,cropsize);

// .lwsを切り取ります

sceneNameには読み込むシーンのフルのファイル名称が入っているため、拡張子".lws"も含ま
れています。新しいバージョン番号をシーンファイル名称の終端につけたいのですが、拡張子
".lws"の後ろにつけたくはありません。ですから処理を行う前にファイル名称から".lws"を切り取る
必要があります。

まずは拡張子".lws"を取り除いたファイル名称の大きさを割り出します。LScriptのビルトイン変
数メソッドで、この処理が簡単になります。size()メソッドは全ての変数に対し利用可能であり、
変数のサイズを返します。ファイル名称のような文字列の場合、名称内にある文字数を返します。
全体の長さから 4引くことで、拡張子".lws"を除いたファイル名称の長さを取得します。

ファイル名称の長さがわかったので、拡張子を取り除くことが出来ます。LScriptでは文字列処
理とこれらの文字列の部分抽出を行うコマンド群を提供しています。これらコマンドの一つに
strsub()コマンドがあります。strsub()では文字列を受け取り、引数として渡した切り取り
用変数に基づいて切り取られた文字列のサブセットを返します。

例：

strsub()は以下のパラメータを受け取ります:

strsub(<sceneName>,<first character position>,<length>)

ファイル名称が以下の場合

sceneName = “X:/myscenes/scene.lws”.

ファイル名称の総文字列は21です。最後の4文字を除去したいので文字数は21-4=17になります。
ですから最後4文字を取り除くstrsub()への呼び出しは以下のようになります。

strsub(sceneName,1,17);

これは"文字列sceneNameを受け取り文字列の最初の文字から 17文字を抽出する"と解釈され
ます。名称全体は21文字の長さですから最後の4文字が返されないままとなります。

拡張子".lws"が取り除かれファイル名称にバージョン番号を付加できるようになりました。

if (strsub(sceneName,cropsize - 4,1) == "_")

第１８章：ジェネリック（総括）スクリプト 18.5



スクリプトはファイル名称に"_vXXX"を追加します。このxxxの個所には番号が入ります（たと
えばv001,v002）。4文字の固定した値ですので、ファイル名称内にバージョン番号が存在するかど
うかが確認できます。ファイル名称終端からの4文字めはアンダースコア（_）となります。

これを頭に入れた上で、if()文は終端からの4文字が"_"であるかどうかをチェックします。

ver = integer(strsub(sceneName,cropsize - 2,3)) + 1;

終端からの 4文字が"_"であれば、カレントシーンの終端にバージョン番号が存在しています。
シーンを保存する前にバージョンの値がわかれば、バージョン番号を上げていくことが出来ます。

既にstrsub()コマンドを使用していますので、文字列から一連の文字を抜き出す方法はわ
かっています。integer()コマンドを使用して組み合わせていきましょう。文字列値は
integer()コマンドを通して値を渡すことにより整数へと変換が可能です。これでカレントシー
ンの整数バージョン番号は文字から整数値へと変換されます。その後バージョン番号に 1を加算し
て、バージョン番号を上げていきます。

sceneName = strsub(sceneName,1,(sceneName.size()) - 5));

このコード行は既存の"_vXXX"文字列をファイル名称から除去し、新しいバージョン番号を適
用させる準備を行います。

else

ver = 1;

if()文の'else'部分は、終端からの 4番目の文字が"_"でない場合、つまりバージョン名称が
シーンに割り当てられていない場合に処理を行います。この場合にはシーンを保存する前にバー
ジョン番号が'1'に設定されているかどうかを確認する必要があります。

pad = "_v" + ver.asStr(3,true);

可変メソッドasStr()を使用してデータを整数タイプから文字列タイプへと変換します。最初
の引数(3)を渡すことで、文字列の長さを特定しています。こうすると文字列は常に文字数 3となり
ます。

return((sceneName + pad + ".lws"));

setupscene()関数を終了させるには、調整したシーンを保存可能にするためreturn()
文が必要となります。

return()文の中で切り取られたシーン名称scenenameは新規バージョン番号を受け取り、
再び拡張子".lws"をつけます。

スクリプトを保存しレイアウトに読み込んだ後、スクリプトをキーストロークに割り当てておけ
ば、キーを押すだけでバージョン付きシーンを保存することが出来ます。

INCREMENTSCENE.LS

//-----------------------------------------

// Incremental Scene Saver v1.0

//

// Scott Wheeler

// 08-03-01

@version 2.3

@warnings

@script generic

@name LW_IncrementalSceneSaver

generic

{

scene = Scene();

if(scene.filename == "(unnamed)")

SaveSceneAs();

else

SaveSceneAs(setupscene(Scene().filename));

}

setupscene : sceneName

{

cropsize = (sceneName.size()) - 4;

sceneName = strsub(sceneName,1,cropsize);

//  .lws を切り取ります

if (strsub(sceneName,cropsize - 4,1) == "_")

{

ver = integer(strsub(sceneName,cropsize - 2,3)) + 1;

sceneName = strsub(sceneName,1,(size(sceneName) - 5));

}

else

ver = 1;

pad = "_v" + ver.asStr(3,true);

return(sceneName + pad + ".lws");

}

第１９章：マスタークラス 19.1

第19章 : マスタークラス

LScriptのMaster Class（MC）スクリプトは、開発者に対し全く異なるレベルの機能を提供し
ます。これらのスクリプトではレイアウトのCommand Sequence（CS）コマンドが実行でき
（Genericスクリプトにおける処理と同様）、またユーザーによってアクションが更新できるという
独自の能力があります。オブジェクト選択やフレーム変更、設定の調整といったアクションは、
ユーザーがアクションを実行する際にCSコマンドへと全て変換されています。MCスクリプトはこ
れらコマンドがいつ発生しているのかを監視するだけで、特定の状況になった時点でアクションを
更新するようコード化することが出来るのです。

ユーザーがアニメーションや設定編集、シーンのレンダリングを行っている一方で、バックグラ
ウンドではアクティブなMasterスクリプトはアクションの記録と更新を連続的に実行しています。
MCスクリプトをこんなにも強力なツールにしているのはこの機能なのです。レイアウトでアニ
メーションさせている時にアニメーターを手助けするツールが書ければ、レイアウトのユーザーイ
ンターフェイスを拡張することが出来ます。ツールの選択からマクロの記録まで、MCスクリプト
の能力にはほとんど際限がありません。

追加されているこの機能は信じられないほど強力な機能ではありますが、スクリプトは常に実行
可能な状態にしておかなくてはなりません。優れたMasterスクリプトは、アニメーション中にユー
ザーが挿入する多くの状況を扱えるようにしてあります。より大切なことは、これらのスクリプト
は常時実行しているわけですから、コードを速く効率的に実行しなくてはならず、レイアウトの
ユーザーインターフェイスのインタラクティブ性が損なわれないよう注意しておかなくてはならな
いということです。コード作成におけるこの側面を見落としてしまうと、便利なMCスクリプトを
作るという目標が大きく覆されてしまいます。



サポートされている関数
MCスクリプトではレイアウトスクリプト内にある一般的な関数、create()やdestroy()、

save()、load()それにoptions()に加え、flags()関数が使用できます。

flags()

flags()関数はレイアウトから使用され、MCスクリプト実行時のオプションを設定します。
スクリプトが初期化された度にLScriptから自動的に呼び出されます。一度でもflags()関数が
呼び出されればスクリプトの設定をLScriptに再通知する必要はないので、もう一度スクリプト内
部で呼び出されることもありません。

SCENEはレイアウトに対し、このスクリプトをシーンと共にクリアするよう指示します。
flags()関数が定義されていない場合はデフォルト設定となります。

LAYOUTを指定すると、シーンがクリアされようと新規シーンが読み込まれようと、MCスク
リプトは読み込まれ起動状態のままでいることが可能です。

例：

flags
{
return(SCENE);

}

process()

Master Classスクリプトにおけるprocess()関数は、レイアウトが新規コマンドを発行する
たびに自動的に呼び出されます。この関数が呼び出されると、シーン内でユーザーが実行した処理
をスクリプトが正確に確定出来るよう、レイアウトから二つの引数を受け取ります。

引数 eventは process()関数へ送られるイベントの種類を返します。現在この引数には
NOTHING, COMMAND, TIME, SELECTもしくはRENDER_DONEという定数値が返されます。

文字列 commandには、レイアウトが一番最近発行したコマンドが引数付きで入っています。
返される文字列には個々のデータを区別するためスペースが使用されています。

例：

process: event, command
{
info(command);

}

アイテムが選択された場合、process()関数のinfo()リクエスタは以下のように状況を表
示します。

“SelectItem 10000000”

例 1：masterTest.ls

Master Classのスクリプトは今まで解説してきたスクリプトタイプの機能とは異なっていますの
で、最初のスクリプトでは新しい機能についてのみを解説していくようにしましょう。最初のスク
リプトは単純なものですが、二つの効果をもたらしてくれます。一つはMasterスクリプトの構造が
他のスクリプトタイプとどのように違うのかを把握しやすくしてくれること、もう一つは完成版ス
クリプトを作業ツールとして実行できるようになることです。このツールを二番目の例として使用
し、コマンドをより理解できるように解説していきます。

masterTest.lsスクリプトは、レイアウトのコマンド履歴ウィンドウと似たような動作を行います。
発行されるレイアウトコマンドの動作はスクリプトで捕らえられ、info()リクエスタに表示され
ます。どのアクションも表示されると言いましたが、各々のアクションという意味です。ですから
スクリプト実行が決断されると、行った特定の動作は全てinfo()リクエスタを表示します。言う
までもないことですが、これは実用的なスクリプトではないものの、目的は達成できます。

ではMaster Classスクリプトにヘッダーを追加していきましょう

//

// masterTest.ls: レイアウトから Master Classスクリプトへと

// 送信されたコマンドを紹介するスクリプト

//

@version 2.3

@warnings

@script master

@name masterTest

何も新しいことは行っていません。短い解説文、バージョン番号、警告レベル、作成するスクリプ
トの種類、スクリプトの名称などを指定しています。一般的なスクリプトの開始部分と同じです。

前述の通り、Master Classスクリプトはレイアウト内でユーザーがコマンドとして実行する動作
を監視することが出来ます。GenericスクリプトではなくMaster Classスクリプトを使用する際の
主な利点の一つが、この機能にあります。しかしユーザーからの情報を取得する前に、まずスクリ
プトとレイアウト間の関係を定義しておく必要があります。flags()という関数の中でその処理
を行います。

Flags()関数は唯一、LScriptから内部的に使用される関数です。コードの任意の場所から手
動でこの関数を呼び出すことはありません。この関数は、スクリプトが適用されているシーンがク
リアされたときに、レイアウトがスクリプトをどのように扱うかを設定するための関数です。

Master Classスクリプト起動時には、LScriptはスクリプトのコードを通して自動的にこの関数
を検索し、定義されているflag()関数を呼び出し、レイアウトとスクリプトの関係を確定します。
この関数にはreturn()文が含まれており、レイアウトのオプションを設定するために使用され
ます。前述通り、この関数には二つのオプション、SCENEとLAYOUTが利用出来ます。

以下の例では、このスクリプト用のflags()関数がどのように見えるのかを示しています。

flags

{

// レイアウトのイベント全てを監視するためのフラグを設定します

return(SCENE);

}

ここまで見てきたレイアウトスクリプトの大半と同様、スクリプトの動作のほとんどが
process()関数内で処理されます。Master Classスクリプトの一部として、process()関数
のジョブはユーザーからコマンドが発行される度にレイアウトから呼び出されます。スクリプトの
process()関数ではレイアウトから自動的に渡される二つの引数eventとcommandを受け
取ります。以下の例で関数を解説していきます。

process: event, command

{

// 二つの渡された引数の値を表示します

}

目標を達成するには、info()関数へと値を入れるだけです。process()関数がレイアウト
から呼び出されるたびに二つの値を表示します。

...

// 二つの渡された引数の値を表示します

info(event, " ", command);

...

このスクリプトを実行してみると、何かをクリックしたり、値を修正したり、フレームを変更す
る度にinfo()リクエスタが現れ、eventの数値や使用されているコマンド、コマンドで使用さ
れているパラメータ（もしあれば）が表示されます。引数 eventから返される整数値は定数
NOTHING, COMMAND, TIME, SELECTまたはRENDER_DONEを表しています。この表示デー
タによって何がレイアウトコマンドなのか、そうでないのかをよく表してくれます。さらに重要な
のは、process()関数がアニメーション時にどのくらいの頻度で呼び出されるのかがわかること
でしょう。アニメーション中にこのスクリプトを使用してみると、大量のinfo()リクエスタが表
示されてしまい、物凄くいらいらしてしまいますね。

このためprocess()コードで莫大な処理時間を浪費したくはないと思うでしょう。現在ユー
ザーが処理を行う度にinfo()リクエスタを手動で閉じなければならないため、スクリプトの生産
性が著しく落ちているような状態です。しかしinfo()文ではなく、プロセッサ集中型コマンドが
多くある場合にも、レイアウトとの対話は遅くなってしまいます。これを実演するため、info()
コマンドを以下のコードで置き換えてみましょう。

...

for(i = 0; i < 100000; i++)

{

}

...

このコードではLScriptに時間ループ処理を挿入しています。勿論、forループ文にはコマンド
や関数は何も割り当てられていません。何の処理も行いませんが、時間だけはかかります。時間が
かかることで、ループもまたレイアウトが何かの処理を行うのを妨害してしまいます。レイアウト
がprocess()関数を呼び出す度に瞬間、感知できる程度の一時停止を引き起こします。

注意
一時停止を生成する時間の長さは、実際にはコンピューターの処理速度と forループの実行時間に依
存しています。

スクリプトを実行し、画面上の周りをヌルが移動するシーンを作ってみてください。ループの処
理がレイアウトのインタラクティブ性をどれほど遅らせているのかが体感できるでしょう。レイア
ウトがprocess()関数を呼び出すたびにタイムループ処理を実行しているからです。では以前
のようにforループをinfo()文へと置き換えます。

ここに最終的なコードがあります。

//

// masterTest.ls: レイアウトから Master Classスクリプトへと

// 送信されたコマンドを紹介するスクリプト

//

@version 2.3

@warnings

@script master

@name masterTest

flags

{

// レイアウトのイベント全てを監視するためフラグを設定します

return(SCENE);

}

process: event, command

{

// 二つの渡された引数の値を表示します

info(event, " ", command);

}

このスクリプトを保存し取って置いてください。後ほど例 2で使用します。

例 2：selected.ls

次に取り組む例は、インターフェイス内で選択されているアイテムを表示するだけのものです。
このスクリプトでは、レイアウトからのコマンドを監視し選択状態を更新することで、Master
Classスクリプトについて既に学んだことを広げていきます。具体的には、選択アイテムを扱うコ
マンドを探していきます。

この例では、process()関数でスクリプト処理の大半を実行します。しかしこのスクリプト
の最終版ではユーザーインターフェイスが必要となります。このマニュアルの章全体でLScriptに
おけるインターフェイス作成の議論にささげてきました。ですから何が起きているのかを詳細に解
説するよりも何をしているのかだけをなぞるだけにします。インターフェイスについては、この後
の"インターフェイス"の章、および"LSIDE"の章でカバーします。

スクリプトの設定

まずはヘッダー情報から始めましょう。

//

// selected.ls - 現在選択されているアイテムを

// 列挙する Master Classのスクリプト

//

@version 2.2

@warnings

@script master

@name selected

ではこの関数で使用する関数を挿入しましょう。最終的には、これらの関数にはさらにコードが
増えることになりますが、現段階ではスクリプトの設定だけです。

create

{

// ここに初期化作業が必要となります

}

flags

{

// この関数ではシーンがクリアされた場合に

// レイアウトにスクリプトをクリアする必要があるかどうかを指示します

return(SCENE);

}

process: event, command

{

// ここではレイアウトが送信するコマンドを調べます

}

前例でinfo()リクエスタを除去して記述したスクリプトによく似ていますね。何度スクリプ
トを実行してみても何も起こりませんが、この時点では何もない状態で正しいのです。

次にユーザーが現在何を選択しているのかを調べる方法を設定します。アイテムの選択などに関
しては、Scene Object Agentを作成すればシーン設定を取得することが可能です。これは一度だけ、
しかもスクリプト内のかなり速い段階で処理しておきたいので、create()関数の中に追加する
ことにしましょう。この関数はスクリプトが呼び出されたときに自動的に実行されます。

create

{

// Scene Objectを作成します

sceneObj = Scene();

}

sceneObj Object Agentをcreate()関数内部においてローカルに宣言していますので、ス
クリプトにおいて他の関数内ではそのデータにアクセスすることは出来ません。ローカルではなく
グローバルで変数を宣言しなくてはなりません。そのために@nameプラグマとcreate()関数
の間に変数の宣言文を挿入します。

...

@name selected

// ここで大域変数の宣言

sceneObj;

create

...

この変数はグローバルとして宣言されましたので、スクリプト内部にある関数は全てObject
Agentのメソッドとデータメンバを共有できます。

注意
Scene Object Agentで利用可能なデータメンバとメソッドの完全なリストについては、リファレ
ンスマニュアルの 第11章：Scene Object Agentを参照してください。

作成されたsceneObj Object Agentを使えば、getselected()メソッドを使用してユー
ザーが現在選択している全アイテムを表すObject Agent（Mesh、Light、Camera）の配列を
取得することが出来ます。process()関数内に以下のコードを追加します。

...

process: event, command

{

// 現在選択されているオブジェクトを取得します

If(event == COMMAND)

{

selItems = sceneObj.getSelect();

...

まずは引数eventが定数COMMANDを返しているかどうかを確かめましょう。こうすれば監視
する必要のない他の種類のイベント全てを除くようになります。作成されたこの配列で様々な
Object Agentを反復することが出来、forループを使用してメソッドやデータメンバにアクセスが
可能になります。

...

{

selItems = sceneObj.getSelect();

for(i = 1; i <= selItems.size(); i++)

info(selItems[i].name);

}

}

このforループ文は、1番目のインデックス(1)からselItems[]配列のサイズをカウントし
ています。配列のサイズは格納されているアイテム数により確定されます。選択アイテムが三つあ
る場合には、三つのアイテムがselItems[]配列に保存されています。ですから配列のサイズは
3となります。この文を設定すれば、forループ文を使用して配列内の各インデックスを通じて、
メソッドやデータメンバにアクセスできるようになります。

現時点では簡単にするため、forループ文にはinfo()関数だけをつけています。このコード
行はレイアウトから取得するデータの種類が何であるかを見るためだけの一時的なものです。これ
で正しい軌道上にいるのかどうかを確定することが出来ます。スクリプトを組み始めたばかりの頃
は、これらの状況判断を設定するのが非常に重要なことなので、スクリプトが想定したとおりに動
いているかどうかもわからないまま様々な処理は出来ません。数行もしくはスクリプト全体をデ
バッグするよりも、バックアップを取り誤っている 1行だけを修正する方がはるかに簡単なのです。

ここにコードを紹介します。

//

// selected.ls - 現在選択されているアイテムを

// 列挙する Master Classのスクリプト

//

@version 2.2

@warnings

@script master

@name selected

// ここで大域変数の宣言

sceneObj;

create

{

// Scene Objectを作成します

sceneObj = Scene();

}

flags

{

// この関数ではシーンがクリアされた場合に

// レイアウトにスクリプトをクリアする必要があるかどうかを指示します

return(SCENE);

}

process: event, command

{

// 現在選択されているオブジェクトを取得します

selItems = sceneObj.getSelect();

if(event == COMMAND)

{

for(i = 1; i <= selItems.size(); i++)

info(selItems[i].name);

}

}

スクリプトのテスト

現在のシーンをクリアし、Masterプラグインパネル上のプラグインリストへMaster Classスク
リプト"selected.ls"を追加して下さい。

図 19-1. Master プラグインパネル

1 Cameraアイテムを選択します。info()リクエスタが現れ、カメラの名称を表示します。

2 OKボタンを押しパネルを閉じます。

3 Lightアイテムを選択します。同様のリクエスタが出てきますが、今度はライトの名称が表示さ
れます。

これらのステップでは、単一のアイテムが選択されているときにスクリプトが動作するかどうか
をテストしています。スクリプトは予測どおりに動いているように見えますが、これで終わりでは
ありません。複数のアイテムが選択されている状態でスクリプトがうまく動くのかどうかをテスト
してみましょう。カレントシーンに 2～ 3個オブジェクトを読み込んでみるのが一番簡単な方法で
しょう。

1 最初のオブジェクトを選択します。選択しているオブジェクトの名称が表示されているリクエス
タ一つ現れます。

2 OKボタンを押してパネルを閉じます。

3 SHIFT選択で二番目のオブジェクトを選択します。もう一度選択されているオブジェクトを表示
するリクエスタが現れます。

4 OKボタンを押すとパネルが閉じます。しかし二番目のリクエスタが現れ選択していた最初のオ
ブジェクトが表示されているリクエスタが現れます。

選択しているオブジェクトの数だけこの作業が繰り返され、前回選択したオブジェクト表示され
ているリクエスタが連続して開くのがわかりますね。最終的にリクエスタが表示される個数は選択
したアイテムの個数に依存します。見てわかるとおりデータ表示にこのメソッドを使用することで、
スクリプトはもしかすると多量のリクエスタを生成することになるかもしれません。これは効果的
とはとても言えませんが、このテストはスクリプトのforループ選択が適切に動作していること
を証明しています。まずい設計ではありますが動作はしています。

このスクリプトでは一つ大きな問題があります。問題を解説していくためにカレントシーンの 0
フレームを 1フレームに変更してみてください。リクエスタ群が再び現れてきましたね。これは前
述の通り、process()関数をトリガーしている全てのイベントに対して発生するからです。しば
らくの間、複数のinfo()リクエスタが出てきますが、この問題を我慢すると想定するなんてユー
ザーには耐えがたいことです。

CS コマンドの決定
選択しているアイテムを表示するコードを、いつ実行するのかを決定しなければいけません。実

演して見せたとおり、process()関数が呼び出されるたびにコードを実行させたくはありません。
情報がユーザーにとって価値を持つ場合にだけ、表示用コードを実行させたいですね。今回の場合、
コードはオブジェクトが選択されたときに実行されるようにしましょう。それではこのコード実行
の時期を選択させるにはどのように取り組めばよいのでしょうか？

前にも解説しましたが、process()関数に渡される引数の一つには、変数command内に記
憶されている値が入っています。例"masterTest.ls"ではprocess()関数が呼び出されるタイミン
グ、引数commandには文字列が含まれていることなどを学びました。この文字列はレイアウトか
ら発行されたコマンドとその引数が含まれています。この文字列を解析することにより、どのコマ
ンドが発行されたのかを把握することが出来、またコードがそのコマンドに対して注意を払うべき
かどうかも確定できます。

前述の例masterTest.lsを使用すれば何百という使用可能なコマンドの中から、一体どのコマンド
がコードで必要となるのかを確定することが出来ます。masterTest.lsスクリプトを実行し、Camera
アイテムを選択します。以下のような文字が描画されている info()リクエスタが表示されましたね。

“SelectItem 300000000”

ここで使用しているコマンドはSelectItemで、その引数は 300000000です。現時点で引数
はそれほど重要ではありませんが、この引数の値はLightWaveが内部的にアイテムを識別するため
に使用している番号です。今回のスクリプトではこの引数を使う必要はありません。Select
ITemというコマンドだけが必要なのです。

探しているコマンドが見つかりましたので、このコマンドに遭遇した場合にのみ forループ
コードの処理を行うようにしましょう。forループの真中に行を追加します（括弧を忘れないよう
にして下さい）。

...

if(event == COMMAND)

{

if(command == "SelectItem")

{

// items[]配列のコンテンツを表示します

for(i = 1; i <= selItems.size(); i++)

info(selItems[i].name);

}

}

スクリプトを実行し今までどおりテストして、どのように処理されるかを確認してみてください。
奇妙なことに、このコードを実行してみても何も起こらないことに気がついたでしょう。アイテム
をいくつ選択してもスクリプトは info()リクエスタが表示されなくなりました。どうしてでしょ
う？どこが悪いのでしょうか？

問題は、if-then文が二つの文字列、変数command内の文字列と"SelectItem"が一致す
るかどうかを比較しているところにあります。演繹法を使用すると、forループとinfo()文が
実行されない場合、この二つの文字列値が等しくないのです。そのためにコードはif-then文に
割り当てられている文を見ることがないのです。if-then文で何かまずいことが起こっているの
です。

if-then文の両側は決して等しくはなりません。なぜなら変数 commandの実際の値は
"SelectItem"ではなく"SelectItem 300000000"だからです。このため文字列が等しく
なることはなく、forループ文が実行されることもありえません。これを証明するため、if-
then文の前にもう一つinfo()コマンドを挿入して、変数commandの値の現在値を表示する
ようにしましょう。コードを実行すると、forループ文が実行される機会が来ないことがわかるで
しょう。

文字列command全体から、引数部分を無視し実際のコマンド部分だけを取り出す必要があり
ます。parse()関数を使用すれば取り出すことが出来ます。parseという言葉は、あるもの
（言葉など）を構成している部品へと分解していくことを意味しています。これがこの関数の動作
となります。区切りの文字と分解される文字列を指定すると、parse()関数は初期文字列を構成
していた全ての部分文字列を含む配列を返します。区切り文字列は単にparse()関数が指定した
文字列を分解するためだけに検索する文字列です。例えば

word = parse(“ “, “This is a test.”);

区切り文字列はスペース（" "）で、分解用文字列は“This is a test.”です。結果は以下のよう
に区切られた文字列が返されます。

word[1] = “This”

word[2] = “is”

word[3] = “a”

word[4] = “test.”

process()関数内のif-then文の前に以下の行を挿入してください。

...

selItems = sceneObj.getSelect();

// 変数 commandを解析します

command = parse(“ “, command);

if(event == COMMAND)

{

第１９章：マスタークラス 19.13



if(command == "SelectItem")

...

これらの行ではレイアウトから送られてきた変数commandを受け取り、parse()関数へと
送り出しています。parse()関数から返される配列には、変数commandが分解され保存されて
います。スペース（" "）文字を使用して変数commandを分解しているため、最初のインデック
スはコマンドを、2番目のインデックスは引数を表す配列が生成されるようになります。前述の例
では文字列commandは以下のとおりです。

“SelectItem 300000000”

変数commandは以下のように分解されます。

command[1] = “SelectItem”

command[2] = “30000000”

command[1]に保存されている値と文字列"SelectItem"を比較することにより、info()
リクエスタを多数生成することになります。では、if-then文を配列command[]の 1番目のイ
ンデックスと比較するように修正しなくてはいけませんね。新しいコードは以下のようになりま
す。

// 変数 commandを解析します

command = parse(“ “, command);

if(event == COMMAND)

{

if(command[1] == "SelectItem")

...

もう一度スクリプトを実行しアイテムを選択します。スクリプトは選択処理の場合にのみ更新さ
れることに気がついたでしょう。フレームの変更やウィンドウを開くといった処理に対しては、
forループ文に割り当てられている文が実行されなくなりました。表示用のコードを実行するアク
ションを単に限定することで、このスクリプトの有用性を証明しました。

不幸なことに、この小さな修正には欠点があります。info()リクエスタはオブジェクトを選
択したときには表示されますが、SHIFTクリックメソッドを使用してオブジェクトの選択を追加し
てみてください。何も起こらないでしょう。さらにテストを重ね、最初のアイテムが選択された時
だけ望みどおりの結果となり、複数のオブジェクトが選択されたときには表示されないことがわか
るますね。問題はコード中のエラーではなく、基本検索にエラーがあります。

アイテムが選択されたときにレイアウトが発行するコマンドは"SelectItem"コマンドである
と最初に発見した時に、すぐにそれを実装してコマンドの実例を見てみました。もう少し時間をか
けてレイアウトにおける選択システムがどのように動作しているのかを理解するべきでした。コー
ドを修正する羽目になる前に、必要となる追加情報へと導いてくれるはずだったのです。レイアウ
トがアイテム選択を扱うためのコマンドは一つだけではなく、実際には三つのコマンドを使用して
いるということを理解していませんでした。

もう一度MasterTest.lsスクリプトを実行してテストを繰り返してみると、アイテムを選択に追加
したり除去したりするのにレイアウトは三つのコマンドを使用しているのがわかりますね。これら
のコマンドが重要となるのです。

“SelectItem”

“AddToSelection”

“RemoveFromSelection”

コードが適切に動作するためには一つだけではなく、三つのコマンド全てをテストする必要があ
るのです。幸運なことに、新しい要求を満たすためにスクリプトを修正するのはかなり簡単です。
単にif-then文を一つではなく三つのコマンドと比較するように修正するだけでよいのです。

if( command[1] == "SelectItem" or

command[1] == "AddToSelection" or

command[1] == "RemoveFromSelection")

注意
コードを読みやすくするため、通常1行で足りるコード行を三つに分けました。どちらでも動作しま
すので、実装は開発者に委ねられます。

速度と効率性を保つためには、よく見てみると process()関数が呼び出される度に配列
selItems[]を作成する必要がないことがわかるでしょう。選択が変更された時にのみ、この配
列をもう一度作成する必要が出てくるのです。ですからselItems[]配列を作成するコード 2行
をif-thenブロックの中へと移します。

これが作業コードとなります。

//

// selected.ls - 現在選択されているアイテムを

// 列挙する Master Classのスクリプト

//

@version 2.2

@warnings

@script master

@name selected

// ここで大域変数を宣言

sceneObj;

create

{

// Scene Objectを作成します

sceneObj = Scene();

}

flags

{

// この関数ではシーンがクリアされた場合に

// レイアウトにスクリプトをクリアする必要があるかどうかを指示します

return(SCENE);

}

process: event, command

{

// 変数 commandを解析します

command = parse(" ", command);

if(event == COMMAND)

{

if( command[1] == "SelectItem" or

command[1] == "AddToSelection" or

command[1] == "RemoveFromSelection")

{

// 現在選択されているオブジェクトを取得します

selItems = sceneObj.getSelect();

// 配列 items[]のコンテンツを表示します

for(i = 1; i <= selItems.size(); i++)

info(selItems[i].name);

}

}

}

シンプルに
スクリプトは正常に動くようになりました。ここまでが第 1段階です。今度はコードをじっくり

と眺めてみて、より小さく、より速く、より効率的に実行出来ないかを調べてみます。スクリプト
は小さなものですが、使用するメモリ量を減らせる方法が一つありましたね。ある特定の環境下で
は実行速度が速くすることが出来るかもしれません。

まずは配列 selItems[]をどのように扱っているのかを調べてみましょう。Scene Object
Agentの getselected()メソッド実行時に、前述の selItems[]配列に保存されている
Object Agentsの配列を生成します。これらObject Agentはデータメンバとメソッドの形式でそれ
ぞれにかなりな量のデータを保持しており、それぞれ個々のアイテムに対する様々なプロパティを
記述しています。

さらによく見てみると、それぞれObject Agentのアイテム名称だけが必要であるとわかりますね。
配列から不必要なデータを破棄すれば、保存されているデータは動的に小さくなり、メモリ量も小
さくなります。Object Agentからデータを削除することは不可能ですが、アイテムの名称だけをも
つ新しい配列を作成することは出来ます。こうすれば望みどおりの結果も得られますし、後でスク
リプトに追加する機能も設定可能です。

新規配列を作成するために、selItems[]配列からアイテムの各名称だけをコピーし、
items[]という新しい配列に保存しましょう。selItems[]配列構築の後で、もう一つ別の
forループ文を配置すれば作成できます。

...

selItems = sceneObj.getSelect();

for(i = 1; i <= selItems.size(); i++)

items += selItems[i].name;

// items[]配列の中身を表示します

...

この場合、(+=)命令子で配列を作成すると一つの重要な欠陥が生じてしまいます。関数が呼び出
されるたびに同じ配列を使用しています。配列の値は各呼び出しの間で保存されており、配列
items[]は重複した名称で急激に崩れていきます。この問題を実証するため、現在の状態でスク
リプトを実行してみましょう。

この問題を修正するには、使用する間に変数へnilの値を代入して配列の値を壊すだけでよい
のです。そうすればforループが変数内へアイテム名称を保存するたびに、常に最初のインデッ
クスから始まることになります。

...

selItems = sceneObj.getSelect();

// 変数 itemsをクリアします

items = nil;

// items[]配列をより小さく、より効果的に作成します

...

問題が解決されると同時にスクリプトもより効率的になりました。selItems[]配列で使用さ
れているメモリを開放するため、変数の破棄が必要になります。items[]配列を破棄するのと同
じ方法で行います。selItems[]変数にも、値nilを単に割り当てるだけです。

...

for(i = 1; i <= selItems.size(); i++)

items += selItems[i].name;

// selItems配列で使用されているメモリを開放します

selItems = nil;

// 配列 items[]の中身を表示します

...

修正箇所はもう一つだけあります。二番目のforループにおいて使用される配列の名称です。
selItems[]配列の代わりに、items[]配列を反復処理するように設定しなおします。

...

// 配列 items[]の中身を表示します

for(i = 1; i <= items.size(); i++)

info(items[i]);

process()関数の中身のみを修正した最新版です。

process: event, command

{

// 変数 commandを解析します

command = parse(" ", command);

if(event == COMMAND)

{

if( command[1] == "SelectItem" or command[1] ==
"AddToSelection" or command[1] == "RemoveFromSelection")

{

// 現在選択されているオブジェクトを取得します

selItems = sceneObj.getSelect();

// 変数 itemsをクリアします

items = nil;

// より小さくより効果的な items[]配列を作成します

for(i = 1; i <= selItems.size(); i++)

items += selItems[i].name;

// selItems配列で使用されているメモリを開放します

selItems = nil;

// 配列 items[]の中身を表示します

for(i = 1; i <= items.size(); i++)

info(items[i]);

}

}

}

よりきれいに
解決し残している最も明らかな問題点は、items[]配列の中身をより効率よく表示するにはど

うすればよいのかということです。現在使用している info()リクエスタでも動作するにはしますが
あまりにも扱いにくいですし、最終スクリプトで使用するには邪魔です。アイテム名称の全リスト
を一度で表示したいと思います。残念なことに、これを叶えるためにはインターフェイスを作成し
なければなりません。

注意
インターフェイス作成にはただひたすら努力が必要となるでしょう。入出力の設計やユーザーイン
ターフェイスの実装にこのマニュアルの1章丸ごと費やしました。ですからこの議題に関しては理論
や解説をここではあまり深くは行いません。

基本的には、リクエスタを作成しitems[]配列の中身を表示するためのリストコントロールを
使用します。このコントロールは複数のアイテムをスクロール可能な一つのリストで一度に表示さ
せることができます。でもこれで問題が解決すると同時に、リストボックスコントロールは設定す
るインターフェイスコンポーネントのうち、かなり複雑なコントロールの一つでもあります。適切
に動作するためにはサポート関数をいくつも記述しなくてはなりません。

まずはitem[]配列をグローバルにしておく必要があります。こうすれば全ての関数、特にイ
ンターフェイスコードを扱うoptions()関数から配列のコンテンツにアクセス出来るようにな
ります。items[]配列をグローバルにするため、スクリプト冒頭において変数sceneObjの前
に変数を追加します。

...

@name selected

items, sceneObj;

create

...

次にoptions()関数を設定しましょう。この関数は、ユーザーがMasterプラグインパネル上
のプラグインのインスタンスをダブルクリックした時、もしくはプラグインの編集ドロップリスト
からプロパティオプションが選択されたときに、自動的に呼び出されます。ここではリクエスタを
適切に描画し更新するため必要となるインターフェイスコードが全て含まれています。

options

{

// リクエスタ既に開かれていないかどうかをチェックします

if(reqisopen())

// 開かれている場合には閉じます

reqend();

else

{

// "Selected:"という名称のリクエスタを作成します

reqbegin("Selected:");

// リクエスタのサイズは 130x300とします

reqsize(130, 300);

// リストボックスコントロールを作成します

c0 = ctllistbox("Selected Items:", 100, 290, "c0_count",
"c0_name");

// リクエスタをノンモーダルで開きます

reqopen();

}

その後リストボックスコントロールが適切に動作するために必要な関数を追加します。

c0_count

{

// この関数は変数のサイズを返します

return(items.size());

}

それに

c0_name: index

{

// この関数は indexで指定された配列の値を返します

return(items[index]);

}

最終コード

これで全てです。スクリプトが完成しました。以下はインターフェイスコードや機能をいくつか
追加し、ほんの少し構造を修正した最終スクリプトです。

//

// selected.ls - 現在選択されているアイテムを

// 列挙する Master Classのスクリプト

//

@version 2.2

@warnings

@script master

@name selected

items, sceneObj;

create

{

// Scene Object Agentを作成します

sceneObj = Scene();

// getSelected() UDFを呼び出します

getSelected();

// プラグインリストに表示される解説文を設定します

setdesc("Selected Items: ", items.size(), "item(s)");

}

flags

{

// この関数ではシーンがクリアされた場合に

// レイアウトにスクリプトをクリアする必要があるかどうかを指示します

return(SCENE);

}

process: event, command

{

// 文字列 commandを解析します

command = parse(" ", command);

// どのコマンドが更新をトリガーするのか確定します

if( command[1] == "SelectItem" or command[1] ==
"AddToSelection" or command[1] == "RemoveFromSelection")

{

// getSelected UDFを呼び出します

getSelected();

if(reqisopen())

requpdate();

}

}

options

{

// リクエスタが既に開かれているかどうかをチェックします

if(reqisopen())

// 開かれている場合にはリクエスタを閉じます

reqend();

else

{

// "Selected:"という名称のリクエスタを作成します

reqbegin("Selected:");

// リクエスタのサイズは 130x300とします

reqsize(130, 300);

// リストボックスコントロールを作成します

c0 = ctllistbox("Selected Items:", 100, 290, "c0_count",
"c0_name");

// リクエスタをノンモーダルで開きます

reqopen();

}

}

c0_count

{

// この関数はインターフェイスのリストボックスコントロール用の

// 関数であり items[]配列のサイズを返します

return(items.size());

}

c0_name: index

{

// この関数はインターフェイスのリストボックスコントロール用の

// 関数であり インデックスで指定した配列の値を返します

return(items[index]);

}

getSelected

{

// この UDFはこのスクリプトにおいて二つの機能を提供するために追加されました

// これは単にグローバル変数である item[]配列を構成するためのものです

// items[]配列を破棄します

items = nil;

// 現在選択されているオブジェクトを取得します

selItems = sceneObj.getSelect();

// items[]配列をアイテム名称で構築します

for(i = 1; i <= selItems.size(); i++)

items += selItems[i].name;

}

第２０章：インターフェイス概論 20.1

第20章 : インターフェイス概論

このマニュアルを通し、LScriptがLightWaveアーティストへ信じられないほどの力を提供出来
るということを実証してきました。ここまでで、ポイントやポリゴンの修正、サーフェイスの調節、
シェーダーの作成、アニメーション作成のスクリプトなど様々なスクリプトが書けるようになりま
した。LightWaveコミュニティに提供する用意が万端整ったと思うかもしれませんが、スクリプト
能力は使いやすさという新たな局面に突入しようとしています。

このセクションでは、インターフェイスコード記述の入出力についてカバーしていきます。前章
までと同様、簡単な例から始めインターフェイスコード独自の新規コマンドや関数などを解説して
いきます。インターフェイスコマンド全リストについての解説は行いませんが、コマンドの少数の
サブセットを学ぶことで、残りのコマンドについてもどのように動作するかといった一般的な概念
を取得できるようになります。インターフェイスのセクション全体を読み終える頃には、かなり高
度なユーザーインターフェイス（UI）に取り組んだり、少なくともそれがどういう動作をしている
のか、理解できるようにはなっているでしょう。

インターフェイスとは？
インターフェイスとはどういうものかというのは皆さんお分かりですね。コンピュータをつけた

り、LightWaveなどのアプリケーションを実行するたびにインターフェイスを使用しています。イ
ンターフェイスとはユーザーがボタンを押したりスライダを動かしたり文字を入力することが出来
る領域を指します。この発明によりコンピュータ上の生活はより容易になるのです。不運にもイン
ターフェイス無しのソフトウェアを使用しなければならない人に尋ねて御覧なさい。大抵、大量に
タイプを打たねばならず、何時間も黒画面を見入ることになります。ちっとも面白くありませんね。

スクリプトにインターフェイスを追加すれば、スクリプトに対する親切さが増しますが、新しい
レベルの複雑さをもコードに追加することになるでしょう。インターフェイスのコードは冗長で、
一度は美しくすっきりとしていたスクリプトが、いとも簡単にぐちゃぐちゃになってしまうほどの
反復性コードとなる場合がほとんどです。コードが騒然となってしまうことで、正しく動作させる
ために終にはスクリプトのデバッグまでもを必要とするエラーを生成してしまいかねません。では
何故インターフェイスを使用したがるのでしょう？もっと大切なことは、スクリプトに対しイン
ターフェイスは何を行うというのでしょう？



インターフェイスは単にコードへの窓または入口でしかありません。この入口でユーザーは変数
に保存されている特定の値を見たり変更することが可能になるのです。そう、コード内の変数を変
えるだけで以上の処理は可能ではありますが、インターフェイスはスクリプト実行中にこの処理を
可能にしてしまうのです。以下の状況を想像してみてください。

実際に素晴らしいポリゴンリダクションスクリプトを作成したとします。モデラーに存在する
ツールの大半と同様、処理するリダクションのレベル（1, 2, 3もしくは4）の設定を直接コードに設
定出来ます。処理回数の値は forループ文をコントロールする変数によって制御されています。
では 4回の反復処理を望むユーザーと同様、1回だけの反復処理を望むユーザーを満足させるため
のスクリプトを作成するにはどうしたらよいのでしょうか？

ループ変数にそれぞれの値を持たせたスクリプトを 4つ書くことも出来るでしょう。これでそれ
ぞれぞれの問題は解決されますが、コードの効率的な使用は難しくなりますね。ユーザーが 5回の
反復処理を望む場合にはどうなるのでしょう？答えはなんでしょうか？

答えはループをコントロールする変数にインターフェイスをリンクすることなのです。インター
フェイスはユーザーからの処理する分割用数値を要求します。インターフェイスからの変数はコー
ド内で変数へとリンクされているため（実際には同じ変数となりますが）、画面上の値を変更すれ
ば、コード内の値も変更されます。この機能はスクリプトに信じられないほど多大な量の柔軟性を
追加してくれます。基本的に同じ処理を行う 101個の異なるスクリプトを作成するのではなく、柔
軟性を持つスクリプトを一つ持つことになります。これが静的なスクリプトと動的なスクリプトを
作成する違いとなります。

静的なスクリプト
前章で作成された例は静的なスクリプトです。毎回スクリプトが実行される度にスクリプトの変

数に保存されているのは同じ値のままです。当然のこととして、スクリプトは同じ処理を行うこと
になります。シーンが読み込まれようとサーフェイスが作成されようとも、スクリプトは同じ方式
で何度も何度も処理します。スクリプトの変数が静的変数に固定されて組み込まれているからです。
以下のコードの例で静的変数を解説します。

main

{

objFile = “c:/newtek/objects/animals/cow.lwo”;

load(objFile);

}

何度このスクリプトを実行しようとも、常に同じ処理を実行します。静的変数であるobjFile
に保存されている値は変更できませんので、変わることは決してありません。しかし、もう一つの
オブジェクトを指すパス情報を修正したとしたら、スクリプトの関数は変更され、異なるオブジェ
クトを読み込むことになります。

main

{

objFile = “c:/newtek/objects/animals/spider.lwo”;

load(objFile);

}

spiderオブジェクトを読み込むとても便利なスクリプトを作っても、異なるオブジェクトを読み
込ませる度にコードを修正したり異なるスクリプトを作成しなければならないという、相変わらず
静的なスクリプトのままです。スクリプト実行中にこの変数を修正する方法が欲しいのです。変数
に保存されている値の中身によって異なる動作を実行するスクリプトであって欲しいのです。これ
を行うのが動的なスクリプトです。

第２０章：インターフェイス概論 20.3



動的スクリプト
スクリプト内部においてデータを固定してしまうのではなく、変数に対してインターフェイスを

作成することにより、変数の値を動的に作成することが可能です。スクリプト実行中に変数の値を
変更出来るようにすれば、様々な機能やオプションを提供することが出来ます。以下の例では一箇
所だけ修正を加えた前述の例をもう一度見てみましょう。

main

{

// インターフェイスコード

// ここでユーザーからオブジェクトのパス情報を取得し

// 変数 objFileに保存します

load(objFile);

}

注意
コードを読みやすくするために、実際のインターフェイスコードはまだ書いていません。コードはこ
れから作成していきますのでご心配なく。

変数objFileに入る値を特定の値に決める代わりに、インターフェイスを作成してユーザー
からの情報を要求することが出来ます。インターフェイスを実行した後、パス情報が変数
objFileに保存されます。どのような手段でデータを取得するかに関係なく、objFileが有効
なLightWaveのオブジェクトファイルである限り、モデラーはパスに保存されているオブジェクト
を読み込みます。

スクリプトは動的であるとみなされます。スクリプトの関数を実行する度に変更しなくても、結
果的にはそうなります。開発者ではなくユーザーがどのオブジェクトを読み込むのかを確定出来る
ため、より使いやすいスクリプトになっています。これがインターフェイスを記述する最も大きな
利点の一つです。

リクエスタ

インターフェイスは全てリクエスタから始めなければなりません。リクエスタとはインターフェ
イスが最初に作成される時点で表示されるパネルです。リクエスタには、スクリプトからデータを
要求したり表示したりするためのコントロールと呼ばれるエレメントを配置することが出来ます。
各リクエスタには何十、時には何百ものコントロールがあります。画面上に必要となるコントロー
ル全てを表示する領域だけが、コントロール数の制限となります。

注意
何百ものエレメントを持つ、大きなインターフェイスを作成するのは賢いやり方ではありません。
このようなスタイルは、ユーザーを圧倒し困惑させるだけです。

リクエスタウィンドウを画面上で動かしてみると、リクエスタに付けられているコントロールも
全てパネルと一緒に移動しますね。本来、コントロールはリクエスタにリンクされているのです。
LScriptは追加したコントロールを編集中のカレントリクエスタに対し自動的に割り当てます。です
から、インターフェイスに割り当てずにコントロールを作成することは不可能です。

ユーザーとの対話のレベルは、リクエスタのインターフェイスを使用してスクリプトが何を必要
としているのかに依存します。例えば、頻繁に使われるinfo()関数は、リクエスタを作成し文字
列を表示するだけです。ウィンドウを閉じるために"OK"ボタンを押す以外、インタラクティブにや
り取りすることは何もありません。ですが、他のスクリプトではユーザーがデータを入力したり値
を見るなどの様々なコントロールを持っている場合があります。

図 20-1. シンプルな info()リクエスタ

各リクエスタには、パネルのタイトルバー表示部分に名称を割り当てることが出来ます。パネル
のサイズとリクエスタ領域は画像の座標値と同様、二次元ピクセル値（XとY）で測定されます。
パネルに対しサイズを指定することも出来ますし、LScriptが適切なサイズを自動で確定することも
出来ます。LScriptはどのコントロールが使用されるのか、各コントロールが正しく表示されるには
どのくらいの領域が必要となるのかなどを考慮に入れます。時間が絶対必要となる場合、短時間で
終了するインターフェイスへとたどり着きます。

図 20-2. 空のリクエスタ

LScriptは画面上でのコントロールの配置、各パネルやコントロールの色、リクエスタを開いた
り閉じたり、最小化、最大化を行うために必要とされる全てのボタンなど、ありとあらゆるパネル
属性を自動的に扱います。ですからリクエスタは全て適切に動作するように作成され、他のスクリ
プトやLightWave内にあるプラグインと同じような外観となるのです。これらのデフォルトがなけ
ればインターフェイス作成処理にはさらに時間がかかり、不測のエラーを引き起こしがちになるで
しょう。

このパネルは単独で何か重要な処理を行うわけではありませんが、この章が終了する時点では
ユーザーが対話を必要とするコントロールが全て含まれることになるでしょう。上記リクエスタは
インターフェイスが適切に動作するため必要最小限にしてあります。

コントロール

コントロールとは、ユーザーが対話を行うインターフェイス内のエレメントを指します。コント
ロールそれぞれは、スクリプトからのデータを収集したり、表示するために設計されています。
LScriptはスクリプトで使える多くの利用可能なコントロールを用意しています。ボタンやデータを
通してスクロールできるスライダ、文字入力可能なフィールド、それにカスタムメードの画像コン
トロールなどです。これらコントロールはスクリプトリクエスタ上に配置されます。

図 20-3. リクエスタとサンプルのコントロール

上記リクエスタでは、選択可能な様々なタイプのLScriptのコントロールを図解しています。こ
れらのコントロールには直接フィールドにデータを入力できるものもあれば、オプションリストか
らの選択が可能になるものもあります。スクリプトを実行することで、これらの値を全て取得する
ことが出来、この値はオプションやユーザーが選択した値を確定することもあります。

注意
インターフェイスのコントロールや関数の完全なリストについては、リファレンスマニュアルの第2
章を参照してください。

インターフェイスの種類
LightWaveには二種類のLScriptがあります。ツールやユーティリティのように特定のアイテム

データのフレームから独立した機能を持つスクリプトと、Item Animationスクリプトのように、各
フレームに対しアイテムデータを評価するスクリプトです。どちらのインターフェイスの種類も同
じような方法でコード化され、同じインターフェイスのコマンドや関数を使用することになります。
しかし、この二つの共通点も実際のスクリプト構造を比較する段階までです。

前章では、モデラーやGenericクラスでスクリプトのコマンドを含む単一関数、main()や
generic()の使用法について解説してきました。この単一の関数にはスクリプトを開き処理し、
閉じるまでに必要となるコード全てが含まれています。理論的にはこの関数内にスクリプトのイン
ターフェイスコードも全て配置されることになります。

注意
もちろん、モデラーまたはGenericクラスのスクリプトは実行中、他にいくつものユーザー定義関
数（UDF）を呼び出すことが可能です。

このタイプのスクリプトを実行すると、コード化されているアクションを実行するのみで、コン
トロールは即座にレイアウトまたはモデラーへと返されます。コードが実行終了したら、スクリプ
トも終了となるのです。スクリプトはユーザーが再び起動したときにのみ再実行されます。

もう一方で、レイアウトのスクリプトは複数の関数を使用し、追加したい機能全てをサポートし
ます。レイアウトスクリプトでは、処理の大半を行うコードはprocess()関数内にあります。
重要なコードの大半がprocess()関数にあっても、他の関数も変数を設定したり、読み込みや
書き込みを扱ったり、スクリプト終了後のクリーンアップ処理が行えます。ひとえに何を処理した
いのかによるのです。

注意
以下の章を通し、レイアウトクラスのスクリプトを参照していきます。このスクリプトクラスはレイ
アウトにあるクラスであり、複数の特定化された関数を使用してスクリプトのアクションを実行して
います。しかしGenericスクリプトの構造はレイアウトスクリプトよりもモデラーのスクリプトと
似通っています。ですからレイアウトスクリプトを参照するというのは、Genericクラスを除いた全
レイアウトのスクリプトクラスを参照することになります。

レイアウトスクリプト実行時には、アニメーションの各フレームにおいてコードとシーンデータ
を計算します。これはシーンのレンダリング時と同様、ユーザーがアニメーションをつけたりプレ
ビューさせたりしている時点でも発生します。これらのスクリプトはフレームが変更される度に呼
び出されますから、アニメーション作成の経過途中において何百、何千回も呼び出されることにな
ります。インターフェイスコードの構造の議論において、これは大きな問題を提起します。

第２０章：インターフェイス概論 20.7



例えば、Item Animationスクリプトでインターフェイスを作成するのに、何も考えずにアプ
ローチしてみましょう。重要なコードはprocess()関数内にあるとわかっていますから、イン
ターフェイスを作成するコードをprocess()関数に配置してみようかと思うかもしれませんね。
こうすればインターフェイスは確実に呼び出されますし、ユーザーから変数の設定が確実になるで
しょう。スクリプトは万事うまくいきそうです。

しかし、この方法ではフレームが変化したという情報を受け取るたびに、process()関数が
呼び出されてしまうという大きな問題が生じます。ですから Item Animationスクリプトが割り当て
られているオブジェクトを編集すると、インターフェイスが現れます。これはすぐに古くなってし
まうでしょう。ここで登場するのがoptions()関数です。

インターフェイスコードをprocess()関数から切り離して、options()と呼ばれる関数内
に配置することで、パネルはユーザーが要求した時にのみ呼び出されるようになります。ユーザー
はプラグインのリストウィンドウにあるスクリプトのインスタンスをダブルクリックするか、プラ
グインのコマンドリストからEdit（編集）> Properties（プロパティ）を選択することによって、
インターフェイスを呼び出すことができます。

図 20-4. Item Animationスクリプトのインスタンス

この二つのアクションどちらかを実行することで、LScriptはスクリプトコード全体を通して
options()関数を検索します。関数がスクリプト内部に存在すればそのコードを実行し、その間
はスクリプトを停止させます。しかし、関数がスクリプト内に存在しなかった場合、開発者はこの
スクリプトに対しインターフェイス描画を望んでいなかったのだと想定し、ユーザーに対しオプ
ションは利用できないとの旨を通知します。

LSIDE： Interface Designer（インターフェイスデザイナー）
LScript Integrated Developement Environment（LSIDE）には非常に強力なインターフェイス

設計アプリケーションが含まれています。"見たままのものをお手元に"（WYSIWYG）というコン
セプトのコーディング環境を使用すれば、インターフェイスを手早く簡単にレイアウトし作成する
ことが出来ます。Interface Designerでは描画したりドラッグしたりリクエスタ上のコントロール
を整列させることで、簡単なまたは複雑なインターフェイスを作成することが出来ます。処理が完
了すると、Interface DesignerはリクエスタをLScriptコードとして、スクリプトへコピー
アンドペーストしてエクスポートします。

図 20-5. LScript Interface Designer(インターフェイスデザイナー)

この処理は恐ろしいほど素早く効率的ですが、その中で一体何が起きているのかを知っておくべ
きであると考えます。ある状況下では、Designerからエクスポートされるコードの目的や機能など
について把握し、理解することが必要となるでしょう。また理解することにより、Designerを再起
動し古いコードの上に新しいコードをカットアンドペーストするなどの厄介な作業を行うことな
く、コードを追加したり修正したり出来るようになります。

インターフェイスがどのように動作するのか、またどのようなコードで作成されるのかしっかり
と理解出来るようになる頃には、Interface Designerは間違いなく強力なツールとなり、あなたは
Designerを使用して必要なインターフェイスコードを全て用意できるようになるでしょう。ですか
ら今はDesignerを使うという欲求と戦い、まずはより困難な道を選ぶことにします。初めての大き
なプロジェクトを書こうとする時には、次の章で蓄えていく知識が非常に貴重なものとなるはずで
す。

注意
Interface Designerのより詳細な情報については、このマニュアルの最終章 "LSIDE" を参照して
ください。

第２１章：モデラーとジェネリック（総括）クラスのインターフェイス 21.1

第21章 : モデラークラスと
Generic（総括）クラスのインターフェイス

Generic（総括）クラス、およびモデラーのスクリプトクラスは、根本的に全く異なる関数やコ
マンドを使用しているものの、どちらのスクリプトに対するインターフェイスも同じ様式でコード
化されます。どちらもユーティリティとして機能するため、インターフェイスはユーザーから必要
となる情報を取得すると実際に機能を実行するコードへと進むような形に設計されています。この
簡易性のおかげで、インターフェイスのコーディングはかなり易しく、構造や仕掛けをそれほど必
要とはしません。このタイプのインターフェイスが他のスクリプトタイプのインターフェイスに比
べ強力でないとは言っていません。ただ、より簡易化されているのです。

注意
クラスのスクリプト処理に対する制限と同じ制限が、インターフェイスコードにもつけられていると
いうことが重要です。例えばモデラーのインターフェイスは、レイアウトインターフェイスで見られ
るカレントのフレーム情報についてはアクセスすることが出来ません。

モデラークラスのインターフェイス
モデラースクリプトは、ジオメトリのモデリングをしやすくするために設計されたツールです。

このタイプのスクリプトではシーンフレームやライト、カメラの設定などにアクセスする必要がな
いですし、シーンのフレーム情報が変化するたびに処理する必要もなく、インターフェイスコード
の学習を始めるには非常に良い出発点となるでしょう。スクリプトの構造、特にインターフェイス
コードを限りなく簡易化しています。

注意
再教育講習が必要であれば、リファレンスマニュアルのモデラースクリプトの章を参照してください。

モデラースクリプト用にインターフェイスを作成するのは非常に簡単です。いったんヘッダーや
変数などの通常のスクリプト形式が処理されれば、インターフェイスコードが始まります。イン
ターフェイスデータがユーザーから収集され適切な値が設定されたら、スクリプトのメイン処理
コードが始まります。他のパネルを開かない限りは、スクリプトにおけるインターフェイス部分は
これで終了です。



始めてみよう！
この例ではインターフェイスコードの導入に焦点をあてていますので、驚くほど簡単なスクリプ

ト用のモデラーインターフェイスを作成していきましょう。このスクリプトは二つの数値を足して、
その結果を info()パネルに出力するというものです。それではモデラークラスのスクリプト
"ModInterfaceTest"を設定していきましょう。

@script modeler

@name ModInterfaceTest

@version 2.3

main()

{

}

より簡単にするため、大まかな輪郭を挙げていきます。

1 インターフェイスを描画します

2 インターフェイスから値を取得します

3 二つの値を加算します

4 結果を表示します

これをスクリプトに挿入してみます

...

main()

{

// インターフェイスを描画します

// インターフェイスから値を取得します

// 二つの値を加算します

// 結果を表示します

}

リストの 1番目の事項はインターフェイスの描画となっていますが、まずはコードがうまく動く
かを確認します。テストの値と変数をいくつか作成する所から始めましょう。スクリプトの主要部
分が動作していれば、一時的な値の割り当てを削除してインターフェイスコードを書きます。こう
することで、確実に動作が保証されているコード上でインターフェイスが記述することになります。
後でも解説しますが、長い目で見ればこの方法を取ることで時間が節約できます。ここでは重要な
ことは何も行いませんが、同じページ上に全て存在しているか確認するために、機能コードは以下
のようになります。

...

main()

{

// インターフェイスを描画します

// インターフェイスから値を取得します

val1 = 10;

val2 = 5;

// 二つの値を加算します

val3 = val1 + val2;

// 結果を表示します

info(“sum = “, val3);

}

注意
ここには一時変数val1と val2があります。この二つの変数はインターフェイスが挿入される時点で
除去されます。

このコードを実行すると、info()パネルには 15と表示されているでしょう。10+5の合計値を表示
するスクリプトが必要だったのですが、これで完了です！スクリプトは単純ですが、ユーザーに二
つの値を要求し、合計値を表示するインターフェイスを作成する場合には、さらに便利にすること
が出来ます。

LScriptインターフェイスにおいて、最も重要でなおかつ根本となるものがリクエスタです。パ
ネルが一つもなければ、コントロールを挿入する場所がありません。リクエスタを作成するため、
インターフェイスコードはreqbegin()関数を使用してLScriptからリクエスタモードへと切り
替える必要があります。この関数は二つの処理を行います。一つは指定されたタイトルでリクエス
タを作成すること、もう一つは、LScriptに対しどのパネルが有効になっているのかを伝えることで
す。この呼び出しの後で追加されたコントロールは、アクティブなパネルに割り当てられることに
なります。

インターフェイスパネルの上部に表示されるタイトルは、reqbegin()に渡される文字列によ
り設定されます。ご想像どおり、reqbegin()関数はreqend()という関数と一組になってい
ます。この関数はリクエスタモードから抜け出てインターフェイスを閉じます。カレントのスクリ
プトでこの二つのコマンドを動かしてみましょう。

...

// インターフェイスを描画します

reqbegin(“Sum of two numbers:”);

// インターフェイスから値を取得します

reqend();

val1 = 10;

val2 = 5;

...

ファイルを保存し、このコードをモデラーでテストしてみます。インターフェイスが現れなくて
も心配しないで下さい。処理中における大変重要なステップを残したままなのです。LScriptに伝え
ることはリクエスタを作成し、それを閉じるということだけでした。LScriptにインターフェイス
作成の作業が終了したことを通知し、インターフェイスを表示させる必要があります。これには
reqpost()関数を使用します。

...

reqbegin(“Sum of two numbers:”);

reqpost();

// インターフェイスから値を取得します

reqend();

...

これでリクエスタが表示されるようになり、LScriptは二つのボタンのどちらかが押されるまで、
どんなコードも実行しません。スクリプトを実行すると、以下のようなインターフェイスが表示さ
れていることでしょう。

図 21-1. 何の処理も行いませんが、ここが出発点です

注意
"OK"と"Cancel"ボタンは自動的に作成されます。

ではこのリクエスタにいくつかコントロールを挿入してみましょう。ユーザーから数値をいくつ
か取得したいので、ctlinteger()コントロールを使用します。この関数は呼び出されると二つ
の値を受け取ります。一つはコントロールのラベルを表す文字列、そしてもう一つはデフォルト値
を表す整数値です。新規2行を既存のスクリプトへと挿入します。

...

reqbegin(“Sum of two numbers:”);

c0 = ctlinteger(“Value2: “, 5);

c1 = ctlinteger(“Value1: “, 10);

reqpost();

// インターフェイスから値を取得します

...

有効で実用的なデフォルト値でコントロールを作成することが大切です。この値は、ユーザーに
対しリクエスタ内に入力されるべき値の手がかりを示すものです。ユーザーがツールの適切な使用
法がわからない場合、"OK"ボタンを押してみてデフォルト値ではどのような動作になるのかを見る
かもしれません。デフォルト値が無効であったり間違った選択をしている場合、スクリプトの良い
試運転とはいきません。エラーさえ引き起こすことになりかねません。

上記で追加した二行では、リクエスタ上のコントロールを配置する方法を紹介しています。
ctlinteger()関数は二つの引数値と共に呼び出されます。コントロール作成に成功すれば、関
数はHandleという値を返します。このHandleは基本的にLScriptが各コントロールを一意に
認識するために使用する内部的な値です。このHandleは変数c0に保存されます。

スクリプトを実行してみると、二つのコントロールが画面上に正確に配置されているのがわかり
ますね。

図 21-2. 完成間近のインターフェイス

Value1の上に、Value2とラベルがつけられたコントロールが挿入されているのがわかりま
すね。これにより二つのことが実証されました。一つは、最初に作成されたコントロールがデフォ
ルトとしてリクエスタ上に（上から下に向かって）配置されるという点です。二つ目はリクエスタ
内部におけるコントロールの位置やサイズです。コントロールの移動やサイズ変更の方法を紹介す
るために、ここでctlposition()関数を使用してみます。

...

c0 = ctlinteger(“Value2: “, 5);

c1 = ctlinteger(“Value1: “, 10);

ctlposition(c0, 35, 30);

ctlposition(c1, 35, 5);

reqpost();

// インターフェイスから値を取得します

...

ctlposition()関数は三つの引数（三つのオプション引数）を要求します。1番目は影響を
及ぼしたいコントロールのHandleです。この場合、変数c0、c1に保存されているHandleを
使用します。2番目と 3番目の引数は列数と行数です。画面上のコントロールを修正しようと思う
と、修正結果を確認するためにスクリプトを実行させなければならず、大変な時間の浪費となる場
合もあります。このためにLSIDEインターフェイスエディターを使用すれば、時間を著しく節約で
きます。

スクリプトを実行し結果を見てみましょう。

図 21-3. リクエスタ上にコントロールを手動で配置することにより、全く違う外観を作成できます

コントロールの位置が正確に修正されているのがわかりますね。ctlposition()関数を使え
ば、スクリプト作成時においてインターフェイスの外観達成に必要となる柔軟性が与えられます。
しかし多くの簡単なスクリプトでは、良い計画と適切なコードを持ちさえすれば、リクエスタ上に
コントロールを自動的に配置させ、このステップを回避することが可能なのです。

21.6 LScript 日本語ユーザーガイド



あとはコントロールから値を取得するだけです。getvalue()関数から返される値を、宣言し
た変数へと保存し、値を取得します。この関数は値を取り出したいコントロールのHandleを要
求します。今回の場合は

...

// インターフェイスから値を取得します

val1 = getvalue(c0);

val2 = getvalue(c1);

reqend();

// 結果を表示します

...

注意
以前使用していた一時変数val1と val2を忘れずに削除してください。

これでコントロールの値は変数val1とval2に保存されました。これらの宣言部分が終了す
れば、スクリプトは実行時に想定どおりの処理を行うようになります。テストしてみてください。

うまく動いているように見えますが、もう一度テストしてみてください。今度は二つの値を入力
した後、"Cancel"ボタンを押してみてください。依然として結果が表示されていますね。ユーザー
インターフェイスの慣例に反しています。"Cancel"ボタンが押されたときには、アクションは無視
されるべきなのです。この問題を修正するため、スクリプトを変更しましょう。

この問題が引き起こしている間違いの一つは、現在reqpost()関数の戻り値を無視している
ことにあります。この値は、ユーザーがリクエスタパネルを閉じるのに"OK"ボタンを押したのか、
"Cancel"ボタンを押したのかを示す値です。現在のreqpost()関数を以下のコードで置き換える
ことにより、この値を取得することが可能です。

...

return if !reqpost();

...

このコードではreqpost()関数がブール値'false'（つまり"Cancel"ボタンが押下）の場合
には、スクリプトが呼び出された時点へと関数を返すという処理を行います。この場合、モデラー
へと返されます。基本的にはこれでスクリプトは終了です。ではスクリプトを動かしてみましょ
う。

最終コード：

@script modeler

@name ModInterfaceTest

@version 2.3

main

{

// インターフェイスを描画します

reqbegin("Sum of two numbers:");

c0 = ctlinteger("Value2: ",5);

c1 = ctlinteger("Value1: ",10);

ctlposition(c0, 35, 30);

ctlposition(c1, 35, 5);

return if !reqpost();

// インターフェイスから値を取得します

val1 = getvalue(c0);

val2 = getvalue(c1);

reqend();

// 二つの値を加算します

val3 = val1 + val2;

// 結果を表示します

info("sum = ", val3);

}

Generic（総括）クラスのインターフェイス
Generic（総括）クラスとモデラークラスのスクリプトにおけるインターフェイス作成法には、

何ら違いはありません。構造的に大変似通っているため、インターフェイスのコーディングは事実
上、同一のコードとなります。これを実証するため、前述の例をモデラークラスのスクリプトから
Genericクラスのスクリプトへと変更し、レイアウトから実行してみましょう。モデラースクリプ
トとすべて同じように動作し、表示されるのがわかりますね。

インターフェイスのコードの箇所でのみ比較できます。本来、Genericクラスのスクリプトで利
用可能な関数やコマンドは、レイアウトのみに限定されているからです。

第２２章：レイアウトインターフェイス 22.1

第 22 章 : レイアウトインターフェイス

インターフェイスには明確な機能が二つあります。ひとつはユーザーが作成したり修正を加えた
い設定を取得すること、もうひとつは保存されている値をユーザーに表示し、現在値や設定を通知
することです。優れたインターフェイスというのはこの双方を実現しています。その点ではレイア
ウトスクリプトもモデラースクリプトも同類です。

しかし前述した通り、モデラーとレイアウトのスクリプトの間にある大きな違いは、その構造に
あります。モデラースクリプトには事前定義された一つの関数が含まれますが、レイアウトのスク
リプトが適切に動くためには、複数の関数が必要となります。これはレイアウトスクリプトに対す
るインターフェイスのコーディングの仕方に直接影響を及ぼします。Item Animationスクリプトの
テンプレートを見てください。この例では何も処理していませんが、スクリプトのスケルトンコー
ドを設定しています。

@version 2.2

@warnings

@script motion

create: obj

{

}

destroy

{

}

process: ma, frame, time

{

}

load: what,io

{



if(what == SCENEMODE) // ASCIIシーンファイルを処理

{

}

}

save: what,io

{

if(what == SCENEMODE)

{

}

}

注意
関数内で何の処理も行わない場合、関数を宣言する必要はありません。しかし今回はこれらの関数が
どのようにサポートされるのかを見るために、あえて関数を宣言するようにします。関数宣言の方法
を理解したら、コード部分を含んでいる関数だけを使用するように、コードを切り取ることが出来ま
す。

見てわかるとおり、関数 create()、destroy()、process()、load()、それに
save()は全てスクリプトの実行時に特化された機能があります。重要度に違いはあるものの、全
ての関数をあわせてスクリプトの機能を完全なものにしています。LScriptはこれらの関数をそれぞ
れいつ実行するのかを把握し、スクリプト実行箇所で呼び出します。ではどこにインターフェイス
のコードを挿入すればよいのでしょうか？

インターフェイス概論の章で解説した通り、process()関数にインターフェイスコードを追
加するだけでは出来ません。これでは絶えずリクエスタが現れてしまい、ユーザーはすぐにいらい
らしてしまうでしょう。しかし options()という関数を宣言することで、LScriptに対しイン
ターフェイスコード内で探しに行く場所を正確に指示することが出来ます。前述の例の終端に、次
のコードを単に追加してください。

options

{

reqbegin("An Animation Test");

return if !reqpost();

reqend();

}

これらのコマンドに馴染みはありますか？解説したように、モデラーとレイアウトインターフェ
イスで使用されるコマンドと関数は同じものです。上記で使用されるコマンドは、最初に動作する
インターフェイス用に最低限必要なものです。事実、この特定の状況においてはreqend()関数
が存在する必要はありません。LScriptは options()関数を通り過ぎる時に自動的にインター
フェイスを終了させるからです。しかしパネルを追加する場合、またはインターフェイスコードを
後で追加する場合に備えて、手動でインターフェイスを閉じるよう習慣付けておく方が良いでしょ
う。

この Item Animationスクリプトを実行してみてプラグインリスト内にあるこのインスタンスを
ダブルクリックしてみましょう。以下の画像のようなパネルが現れましたね。

図 22-1. 空のOptionsインターフェイス

options()関数内部において、前章で行ったのと同じ方法でコントロールを追加し、望みど
おりに機能するインターフェイスを作成することが出来ます。モデラーインターフェイスで使用し
たときと全て同じような外観と機能を持ちます。しかしレイアウトインターフェイスがモデラーの
インターフェイスと全く同じような動作し始める前に、取り組むべき箇所がもう一つあるのです。

変数スコープ
関数内に変数を宣言した時に、その変数に対するデータは同じ関数内部のコードでのみ利用可能

です。例えばAとBという二つの関数がある場合、関数Aの仲で変数TEMPを宣言すると、関数B
ではTEMPの存在を知ることなく、カレント値を正しく与えようと気にかけることは決してないで
しょう。これがレイアウトインターフェイスで現在直面している問題なのです。

インターフェイスからデータを集めるコードはoptions()関数内に置かれます。ところが、
実際のスクリプトのタスクを実行するコードは process()関数内に置かれます。そのため、
インターフェイスから集められた値は、スクリプトの処理コードからアクセスできません。変数は
options()関数でローカルに宣言されており、create()やprocess()関数ではその値を読
み取れないのです。つまり可変関数の間で共有される変数は、グローバル変数として宣言されなけ
ればならないのです。

変数名称を指示子セクションの後、コードに記されている最初の関数の前に宣言するだけでよい
のです。例として以前のコードを使用してみましょう。

@version 2.2

@warnings

@script motion

// ここに大域変数を挿入

option1 = “Interface Test.”;

option2 = .01;

option3 = 1;

create: obj

{

}

全ての変数を大域変数として宣言したくはありませんが、変数 option1、option2、
option3は、options()関数だけではなく全関数内からでも利用可能になりました。もっと大
切なことは、process()関数内のコードはインターフェイスから集められたデータを使用するこ
とが出来るようになったということです。

解説文
最後にインターフェイスで選択した重要なオプションをまとめて、スクリプトの解説リスト上に

表示し、ユーザーに通知しなければなりません。setdesc()コマンドを使用して処理を行います。
表示したい情報は、文字列引数として保存されているinfo()関数と同様の働きをします。

注意
この関数はモデラースクリプトでは利用できません。

例えば、ptions()関数の終端にこの行を配置してみます。

...

reqend();

setdesc(“Script: ” + option1);

}

...

リクエスタを閉じると、この関数はoption1に何を選択したのかをユーザーに対し正確に通
知します。お気づきのとおり、選択された値全てを表示することは出来ません。何を選択したかが
問題となるような、重要な値だけを表示するようにします。

例：

アイテムのキーフレームの値から厳密に最大座標値を設定する、簡単な拘束アニメーションスク
リプトを作成してみましょう。これは Item Animationスクリプトで、アイテムが移動するに従って
全ての動きを封じるようにします。いつもどおりヘッダー設定から始めましょう。

@version 2.3

@warnings

@script motion

最大X座標値を保存するための大域変数を宣言します。

maxXPos = 1.0;

まずはprocess()関数から始めましょう。最初に処理コードがすべて正常に動くようにしま
す。それからインターフェイスを開発し、ユーザー入力として使用する値を全て扱うインターフェ
イスを開発します。Item Animationスクリプトの内部作成を終わらせるよりも、インターフェイス
コードの中へ直接飛びましょう。しかしスクリプトは同様の動作を実行しますので、以下の処理機
能を用意します。

process: ma, frame, time

{

// ma Motion Object Agentを使用して現在位置を取得します

currPosition = ma.get(POSITION,time);

// アイテムの位置が maxXPosより大きければ

// maxXPosに変更します

if(currPosition.x > maxXPos)

currPosition.x = maxXPos;

// 位置情報を設定します

ma.set(POSITION, currPosition);

}

注意
この関数内で使用している馴染みの薄いコードについては、コメントで解説します。コメントがない
場合は、リファレンスマニュアル25章のMotion Object Agentを参照してください。リファレン
スマニュアルのこのセクションでは、この関数内で使用されているコマンド全てが解説されています。

任意のアイテムモーションプラグインリストでこのスクリプトを実行させ、アイテムを動かして
みてください。アイテムがmaxXPosの値以上になったら動かなくなるのがわかりますね。今回の
場合、最大X位置（maxXPos）は大域変数として値 1.0で宣言されています。変数maxXPosの
値を手動で変更することで、最大X位置を変更します。

テストを十分に行い、コードの動作を確認したら、インターフェイスを追加していきましょう。
まずはoptions()関数と初期リクエスタを作成するところから始めます。

options

{

// リクエスタを開きます

reqbegin("Clamp");

return if !reqpost();

// インターフェイスを閉じます

reqend();

}

三次元空間内においてX座標値を表現したいので、浮動小数点数を使用します。数値コントロー
ルを追加します。

...

reqbegin("Clamp");

// 数値コントロールを作成します

c0 = ctlnumber("Max. X Position",xPos);

...

数値コントロールのデフォルト値が変数xPosの値に設定されているのがわかりますね。xPos
は値1.0を持つように宣言されていますので、コントロールのデフォルト値は 1.0となります。

パネルを閉じた後にコントロールの値を収集し、グローバル変数xPosに値を保存しましょう。

...

return if !reqpost();

// インターフェイスからの値を取得し大域変数 xPosに保存します

xPos = getvalue(c0);

// インターフェイスを閉じます

...

スクリプトを実行すると、下のパネルのようなインターフェイスが現れます。

図 22-2. Clampインターフェイス

プラグインリストに解説文を追加しましょう。

...

reqend();

// プラグインの解説文に xPosの値を送信します

setdesc("xPos:" + xPos);

}

これでインターフェイスは完成です。

機能追加
この単純なインターフェイスは適切に動作するようになりましたので、他の二つの座標値も

カバーするよう簡単に拡張できます。コピーアンドペーストし、変数の名称を変更し、大域変数を
いくつか追加するだけで、下記に示すようなインターフェイスを持つスクリプトが、いとも簡単に
出来あがります。

図 22-3. 拡張されたClampインターフェイス

他に二つの数値フィールドとそれをサポートするコードを追加しましたので、完全な座標システ
ムをサポートすることになり、スクリプトはより使いやすくなりました。最大値と最小値を拘束す
るスクリプトを作成すればさらに拡張することが出来ます。インターフェイスは以下のようになり
ます。

図 22-4. さらに拡張されたインターフェイス

このスクリプトとインターフェイスを作成するコードは、先ほど動かしたサンプルのスクリプト
を派生させただけに過ぎません。概念も関数もコマンドも、新しいものは何も導入されていません。

このように、インターフェイス作成法は段階を踏んで組み立てていくことでスクリプトの構築法
を強化してくれます。最初に実用的なコードを、それからインターフェイスを、その後でもう一歩
先に進めたコードを書くことにより、非常に柔軟性があってしっかりしたコードを作成しました。
この手法に従えば、より大きなスクリプトやインターフェイスを作成する場合においても、さほど
頭を悩ませることはないはずです。

22.8 LScript 日本語ユーザーガイド



第２３章：インターフェイス設計における規則 23.1

第23章 : インターフェイス設計における規則

初期のスクリプトでは、概念と設計という面で非常に単純なインターフェイスを作るべきです。
インターフェイスで複雑な動作を実装させようとする前に、まずコントロールの基本とコントロー
ルがどのように機能するのかを学び理解して下さい。インターフェイスのコーディングは信じられ
ないほど難しいというものではありません。スクリプトにおける他の全ての側面と同様、全てがう
まく動かせるようになるには、時間とそれに適用するための知識が必要となります。インターフェ
イスを書き続け、サンプルを研究すればするほど、独自のインターフェイスがよりよい外観を持ち、
動作を行うようになるでしょう。

インターフェイスのコードを学習する一方で、従わなければならない規則がいくつかあります。

規則 #1: まずは機能性
インターフェイスはユーザーが直接見て対話するものですから、設計するのはとても楽しい作業

となります。しかし開発の技術はインターフェイスデザインだけで判断されるなどという馬鹿げた
考えは信じないで下さい。初心者にはスクリプトの実際のコードを書くよりもインターフェイスの
プログラムにより時間をかけるという過ちを犯す人もいます。インターフェイスは素晴らしいもの
ですし、つまらないインターフェイスよりかはプロフェッショナル的に見せてくれはしますが、
コードを便利にするため以外の何者でもないのです。

開発を始めたばかりですから、スクリプトは最初に想定していたようには動作しないかもしれな
いと言っても差し支えないでしょう。過ちは経験不足によるものですから安心してください。過ち
を犯すことで時間を浪費してしまうでしょうが、それも全て学習過程の一部なのです。

間違いは避けられないものなのですから、インターフェイスを追加する前にまずコードが適切に
機能しているのかを確かめて下さい。不必要なインターフェイスコードを書く時間の無駄は省きた
いでしょう？インターフェイス作成に時間をかける必要はないと言っているわけではありません。
しかしスクリプトは設計どおりに動作しないのですから、全ての強力なインターフェイスの一部も
しくは全部を処分するより負担が重くなることなどないのです！



可能であれば、インターフェイスを書き始める前に、ダミーオブジェクトと固定値を持つ変数で
動作させるようなスクリプトを書いてみてください。コントロールの代わりにこれらの値を使用す
ることによって、主なタスク、スクリプトの記述とテストなどに焦点を当てることが出来るように
なります。この作業が終了すれば、実際の処理を全て行う根本となるコードが固定されることにな
り、コードは想定どおり機能することが保証されます。スクリプトの機能を担うコードを書き終え
バグからも開放された時点で、インターフェイスの作成を始めてください。この時点で、機能を全
て備え、全てが実装された後でもそれほど変更を加えないテストコードを基に、インターフェイス
コードを書きます。このように作業することで、より効率的に時間を使用できます。

規則 #2: シンプルさを保つ
インターフェイスは視覚的な手段ですから、大なり小なり視覚的なものです。スクリプトやプラ

グインにはどれも素晴らしいデザインを持つインターフェイスを割り当てることが出来ます。イン
ターフェイスは画面上にあまりの多くのボタンがあっても、もしくは配置がばらばらで混乱させる
ようなデザインであっても、目障りなものになってしまいます。実際、デザインが悪いとツールは
難しくなり、場合によっては使用するのが苦痛でさえあります。

スクリーン上の情報量を最小化することで、インターフェイスは使いやすくなるのです。仕切り
線を追加することでコントロールを系統立てて配置したり、タブを追加することでリクエスタのサ
イズ全体を小さくすることが出来ます。このようなトリックは必要であればあるほど複雑になって
きます。

インターフェイスコードを書き始める前に、ユーザーから取得しなければならない値のリストを
作成しておくのが良いでしょう。このリストから設定や変数を固定しておいて、コードの機能をテ
ストするようにしてください（規則 #1参照）。その情報からどのコントロールが必要となるのかを
確定します。その後でリクエスタ上でどう配置するのか、タブを使ってコントロールをグループ化
するのかなどを確定していきます。インターフェイスの輪郭はこのように考えていって下さい。

規則 #3: 美しさよりも速度
"Less is More"（過ぎたるは及ばざるが如し/出来るだけシンプルに）という黄金律を極限まで

極めます。スクリプトにインターフェイスが必要でなければ、作る必要はありません！出来る限り
作らないで下さい。スクリプトをより動的に使いやすくするためにインターフェイスを作成するの
です。やるべきことは、速くそして効率の良いツールを作ることなのです。もう一つ、にわか仕立
ての処理段階を追加することほど、ユーザーを悩ませるものはありません。常識を用いてスマート
に設計するようにして下さい。一日に一回、もしくは一月に一回走らせるかどうかぐらいの"一度限
り"のスクリプトであれば、一日に何百回も使用するスクリプトのように効率良くする必要はないの
です。

規則 #4: 一度に一つのアイテム
パネルのサイズであろうとコントロールの配置であろうと、一度に一つのアイテムだけのコード

を書いてください。そうすれば頻繁に動作をテスト出来るようになり、間違いも早く見つけること
が出来ます。インターフェイスのデバッグに費やす時間を大幅に短縮出来、より簡単になります。
一度に多くのものを実装しようとすると、簡単な間違いでバグ探しに莫大な時間を費やしてしまい
かねません。

規則 #5: 業界標準
どのコントロールを使用するのか、画面上でそのコントロールをどう整えていくのかを決めるの

は芸術形式です。大半のソフト会社には、ソフトウェアインターフェイスの設計と実装を担当する
専門家がいるものです。彼らはクリエイティブな処理を手助けするために何年もの間研究を重ね、
顧客からのフィードバックを元にインターフェイスを設計しています。こういった人々から多くの
ことが学べるはずです。事実、もう学んでいるはずです。

コンピュータ上で数え切れない時間を過ごすうちに、ソフトウェアがどのように動作するのか想
定できるようになります。WindowsであれMacであれ、一つのアプリケーションから別のアプリ
ケーションを通して、ある特定の関数が同じ動作を行っています。これはあなた自身の創作物にも
有効であるはずです。LightWaveのプラグインやその他のスクリプトが、インターフェイスをどの
ように設定しているのかに注意を払うのはさらに重要です。ユーザーはそれらコンポーネントの対
話に慣れており、想定どおりのインターフェイスであれば、スクリプトをより受け入れやすいから
です。

規則 #6: 研究例
お好みのスクリプトのインターフェイスを見つけたり、どのように動作しているのか不思議に

思ったら、コードを調べてみてください。理解したい領域、自身のスクリプトへと組み込みたいた
めに研究が必要となる部分のコードを解析してみてください。これはインターフェイスに限らず
コーディング一般に関して言えることです。時間をかけて他の素晴らしいスクリプト作品を学んで
下さい。

注意
謝辞を授与すべき箇所には忘れずに記述するようにして下さい。あなた独自のものでないアイデアや
コードの一部を使用している場合には、コメントの形でオリジナルの著者の謝辞を与えるようにして
下さい（インターフェイス上ではなくコード内部で述べること）。

規則 #7: ユーザーの批判に耳を傾ける
アニメーターは、開発者が決して想定していなかったような方法でコードやインターフェイスを

使おうとします。コードの目的が理解できないと考えるのではなく、その正反対に捉えるようにし
て下さい。アニメーター達はスクリプトの限界を押し上げようとしているのです。対象とする聴衆
と対話を始めるのと同時に、彼らのコメントを、次期バージョンのスクリプトのための新鮮なヒン
トとして利用してください。これらがスクリプトのアイデアやベータテストコードとして跳ね返り、
これらの人々は大変貴重なリソースとなるのです。

規則 #8: あなた自身が一番の批評家
自分のスクリプトを使用し、実際にテストしていってください。自身のコードのデバッグに役立

つと共に、自らスクリプトを実行することで即時にフィードバックが提供されます。インターフェ
イスの動作方式が自分で気に入らないようであれば、LightWaveコミュニティでもその動作方式が
気に入られない可能性が高いでしょう。

注意
この意見の逆が必ずしも真であるとは限りません。インターフェイスの動作方式が自分では気に入っ
ていても、コミュニティも気に入るとは保証出来ないのです。

第２４章：LSide  24.1

第24章 : LSIDE

LScript Integrated Development Environment（LSIDE）は、LScriptプログラマがスクリプトを
作成するためのツール群の一つです。

LightWaveのツールキットと LScript内部で開発された技術に対し、LSIDEは完全にプラット
フォーム独立です。つまり、LSIDE内にあるツールは全てLightWaveをサポートしている全ての
プラットフォームで上で同じ動作を行うということです。

このセクションで見ていく通り、LSIDEのツールもまた完全に統合されています。これらはワー
クフロー向上のため、スクリプトの開発処理におけるキーとなる時間に相互に通信します。LSIDE
における統合されたツールを使用するにつれ、スクリプト作成はより速く簡単になります。

LScript Editor（エディター）
まず最初に、LSIDEの基幹をなすLScript Editor（エディター）を見ていきましょう。どのテキ

ストエディタでもスクリプトを作成することは出来ますが、LScript Editorの威力を目の前にしたら、
このエディタを使用して全スクリプトを作りたくなることでしょう。以下のリストでは素晴らしい
機能を紹介しています。

・色構文ハイライト化

・マウスサポート

・ LScript テンプレート

・複数ドキュメント

・ LScript構文チェック

・検索置換

・ブロックによるテキスト移動

・アンドゥの複数レベル

・ドキュメントをまたがっての検索置換

LScript Editorの主となるエリアは 5つあります。メニュー、テキスト、コマンド、メッセージ
それに状態エリアです。それでは、それぞれのエリアがどういう動作をするのか順番に見ていきま
しょう。



メニューエリア

想像どおり、メニューエリアを構成しているのは主に4つのポップアップです。

図 24-1. LScript Editor におけるメニューエリア

Fileメニューではテキストエディタからファイルを扱えます。新規ファイルの作成やファイルの
保存、別名で保存、ファイルを閉じるなどの動作です。

このメニューにある唯一新規タイプのコマンドはToggle Writeコマンドです。このコマンドで
は不注意により上書きしないよう、スクリプトの書き込みを許可するトグルを使用できます。いく
つものスクリプトを開いている時に、これは非常に便利な機能となります。不測のエラーを作る機
会を無くします。

EditメニューにはUndoやDelete、Cut、Copy、PasteそれにSearch and Replaceがあります。
Cut、CopyそれにDeleteの機能は、使用する前にハイライト状態になっているテキストが必要で
す。処理を実行したいアイテムを選択し、Editメニューからオプションを選択します。例えばコー
ドのある一行を削除したい場合には行を選択してからEdit > Deleteを選択します。これらのコマン
ドはWindowsのクリップボードとともに動作します。

図 24-2. Edit メニュー

Viewメニューでは、エディタ内部のビューオプションを変更することが出来ます。まず最初の
オプションはFontで表示エリアに対するフォントのサイズを変更することが出来ます。サイズを増
減させます。

図 24-3. Fontは文字サイズを変更します

HighlightセクションではLScirpt特有の構文を判読しやすいように色付けするかどうかを設定で
きます。syntaxをハイライトにするよう選択すると、様々なタイプのコードがそれぞれ目立つよう
に異なる色で表示されます。ハイライトの種類はコードを通して探しやすくし、リファレンスとし
てカラーを使用することが出来ます。

図 24-4. Highlight オプションを使用することでコードを目立たせます

Toolメニューには、エディターにおいて最も重要なツールの一つであるTemplatesオプション
があります。

図 24-5. スクリプト開始時にはテンプレートを選択

お望みのタイプのスクリプトに対し事前に作成されたテンプレートを取得することが出来ます。
スクリプト構造に関する心配をする必要がないわけですから、テンプレートを使うのはスクリプト
処理をあっという間に始められる素晴らしい方法です。

図 24-6. Colorsパネルでコードの色を調整

Colorsパネル上では、テキストの色調整を行うためインターフェイス上で見えている通りにカ
ラーを修正することが可能です。参照するカラーでコメントが表示されるように、また命令子をさ
らに目立たせるためにインターフェイスを修正することが出来ます。またエディターに対して
Background Imageを追加することも出来、背景内部の方向を設定することも可能です。

でも待って下さい、これで全てじゃないんです。Toolsメニューからスクリプトの構文チェック
も行えます。スクリプトの構文チェックを行えば、スクリプトの処理時間を向上できます。
LightWaveにおけるスクリプト実行の必要がなくなり、20もの行の終端にセミコロンを付け忘れた
のではないかと探すこともなくなるので、生産性が向上します。

LScript Editorはメモリ内に一度に複数のスクリプトを置いておけるため、Nextボタンを使用し
て読み込まれているスクリプト全ての間を切り替えることが可能です。

図 24-7. Next ボタンで読み込まれているスクリプトを移動

もちろん、アプリケーションを終了するときにはQuitを使用します。

図 24-8. Quit ボタンでLScript Editor を終了

第２４章：LSide  24.5



テキストエリア

LScript Editorのメインウィンドウはテキストエリアです。このウィンドウ内部で全スクリプト
を編集します。どのワードプロセッサで編集されたテキストを持ってきても、問題なく内容を表示
することが出来ます。実際、LScriptエディタはどのテキストファイルでも読み込んで編集すること
が可能です。

図 24-9. スクリプト編集用のテキストエリア

編集エリアにおける独自の機能は、メインとなるテキストエリアの上にある二つのドロップ
メニューにあります。

左側のドロップメニューはスクリプトドロップメニューです。メニューエリアにあるNextボタ
ン同様、このコントロールを使用してメモリに読み込まれているスクリプト間を切り替えることが
出来ます。Nextボタンとは異なり、順に切り替えるのではなくリストから自由に任意のスクリプト
を選択することが出来ます。

またスクリプトドロップメニューはスクリプトの状態を知らせるためにスクリプト名称の頭に文
字を配置します。

* : アスタリスクはスクリプトが修正されまだ保存されていないことを表しています。

^ : キャラットはスクリプトが読取専用であることを示しています。

図 24-10. 上: スクリプトドロップメニュー 下: 関数リスト

右側のドロップメニューは関数リストです。スクリプトドロップメニューと同様、リスト内の関
数名称にジャンプすることが出来ます。ここでの違いは、アクティブスクリプト内部において定義
されている関数があればその関数へとジャンプすることができることです。Genericスクリプトの
テンプレートがLSEDに読み込まれている場合、ユーザー定義関数はありませんから、リストには
generic()関数のみが表示されます。

コマンドエリア

テキストエリアの下にあるフィールドがコマンドエリアで、特定のコマンドを入力する場所で
す。

図 24-11. 利用可能なボタンを使用する代わりにコマンドを入力するコマンドエリア

コマンドエリアでは、LSEDに対し何を処理させるのかを正確に指定することが出来ます。この
エリアは、厄介な小さなボタンを使用するよりも直接コマンドを打ち込みたい人のためのエリア
です。

以下がCommand入力ウィンドウで認識されるコマンドとなります。

goto <行番号 >

open

s~<検索文字列 >~

s~<検索文字列 >~<置換文字列 >~

home

end

match (brace, bracket or parentheses)

filename

new

close

インターフェイス上のボタンでも上記コマンドと同じ動作が出来ますので上記コマンドはは使用
しないかもしれませんが、ある一つのコマンドはとても役に立つでしょう。

help

helpコマンドを入力すると以下のLSED用のキーボードショートカットが画面に溢れ出します。

CTRL+\ カレントの編集ドキュメントとコマンドウィンドウを切り替えます

CTRL+] 一致する括弧文字にジャンプします

CTRL+[ カレントスクリプトにおいてハイライトになっている構文をトグルします

SHIFT+TAB 選択されているブロックを 4文字分左に移動します
TAB 選択されているブロックを 4文字分右に移動します
CTRL+F7 選択されているブロックを 1文字分左に移動します
CTRL+F8 選択されているブロックを 1文字分右に移動します
CTRL+X 選択されているブロックをクリップボードへ切り取ります

CTRL+C 選択されているブロックをクリップボードへコピーします

CTRL+V クリップボード内のコンテンツをカレントのドキュメントへ貼り付けます

CTRL+D 選択されているブロックを削除します

CTRL+Q 終了します

CTRL+O 新規ドキュメントを開きます

CTRL+S カレントのドキュメントに修正が加えられていれば保存します

CTRL+R バッファとそれに割り当てられているファイルの書き込み状態をトグルします

CTRL+N 次のドキュメントへ切り替えます

CTRL+G カレントドキュメント内にある指定した行番号にジャンプします

CTRL+1-9 最初の 9個のドキュメントの一つを選択します
CTRL+F9 フォントサイズを大きくします

CTRL+F10 フォントサイズを小さくします

メッセージエリア

LSIDEエディタの一番下の部分がメッセージエリアです。

図 24-12. スクリプトに関するメッセージを表示するメッセージエリア

LSEDがドキュメントに関して何かを伝える必要がある場合には、このメッセージエリアに表示
されます。

ポジションエリアとモードエリア

ウィンドウ右下はポジションエリアとモードエリアです。ポジションエリアにはスクリプト内に
おける現在のカーソル位置が表示されています。位置は行番号と列番号で表されています。

図 24-13. 左側がポジションエリア、右側がモードエリア

モードエリアでは InsertモードかまたはOverwriteモードかを表示します。Insertモードでは編
集時においてカーソルの位置に文字を挿入します。Overwriteモードでは編集時に文字を上書きし
ます。

LScript Interface Designer（インターフェイスデザイナー）
LScript Interface Designer（LSID）は"見たままのものをお手元に"（WYSIWYG）というコンセ

プトの基に作られた、スクリプト用のインターフェイスをデザインできるツールです。LSIDを使用
すると、リクエスタコントロールを編集することが出来ます。インタラクティブに好きな場所へコ
ントロールを配置し、複雑なインターフェイスを作成する時間を劇的に減らすことが可能です。

図 24-22. LScript Interface Designer(インターフェイスデザイナー)

他のLSIDEツールと同様、LScript Interface Designerは幾つかのエリアに分けられます。メ
ニューエリア、コンポーネントツリーエリア、それにシステムメッセージエリアです。それぞれの
エリアを見ていきましょう。

メニューエリア

Interface Designerにあるメニューはインターフェイスを処理する関数に対応しています。まず
始めにインターフェイス用の空白テンプレートを作成するか、以前に保存してあるインターフェイ
スを読み込まなくてはなりません。

図 24-23. LScript Interface Designer のメニューエリア

テンプレートを読み込んだり保存するためのコマンドはFileメニューの下にあり、自己説明形式
のメニューとなっています。ここでの一つ大きな新機能として、サブメニューExportがあります。
SaveやSave Asメニューオプションが Interface Designerだけが読み込み可能なインターフェイス
ファイルを保存するのに対し、Exportは実際のスクリプトコードを保存します。

図 24-24. 左: File メニューにおけるオプション. 右: Export メニューと利用可能なオプション

Interface DesignerとLScript Editorが双方とも開かれている場合、Export関数を使用してイン
ターフェイスをエディターへと直接エクスポートすることが可能です。Export関数に書き込むコー
ドタイプ（モデラーLScript、レイアウトLScriptまたはPanel）を指定します。Panelを選択した
場合、LSIDはCのコードをエクスポートします。

コンポーネントメニューでは様々なインターフェイスコンポーネントを選択できます。インター
フェイス上に追加したいコンポーネントを選択するだけで、テンプレート上に作成されます。

図 24-25. コンポーネントメニューから追加出来る使用可能なコンポーネント

ツールメニューではインターフェイスに対し新規レイヤーを追加できます。スクリプトの見え方
には影響を及ぼしませんが、個別のレイヤーに似たようなボタンを挿入することが出来るため、複
数のボタンを持つインターフェイスをより簡単に編集できるようになります。これはモデラーにお
けるレイヤーと同じようなものです。各レイヤーにはオブジェクトの個別の部品がありますが、全
て同一オブジェクトの一部とみなされます。

図 24-26. ツールメニューの New Layer オプション

インターフェイス上のコンポーネント群を選択して整列させることが出来ます。Alignのサブメ
ニューではどのように整列させるのかを選択します。

Align（整列）オプション

垂直方向にコントロールを積み上げたい場合には、Left または Right で整列させます。

図 24-27. Left: 整列させるアイテムを選択. Right: 右側にアイテムが整列

コントロールを水平方向に修正したい場合には、Top または Bottom で整列可能です。

図 24-28. Top: 整列させるアイテムを選択. Bottom: 下に合わせて整列

SHIFTを押しながらクリックするとコントロールを複数選択できます。Spaceメニューオプショ
ンは、水平方向または垂直方向に複数のコントロールを均等に配置します。

図 24-29. Space メニュー

注意
整列メニューツールと一緒に矢印キーを使用すれば、一度に位置ピクセルずつ矢印の方向へとコント
ロールをずらすことが出来ます。

コンポーネントツリー

コンポーネントツリーではインターフェイスの外観を組織的に見ることが出来ます。左側の列に
は二つのトグルがあります。その一つであるQuillは現在どのレイヤーを修正しているのかを示し、
もう一つのトグルEyeballは、各レイヤーを可視状態にします。レイヤーの可視状態は Interface
Designerにおいてのみ影響を受けます。レイヤーを見えない状態のままでエクスポートしても、コ
ントロールはインターフェイスの一部としてエクスポートされます。ですから必要のないコント
ロールはエクスポートする前に削除しておいてください。

図 24-30. コンポーネントツリー

コントロールを拡張することにより、任意のコントロールの編集可能な値にすべてアクセスでき
ます。矢印を下向きにクリックすることでコントロールを拡張します。値を編集するには修正を加
えたい値をダブルクリックするだけでダイアログボックスが表示されます。

図 24-31. 矢印が下向きになるようにクリックするとコントロールが拡張され値が見えます

図 24-32. 値をダブルクリックするとダイアログボックスが開きます

コンポーネントツリーのもう一つのすばらしい機能として、コントロールを他のコントロールに
対し親子関係を設定することが出来ます。この機能はLightWaveのScene Editor（シーン編集）で
アイテムに親子関係を設定するのと同じ方法で動作します。コンポーネントツリーウィンドウでは
子コンポーネントを親コンポーネントの下にドラッグします。すると親アイテムの下に黄色い線が
表示されます。 選択したアイテムに親子関係がつけられたことを示すため、コンポーネントは少し
ずれて表示されます。

図 24-33. 一番目のコンポーネントNumberの下にある短い線は子アイテムを示しています

親アイテムを選択したとき、インターフェース上では子アイテムにそれぞれ直線が引かれます。

図 24-34. 親子関係を示すNumberコンポーネント間の直線

今は親アイテムが移動しても子アイテムはその場に張り付いたままになっています。これは
ctlgroup()コマンドのインタラクティブ版です。LSIDにおいてインターフェイスを作成すると
きに親子付けがすべて発生するために、ctlgroup()関数はエクスポートされるコードに含まれ
ません。

タブつきのインターフェイスを作成する場合には、親子関係もまた重要な要素となります。親子
関係を使用して各タブにコントロールを配置することが出来るのです。

第２４章：LSide  24.17



タブと親子付けを見てみましょう。

1 まずはTabコントロールを追加します。

図 24-35. Layer 1に Tabコントロールを追加

2 初期状態でタブが 3つありますから、それぞれのタブに異なるコントロール、Booleanに
NumberそれにChoiceを追加しましょう。

図 24-36. Boolean、NumberそれにChoiceコントロールを追加

24.18 LScript 日本語ユーザーガイド



3 何もコントロールが割り当てられていませんので、インターフェイス上のすべてが可視状態に
なっています。コントロールをタブへと親子付けするために、コントロールをクリックし Item
の行までドラッグします。たとえば、BooleanをTab1に、NumberをTab2に、ChoiceをTab3に
挿入してみると、インターフェイスとコンポーネントツリーは以下の画像のようになっているは
ずです。

図 24-37. コントロールをそれぞれ Item行の下へドラッグ

4 各タブの下にあるコントロールを見るため、タブをクリックしてみてください。ただしツリー
ビューにおいて個々のタブを選択しなければなりません。作業ダイアログにおいて個別のタブを
選択することは出来ないのです。

図 24-38. Tab1をクリックしてコントロールを確認

図 24-39. Tab2を選択

図 24-40. Tab2をクリックしてコントロールを確認

図 24-41. Tab3を選択

図 24-42. Tab3をクリックしてコントロールを確認

5 Export > Layout LScriptを選択すると、LScript Editorへとインターフェイスを送信することが
出来ます。もしくはEditorが現在起動していない場合には、LSIDは生成されたコードをディス
クファイルへと保存するために保存名称の入力を促します。

図 24-43. Export > Layout LScript を選択

LSIDは以下の画像にあるようなコードを作成しますので、コードを切り取り、スクリプト内に
貼り付けることが出来ます。

図 24-44. エクスポート時に作成されるコード

この時点で少し時間を取り、Interface Designer（.id）ファイルとしてダイアログを保存するの
がよいでしょう。そうすれば後日、Designerでプロジェクトを再編集することが可能になるからで
す。先ほど編集・エクスポートして生成されたコードは、スクリプト内部においてのみ使用可能に
なっています。これは今まで作業してきたプロジェクトを保存するということではありません。

メッセージエリア

ウィンドウ下にあるメッセージエリアはLSIDEのシステムがユーザーに対し関連する任意の情
報を提供するエリアとなります。

